
Getting Started

This guide is an introduction to Ecto,
the database wrapper and query generator for Elixir. Ecto provides a
standardised API and a set of abstractions for talking to all the different
kinds of databases, so that Elixir developers can query whatever database
they're using by employing similar constructs.

In this guide, we're going to learn some basics about Ecto, such as creating,
reading, updating and destroying records from a PostgreSQL database. If you want
to see the code from this guide, you can view it at ecto/examples/friends on GitHub.

This guide will require you to have setup PostgreSQL beforehand.

Adding Ecto to an application

To start off with, we'll generate a new Elixir application by running this command:

mix new friends --sup

The --sup option ensures that this application has a supervision tree, which we'll need for Ecto a little later on.

To add Ecto to this application, there are a few steps that we need to take. The first step will be adding Ecto and a driver called Postgrex to our mix.exs file, which we'll do by changing the deps definition in that file to this:

defp deps do
  [
    {:ecto_sql, "~> 3.0"},
    {:postgrex, ">= 0.0.0"}
  ]
end

Ecto provides the common querying API, but we need the Postgrex driver installed too, as that is what Ecto uses to speak in terms a PostgreSQL database can understand. Ecto talks to its own Ecto.Adapters.Postgres module, which then in turn talks to the postgrex package to talk to PostgreSQL.

To install these dependencies, we will run this command:

mix deps.get

The Postgrex application will receive queries from Ecto and execute them
against our database. If we didn't do this step, we wouldn't be able to do any
querying at all.

That's the first two steps taken now. We have installed Ecto and Postgrex as
dependencies of our application. We now need to setup some configuration for
Ecto so that we can perform actions on a database from within the
application's code.

We can set up this configuration by running this command:

mix ecto.gen.repo -r Friends.Repo

This command will generate the configuration required to connect to a database. The first bit of configuration is in config/config.exs:

config :friends, Friends.Repo,
  database: "friends_repo",
  username: "user",
  password: "pass",
  hostname: "localhost"

NOTE: Your PostgreSQL database may be setup to


	not require a username and password. If the above configuration doesn't work, try removing the username and password fields, or setting them both to "postgres".


	be running on a non-standard port. The default port is 5432. You can specify your specific port by adding it to the config: e.g. port: 15432.




This configures how Ecto will connect to our database, called "friends". Specifically, it configures a "repo". More information about Ecto.Repo can be found in its documentation.

The Friends.Repo module is defined in lib/friends/repo.ex by our mix ecto.gen.repo command:

defmodule Friends.Repo do
  use Ecto.Repo,
    otp_app: :friends,
    adapter: Ecto.Adapters.Postgres
end

This module is what we'll be using to query our database shortly. It uses the Ecto.Repo module, and the otp_app tells Ecto which Elixir application it can look for database configuration in. In this case, we've specified that it is the :friends application where Ecto can find that configuration and so Ecto will use the configuration that was set up in config/config.exs. Finally, we configure the database :adapter to Postgres.

The final piece of configuration is to setup the Friends.Repo as a supervisor within the application's supervision tree, which we can do in lib/friends/application.ex, inside the start/2 function:

def start(_type, _args) do
  children = [
    Friends.Repo,
  ]

  ...

This piece of configuration will start the Ecto process which receives and executes our application's queries. Without it, we wouldn't be able to query the database at all!

There's one final bit of configuration that we'll need to add ourselves, since the generator does not add it. Underneath the configuration in config/config.exs, add this line:

config :friends, ecto_repos: [Friends.Repo]

This tells our application about the repo, which will allow us to run commands such as mix ecto.create very soon.

We've now configured our application so that it's able to make queries to our database. Let's now create our database, add a table to it, and then perform some queries.

Setting up the database

To be able to query a database, it first needs to exist. We can create the database with this command:

mix ecto.create

If the database has been created successfully, then you will see this message:

The database for Friends.Repo has been created.

NOTE: If you get an error, you should try changing your configuration in config/config.exs, as it may be an authentication error.

A database by itself isn't very queryable, so we will need to create a table within that database. To do that, we'll use what's referred to as a migration. If you've come from Active Record (or similar), you will have seen these before. A migration is a single step in the process of constructing your database.

Let's create a migration now with this command:

mix ecto.gen.migration create_people

This command will generate a brand new migration file in priv/repo/migrations, which is empty by default:

defmodule Friends.Repo.Migrations.CreatePeople do
  use Ecto.Migration

  def change do

  end
end

Let's add some code to this migration to create a new table called "people", with a few columns in it:

defmodule Friends.Repo.Migrations.CreatePeople do
  use Ecto.Migration

  def change do
    create table(:people) do
      add :first_name, :string
      add :last_name, :string
      add :age, :integer
    end
  end
end

This new code will tell Ecto to create a new table called people, and add three new fields: first_name, last_name and age to that table. The types of these fields are string and integer. (The different types that Ecto supports are covered in the Ecto.Schema documentation.)

NOTE: The naming convention for tables in Ecto databases is to use a pluralized name.

To run this migration and create the people table in our database, we will run this command:

mix ecto.migrate

If we found out that we made a mistake in this migration, we could run mix ecto.rollback to undo the changes in the migration. We could then fix the changes in the migration and run mix ecto.migrate again. If we ran mix ecto.rollback now, it would delete the table that we just created.

We now have a table created in our database. The next step that we'll need to do is to create the schema.

Creating the schema

The schema is an Elixir representation of data from our database. Schemas are commonly associated with a database table, however they can be associated with a database view as well.

Let's create the schema within our application at lib/friends/person.ex:

defmodule Friends.Person do
  use Ecto.Schema

  schema "people" do
    field :first_name, :string
    field :last_name, :string
    field :age, :integer
  end
end

This defines the schema from the database that this schema maps to. In this case, we're telling Ecto that the Friends.Person schema maps to the people table in the database, and the first_name, last_name and age fields in that table. The second argument passed to field tells Ecto how we want the information from the database to be represented in our schema.

We've called this schema Person because the naming convention in Ecto for schemas is a singularized name.

We can play around with this schema in an IEx session by starting one up with iex -S mix and then running this code in it:

person = %Friends.Person{}

This code will give us a new Friends.Person struct, which will have nil values for all the fields. We can set values on these fields by generating a new struct:

person = %Friends.Person{age: 28}

Or with syntax like this:

person = %{person | age: 28}

We can retrieve values using this syntax:

person.age # => 28

Let's take a look at how we can insert data into the database.

Inserting data

We can insert a new record into our people table with this code:

person = %Friends.Person{}
Friends.Repo.insert(person)

To insert the data into our database, we call insert on Friends.Repo, which is the module that uses Ecto to talk to our database. This function tells Ecto that we want to insert a new Friends.Person record into the database corresponding with Friends.Repo. The person struct here represents the data that we want to insert into the database.

A successful insertion will return a tuple, like so:

{:ok,
 %Friends.Person{__meta__: #Ecto.Schema.Metadata<:loaded>, age: nil,
  first_name: nil, id: 1, last_name: nil}}

The :ok atom can be used for pattern matching purposes to ensure that the insertion succeeds. A situation where the insertion may not succeed is if you have a constraint on the database itself. For instance, if the database had a unique constraint on a field called email so that an email can only be used for one person record, then the insertion would fail.

You may wish to pattern match on the tuple in order to refer to the record inserted into the database:

{:ok, person} = Friends.Repo.insert person

Validating changes

In Ecto, you may wish to validate changes before they go to the database. For instance, you may wish that a person has both a first name and a last name before a record can be entered into the database. For this, Ecto has changesets.

Let's add a changeset to our Friends.Person module inside lib/friends/person.ex now:

def changeset(person, params \\ %{}) do
  person
  |> Ecto.Changeset.cast(params, [:first_name, :last_name, :age])
  |> Ecto.Changeset.validate_required([:first_name, :last_name])
end

This changeset takes a person and a set of params, which are to be the changes to apply to this person. The changeset function first casts the first_name, last_name and age keys from the parameters passed in to the changeset. Casting tells the changeset what parameters are allowed to be passed through in this changeset, and anything not in the list will be ignored.

On the next line, we call validate_required which says that, for this changeset, we expect first_name and last_name to have values specified. Let's use this changeset to attempt to create a new record without a first_name and last_name:

person = %Friends.Person{}
changeset = Friends.Person.changeset(person, %{})
Friends.Repo.insert(changeset)

On the first line here, we get a struct from the Friends.Person module. We know what that does, because we saw it not too long ago. On the second line we do something brand new: we define a changeset. This changeset says that on the specified person object, we're looking to make some changes. In this case, we're not looking to change anything at all.

On the final line, rather than inserting the person, we insert the changeset. The changeset knows about the person, the changes and the validation rules that must be met before the data can be entered into the database. When this third line runs, we'll see this:

{:error,
 #Ecto.Changeset<action: :insert, changes: %{},
  errors: [first_name: "can't be blank", last_name: "can't be blank"],
  data: #Friends.Person<>, valid?: false>}

Just like the last time we did an insertion, this returns a tuple. This time however, the first element in the tuple is :error, which indicates something bad happened. The specifics of what happened are included in the changeset which is returned. We can access these by doing some pattern matching:

{:error, changeset} = Friends.Repo.insert(changeset)

Then we can get to the errors by doing changeset.errors:

[first_name: "can't be blank", last_name: "can't be blank"]

And we can ask the changeset itself if it is valid, even before doing an insertion:

changeset.valid?
#=> false

Since this changeset has errors, no new record was inserted into the people
table.

Let's try now with some valid data.

person = %Friends.Person{}
changeset = Friends.Person.changeset(person, %{first_name: "Ryan", last_name: "Bigg"})

We start out here with a normal Friends.Person struct. We then create a changeset for that person which has a first_name and a last_name parameter specified. At this point, we can ask the changeset if it has errors:

changeset.errors
#=> []

And we can ask if it's valid or not:

changeset.valid?
#=> true

The changeset does not have errors, and is valid. Therefore if we try to insert this changeset it will work:

Friends.Repo.insert(changeset)
#=> {:ok,
     %Friends.Person{__meta__: #Ecto.Schema.Metadata<:loaded>, age: nil,
      first_name: "Ryan", id: 3, last_name: "Bigg"}}

Due to Friends.Repo.insert returning a tuple, we can use a case to determine different code paths depending on what happens:

case Friends.Repo.insert(changeset) do
  {:ok, person} ->
    # do something with person
  {:error, changeset} ->
    # do something with changeset
end

NOTE: changeset.valid? will not check constraints (such as uniqueness_constraint). For that, you will need to attempt to do an insertion and check for errors from the database. It's for this reason it's best practice to try inserting data and validate the returned tuple from Friends.Repo.insert to get the correct errors, as prior to insertion the changeset will only contain validation errors from the application itself.

If the insertion of the changeset succeeds, then you can do whatever you wish with the person returned in that result. If it fails, then you have access to the changeset and its errors. In the failure case, you may wish to present these errors to the end user. The errors in the changeset are a keyword list that looks like this:

[first_name: {"can't be blank", []},
 last_name: {"can't be blank", []}]

The first element of the tuple is the validation message, and the second element is a keyword list of options for the validation message. The validate_required/3 validations don't return any options, but other methods such as validate_length/3 do. Imagine that we had a field called bio that we were validating, and that field has to be longer than 15 characters. This is what would be returned:

[first_name: {"can't be blank", []},
 last_name: {"can't be blank", []},
 bio: {"should be at least %{count} characters", [count: 15]}]

To display these error messages in a human friendly way, we can use Ecto.Changeset.traverse_errors/2:

traverse_errors(changeset, fn {msg, opts} ->
  Enum.reduce(opts, msg, fn {key, value}, acc ->
    String.replace(acc, "%{#{key}}", to_string(value))
  end)
end)

This will return the following for the errors shown above:

%{
  first_name: ["can't be blank"],
  last_name: ["can't be blank"],
  bio: ["should be at least 15 characters"],
}

One more final thing to mention here: you can trigger an exception to be thrown by using Friends.Repo.insert!/2. If a changeset is invalid, you will see an Ecto.InvalidChangesetError exception. Here's a quick example of that:

Friends.Repo.insert! Friends.Person.changeset(%Friends.Person{}, %{first_name: "Ryan"})

** (Ecto.InvalidChangesetError) could not perform insert because changeset is invalid.

* Changeset changes

%{first_name: "Ryan"}

* Changeset params

%{"first_name" => "Ryan"}

* Changeset errors

[last_name: "can't be blank"]

    lib/ecto/repo/schema.ex:111: Ecto.Repo.Schema.insert!/4

This exception shows us the changes from the changeset, and how the changeset is invalid. This can be useful if you want to insert a bunch of data and then have an exception raised if that data is not inserted correctly at all.

Now that we've covered inserting data into the database, let's look at how we can pull that data back out.

Our first queries

Querying a database requires two steps in Ecto. First, we must construct the query and then we must execute that query against the database by passing the query to the repository. Before we do this, let's re-create the database for our app and setup some test data. To re-create the database, we'll run these commands:

mix ecto.drop
mix ecto.create
mix ecto.migrate

Then to create the test data, we'll run this in an iex -S mix session:

people = [
  %Friends.Person{first_name: "Ryan", last_name: "Bigg", age: 28},
  %Friends.Person{first_name: "John", last_name: "Smith", age: 27},
  %Friends.Person{first_name: "Jane", last_name: "Smith", age: 26},
]

Enum.each(people, fn (person) -> Friends.Repo.insert(person) end)

This code will create three new people in our database, Ryan, John and Jane. Note here that we could've used a changeset to validate the data going into the database, but the choice was made not to use one.

We'll be querying for these people in this section. Let's jump in!

Fetching a single record

Let's start off with fetching just one record from our people table:

Friends.Person |> Ecto.Query.first

That code will generate an Ecto.Query, which will be this:

#Ecto.Query<from p in Friends.Person, order_by: [asc: p.id], limit: 1>

The code between the angle brackets <...> here shows the Ecto query which has been constructed. We could construct this query ourselves with almost exactly the same syntax:

require Ecto.Query
Ecto.Query.from p in Friends.Person, order_by: [asc: p.id], limit: 1

We need to require Ecto.Query here to enable the macros from that module. Then it's a matter of calling the from function from Ecto.Query and passing in the code from between the angle brackets. As we can see here, Ecto.Query.first saves us from having to specify the order and limit for the query.

To execute the query that we've just constructed, we can call Friends.Repo.one:

Friends.Person |> Ecto.Query.first |> Friends.Repo.one

The one function retrieves just one record from our database and returns a new struct from the Friends.Person module:

%Friends.Person{__meta__: #Ecto.Schema.Metadata<:loaded>, age: 28,
 first_name: "Ryan", id: 1, last_name: "Bigg"}

Similar to first, there is also last:

Friends.Person |> Ecto.Query.last |> Friends.Repo.one
#=> %Friends.Person{__meta__: #Ecto.Schema.Metadata<:loaded>, age: 26,
     first_name: "Jane", id: 3, last_name: "Smith"}

The Ecto.Repo.one function will only return a struct if there is one record in the
result from the database. If there is more than one record returned, an
Ecto.MultipleResultsError exception will be thrown. Some code that would
cause that issue to happen is:

Friends.Person |> Friends.Repo.one

We've left out the Ecto.Query.first here, and so there is no limit or order clause applied to the executed query. We'll see the executed query in the debug log:

[timestamp] [debug] SELECT p0."id", p0."first_name", p0."last_name", p0."age" FROM "people" AS p0 [] OK query=1.8ms

Then immediately after that, we will see the Ecto.MultipleResultsError exception:

** (Ecto.MultipleResultsError) expected at most one result but got 3 in query:

from p in Friends.Person

    lib/ecto/repo/queryable.ex:67: Ecto.Repo.Queryable.one/4

This happens because Ecto doesn't know what one record out of all the records
returned that we want. Ecto will only return a result if we are explicit in
our querying about which result we want.

If there is no record which matches the query, one will return nil.

Fetching all records

To fetch all records from the schema, Ecto provides the all function:

Friends.Person |> Friends.Repo.all

This will return a Friends.Person struct representation of all the records that currently exist within our people table:

[%Friends.Person{__meta__: #Ecto.Schema.Metadata<:loaded>, age: 28,
  first_name: "Ryan", id: 1, last_name: "Bigg"},
 %Friends.Person{__meta__: #Ecto.Schema.Metadata<:loaded>, age: 27,
  first_name: "John", id: 2, last_name: "Smith"},
 %Friends.Person{__meta__: #Ecto.Schema.Metadata<:loaded>, age: 26,
  first_name: "Jane", id: 3, last_name: "Smith"}]

Fetch a single record based on ID

To fetch a record based on its ID, you use the get function:

Friends.Person |> Friends.Repo.get(1)
#=> %Friends.Person{__meta__: #Ecto.Schema.Metadata<:loaded>, age: 28,
     first_name: "Ryan", id: 1, last_name: "Bigg"}

Fetch a single record based on a specific attribute

If we want to get a record based on something other than the id attribute, we can use get_by:

 Friends.Person |> Friends.Repo.get_by(first_name: "Ryan")
 #=> %Friends.Person{__meta__: #Ecto.Schema.Metadata<:loaded>, age: 28,
      first_name: "Ryan", id: 1, last_name: "Bigg"}

Filtering results

If we want to get multiple records matching a specific attribute, we can use where:

Friends.Person |> Ecto.Query.where(last_name: "Smith") |> Friends.Repo.all

[%Friends.Person{__meta__: #Ecto.Schema.Metadata<:loaded>, age: 27,
  first_name: "John", id: 2, last_name: "Smith"},
 %Friends.Person{__meta__: #Ecto.Schema.Metadata<:loaded>, age: 26,
  first_name: "Jane", id: 3, last_name: "Smith"}]

If we leave off the Friends.Repo.all on the end of this, we will see the query Ecto generates:

#Ecto.Query<from p in Friends.Person, where: p.last_name == "Smith">

We can also use this query syntax to fetch these same records:

Ecto.Query.from(p in Friends.Person, where: p.last_name == "Smith") |> Friends.Repo.all

One important thing to note with both query syntaxes is that they require variables to be pinned, using the pin operator (^). Otherwise, this happens:

last_name = "Smith"
Friends.Person |> Ecto.Query.where(last_name: last_name) |> Friends.Repo.all

** (Ecto.Query.CompileError) variable `last_name` is not a valid query expression.
  Variables need to be explicitly interpolated in queries with ^
             expanding macro: Ecto.Query.where/2
             iex:1: (file)
    (elixir) expanding macro: Kernel.|>/2
             iex:1: (file)

The same will happen in the longer query syntax too:

Ecto.Query.from(p in Friends.Person, where: p.last_name == last_name) |> Friends.Repo.all

** (Ecto.Query.CompileError) variable `last_name` is not a valid query expression.
  Variables need to be explicitly interpolated in queries with ^
             expanding macro: Ecto.Query.where/3
             iex:1: (file)
             expanding macro: Ecto.Query.from/2
             iex:1: (file)
    (elixir) expanding macro: Kernel.|>/2
             iex:1: (file)

To get around this, we use the pin operator (^):

last_name = "Smith"
Friends.Person |> Ecto.Query.where(last_name: ^last_name) |> Friends.Repo.all

Or:

last_name = "Smith"
Ecto.Query.from(p in Friends.Person, where: p.last_name == ^last_name) |> Friends.Repo.all

The pin operator instructs the query builder to use parameterised SQL queries protecting against SQL injection.

Composing Ecto queries

Ecto queries don't have to be built in one spot. They can be built up by calling Ecto.Query functions on existing queries. For instance, if we want to find all people with the last name "Smith", we can do:

query = Friends.Person |> Ecto.Query.where(last_name: "Smith")

If we want to scope this down further to only people with the first name of "Jane", we can do this:

query = query |> Ecto.Query.where(first_name: "Jane")

Our query will now have two where clauses in it:

#Ecto.Query<from p in Friends.Person, where: p.last_name == "Smith",
 where: p.first_name == "Jane">

This can be useful if you want to do something with the first query, and then build off that query later on.

Updating records

Updating records in Ecto requires us to first fetch a record from the database. We then create a changeset from that record and the changes we want to make to that record, and then call the Ecto.Repo.update function.

Let's fetch the first person from our database and change their age. First, we'll fetch the person:

person = Friends.Person |> Ecto.Query.first |> Friends.Repo.one

Next, we'll build a changeset. We need to build a changeset because if we just create a new Friends.Person struct with the new age, Ecto wouldn't be able to know that the age has changed without inspecting the database. Let's build that changeset:

changeset = Friends.Person.changeset(person, %{age: 29})

This changeset will inform the database that we want to update the record to have the age set to 29. To tell the database about the change we want to make, we run this command:

Friends.Repo.update(changeset)

Just like Friends.Repo.insert, Friends.Repo.update will return a tuple:

{:ok,
 %Friends.Person{__meta__: #Ecto.Schema.Metadata<:loaded>, age: 29,
  first_name: "Ryan", id: 1, last_name: "Bigg"}}

If the changeset fails for any reason, the result of Friends.Repo.update will be {:error, changeset}. We can see this in action by passing through a blank first_name in our changeset's parameters:

changeset = Friends.Person.changeset(person, %{first_name: ""})
Friends.Repo.update(changeset)
#=> {:error,
     #Ecto.Changeset<action: :update, changes: %{first_name: ""},
      errors: [first_name: "can't be blank"], data: #Friends.Person<>,
      valid?: false>}

This means that you can also use a case statement to do different things depending on the outcome of the update function:

case Friends.Repo.update(changeset) do
  {:ok, person} ->
    # do something with person
  {:error, changeset} ->
    # do something with changeset
end

Similar to insert!, there is also update! which will raise an exception if the changeset is invalid:

changeset = Friends.Person.changeset(person, %{first_name: ""})
Friends.Repo.update! changeset

** (Ecto.InvalidChangesetError) could not perform update because changeset is invalid.

* Changeset changes

%{first_name: ""}

* Changeset params

%{"first_name" => ""}

* Changeset errors

[first_name: {"can't be blank", []}]

    lib/ecto/repo/schema.ex:132: Ecto.Repo.Schema.update!/4

Deleting records

We've now covered creating (insert), reading (get, get_by, where) and updating records. The last thing that we'll cover in this guide is how to delete a record using Ecto.

Similar to updating, we must first fetch a record from the database and then call Friends.Repo.delete to delete that record:

person = Friends.Repo.get(Friends.Person, 1)
Friends.Repo.delete(person)
#=> {:ok,
 %Friends.Person{__meta__: #Ecto.Schema.Metadata<:deleted>, age: 29,
  first_name: "Ryan", id: 2, last_name: "Bigg"}}

Similar to insert and update, delete returns a tuple. If the deletion succeeds, then the first element in the tuple will be :ok, but if it fails then it will be an :error.

  
Testing with Ecto

After you have successfully set up your database connection with Ecto for your application,
its usage for your tests requires further changes, especially if you want to leverage the
Ecto SQL Sandbox that allows you to run tests that talk to the database concurrently.

Create the config/test.exs file or append the following content:

use Mix.Config

config :my_app, MyApp.Repo,
  username: "postgres",
  password: "postgres",
  database: "myapp_test",
  hostname: "localhost",
  pool: Ecto.Adapters.SQL.Sandbox


Thereby, we configure the database connection for our test setup.
In this case, we use a Postgres database and set it up to use the sandbox pool that will wrap each test in a transaction.

Make sure we import the configuration for the test environment at the very bottom of config/config.exs:

import_config "#{Mix.env()}.exs"

We also need to add an explicit statement to the end of test/test_helper.exs about the sandbox mode:

Ecto.Adapters.SQL.Sandbox.mode(MyApp.Repo, :manual)

Lastly, you need to establish the database connection ahead of your tests.
You can enable it either for all of your test cases by extending the ExUnit template or by setting it up individually for each test. Let's start with the former and place it to the test/support/repo_case.ex:

defmodule MyApp.RepoCase do
  use ExUnit.CaseTemplate

  using do
    quote do
      alias MyApp.Repo

      import Ecto
      import Ecto.Query
      import MyApp.RepoCase

      # and any other stuff
    end
  end

  setup tags do
    :ok = Ecto.Adapters.SQL.Sandbox.checkout(MyApp.Repo)

    unless tags[:async] do
      Ecto.Adapters.SQL.Sandbox.mode(MyApp.Repo, {:shared, self()})
    end

    :ok
  end
end

The case template above brings Ecto and Ecto.Query functions into your tests and checkouts a database connection. It also enables a shared sandbox connection mode in case the test is not running asynchronously. See Ecto.Adapters.SQL.Sandbox for more information.

To add test/support/ folder for compilation in test environment we need to update mix.exs configuration

  def project do
    [
      # ...
      elixirc_paths: elixirc_paths(Mix.env())
    ]
  end

  # Specifies which paths to compile per environment.
  defp elixirc_paths(:test), do: ["lib", "test/support"]
  defp elixirc_paths(_), do: ["lib"]

And then in each test that uses the repository:

defmodule MyApp.MyTest do
  use MyApp.RepoCase

  # Tests etc...
end

In case you don't want to define a "case template", you can checkout on each individual case:

defmodule MyApp.MyTest do
  use ExUnit.Case

  setup do
    :ok = Ecto.Adapters.SQL.Sandbox.checkout(MyApp.Repo)
  end

  # Tests etc...
end

For convenience reasons, you can also define aliases to automatically set up your database at the execution of your tests.
Change the following content in your mix.exs.


  def project do
    [app: :my_app,

     ...

     aliases: aliases()]
  end

  defp aliases do
    [ ...
     "test": ["ecto.create --quiet", "ecto.migrate", "test"]
    ]
  end

  
Aggregates and subqueries

Now it's time to discuss aggregates and subqueries. As we will learn, one builds directly on the other.

Aggregates

Ecto includes a convenience function in repositories to calculate aggregates.

For example, if we assume every post has an integer column named visits, we can find the average number of visits across all posts with:

MyApp.Repo.aggregate(MyApp.Post, :avg, :visits)
#=> #Decimal<1743>

Behind the scenes, the query above translates to:

MyApp.Repo.one(from p in MyApp.Post, select: avg(p.visits))

The Ecto.Repo.aggregate/4 function supports any of the aggregate operations listed in the Ecto.Query.API module.

At first, it looks like the implementation of aggregate/4 is quite straight-forward. You could even start to wonder why it was added to Ecto in the first place. However, complexities start to arise on queries that rely on limit, offset or distinct clauses.

Imagine that instead of calculating the average of all posts, you want the average of only the top 10. Your first try may be:

MyApp.Repo.one(
  from p in MyApp.Post,
    order_by: [desc: :visits],
    limit: 10,
    select: avg(p.visits)
)
#=> #Decimal<1743>

Oops. The query above returned the same value as the queries before. The option limit: 10 has no effect here since it is limiting the aggregated result and queries with aggregates return only a single row anyway. In order to retrieve the correct result, we would need to first find the top 10 posts and only then aggregate. That's exactly what aggregate/4 does:

query =
  from MyApp.Post,
    order_by: [desc: :visits],
    limit: 10

MyApp.Repo.aggregate(query, :avg, :visits)
#=> #Decimal<4682>

When limit, offset or distinct is specified in the query, aggregate/4 automatically wraps the given query in a subquery. Therefore the query executed by aggregate/4 above is rather equivalent to:

inner_query =
  from MyApp.Post,
    order_by: [desc: :visits],
    limit: 10

query =
  from q in subquery(inner_query),
  select: avg(q.visits)

MyApp.Repo.one(query)

Let's take a closer look at subqueries.

Subqueries

In the previous section we have already learned some queries that would be hard to express without support for subqueries. That's one of many examples that caused subqueries to be added to Ecto.

Subqueries in Ecto are created by calling Ecto.Query.subquery/1. This function receives any data structure that can be converted to a query, via the Ecto.Queryable protocol, and returns a subquery construct (which is also queryable).

In Ecto, it is allowed for a subquery to select a whole table (p) or a field (p.field). All fields selected in a subquery can be accessed from the parent query. Let's revisit the aggregate query we saw in the previous section:

inner_query =
  from MyApp.Post,
    order_by: [desc: :visits],
    limit: 10

query =
  from q in subquery(inner_query),
    select: avg(q.visits)

MyApp.Repo.one(query)

Because the query does not specify a :select clause, it will return select: p where p is controlled by MyApp.Post schema. Since the query will return all fields in MyApp.Post, when we convert it to a subquery, all of the fields from MyApp.Post will be available on the parent query, such as q.visits. In fact, Ecto will keep the schema properties across queries. For example, if you write q.field_that_does_not_exist, your Ecto query won't compile.

Ecto also allows an Elixir map to be returned from a subquery, making the map keys directly available to the parent query.

Let's see one last example. Imagine you manage a library (as in an actual library in the real world) and there is a table that logs every time the library lends a book. The "lendings" table uses an auto-incrementing primary key and can be backed by the following schema:

defmodule Library.Lending do
  use Ecto.Schema

  schema "lendings" do
    belongs_to :book, MyApp.Book       # defines book_id
    belongs_to :visitor, MyApp.Visitor # defines visitor_id
  end
end

Now consider we want to retrieve the name of every book alongside the name of the last person the library has lent it to. To do so, we need to find the last lending ID of every book, and then join on the book and visitor tables. With subqueries, that's straight-forward:

last_lendings =
  from l in MyApp.Lending,
    group_by: l.book_id,
    select: %{
      book_id: l.book_id,
      last_lending_id: max(l.id)
    }

from l in Lending,
  join: last in subquery(last_lendings),
  on: last.last_lending_id == l.id,
  join: b in assoc(l, :book),
  join: v in assoc(l, :visitor),
  select: {b.name, v.name}

  
Composable transactions with Multi

Ecto relies on database transactions when multiple operations must be performed atomically. The most common example used for transaction are bank transfers between two people:

Repo.transaction(fn ->
  mary_update =
    from Account,
      where: [id: ^mary.id],
      update: [inc: [balance: +10]]

  {1, _} = Repo.update_all(mary_update)

  john_update =
    from Account,
      where: [id: ^john.id],
      update: [inc: [balance: -10]]

  {1, _} = Repo.update_all(john_update)
end)

In Ecto, transactions can be performed via the Repo.transaction function. When we expect both operations to succeed, as above, transactions are quite straight-forward. However, transactions get more complicated if we need to check the status of each operation along the way:

Repo.transaction(fn ->
  mary_update =
    from Account,
      where: [id: ^mary.id],
      update: [inc: [balance: +10]]

  case Repo.update_all query do
    {1, _} ->
      john_update =
        from Account,
          where: [id: ^john.id],
          update: [inc: [balance: -10]]

      case Repo.update_all query do
        {1, _} -> {mary, john}
        {_, _} -> Repo.rollback({:failed_transfer, john})
      end
    {_, _} ->
      Repo.rollback({:failed_transfer, mary})
  end
end)

Transactions in Ecto can also be nested arbitrarily. For example, imagine the transaction above is moved into its own function that receives both accounts, defined as transfer_money(mary, john, 10), and besides transferring money we also want to log the transfer:

Repo.transaction(fn ->
  case transfer_money(mary, john, 10) do
    {:ok, {mary, john}} ->
      transfer = %Transfer{
        from: mary.id,
        to: john.id,
        amount: 10
      }

      Repo.insert!(transfer)

    {:error, error} ->
      Repo.rollback(error)
  end
end)

The snippet above starts a transaction and then calls transfer_money/3 that also runs in a transaction. In case of multiple transactions, they are all flattened, which means a failure in an inner transaction causes the outer transaction to also fail. That's why matching and rolling back on {:error, error} is important.

While nesting transactions can improve the code readability by breaking large transactions into multiple smaller transactions, there is still a lot of boilerplate involved in handling the success and failure scenarios. Furthermore, composition is quite limited, as all operations must still be performed inside transaction blocks.

A more declarative approach when working with transactions would be to define all operations we want to perform in a transaction decoupled from the transaction execution. This way we would be able to compose transactions operations without worrying about its execution context or about each individual success/failure scenario. That's exactly what Ecto.Multi allows us to do.

Composing with data structures

Let's rewrite the snippets above using Ecto.Multi. The first snippet that transfers money between mary and john can rewritten to:

mary_update =
  from Account,
    where: [id: ^mary.id],
    update: [inc: [balance: +10]]

john_update =
  from Account,
    where: [id: ^john.id],
    update: [inc: [balance: -10]]

Ecto.Multi.new()
|> Ecto.Multi.update_all(:mary, mary_update)
|> Ecto.Multi.update_all(:john, john_update)

Ecto.Multi is a data structure that defines multiple operations that must be performed together, without worrying about when they will be executed. Ecto.Multi mirrors most of the Ecto.Repo API, with the difference each operation must be explicitly named. In the example above, we have defined two update operations, named :mary and :john. As we will see later, the names are important when handling the transaction results.

Since Ecto.Multi is just a data structure, we can pass it as argument to other functions, as well as return it. Assuming the multi above is moved into its own function, defined as transfer_money(mary, john, value),  we can add a new operation to the multi that logs the transfer as follows:

transfer = %Transfer{
  from: mary.id,
  to: john.id,
  amount: 10
}

transfer_money(mary, john, 10)
|> Ecto.Multi.insert(:transfer, transfer)

This is considerably simpler than the nested transaction approach we have seen earlier. Once all operations are defined in the multi, we can finally call Repo.transaction, this time passing the multi:

transfer = %Transfer{
  from: mary.id,
  to: john.id,
  amount: 10
}

transfer_money(mary, john, 10)
|> Ecto.Multi.insert(:transfer, transfer)
|> Repo.transaction()
|> case do
  {:ok, %{transfer: transfer}} ->
    # Handle success case
  {:error, name, value, changes_so_far} ->
    # Handle failure case
end

If all operations in the multi succeed, it returns {:ok, map} where the map contains the name of all operations as keys and their success value. If any operation in the multi fails, the transaction is rolled back and Repo.transaction returns {:error, name, value, changes_so_far}, where name is the name of the failed operation, value is the failure value and changes_so_far is a map of the previously successful multi operations that have been rolled back due to the failure.

In other words, Ecto.Multi takes care of all the flow control boilerplate while decoupling the transaction definition from its execution, allowing us to compose operations as needed.

Dependent values

Besides operations such as insert, update and delete, Ecto.Multi also provides functions for handling more complex scenarios. For example, prepend and append can be used to merge multis together. And more generally, the Ecto.Multi.run/3 and Ecto.Multi.run/5 can be used to define any operation that depends on the results of a previous multi operation.

Let's study a more practical example. In Constraints and Upserts, we want to modify a post while possibly giving it a list of tags as a string separated by commas. At the end of the guide, we present a solution that insert any missing tag and then fetch all of them using only two queries:

defmodule MyApp.Post do
  use Ecto.Schema

  # Schema is the same
  schema "posts" do
    field :title
    field :body
    many_to_many :tags, MyApp.Tag,
      join_through: "posts_tags",
      on_replace: :delete
    timestamps()
  end

  # Changeset is the same
  def changeset(struct, params \\ %{}) do
    struct
    |> Ecto.Changeset.cast(params, [:title, :body])
    |> Ecto.Changeset.put_assoc(:tags, parse_tags(params))
  end

  # Parse tags has slightly changed
  defp parse_tags(params)  do
    (params["tags"] || "")
    |> String.split(",")
    |> Enum.map(&String.trim/1)
    |> Enum.reject(& &1 == "")
    |> insert_and_get_all()
  end

  defp insert_and_get_all([]) do
    []
  end
  defp insert_and_get_all(names) do
    timestamp = NaiveDateTime.utc_now() |> NaiveDateTime.truncate(:second)
    maps = Enum.map(names, &%{name: &1, inserted_at: timestamp, updated_at: timestamp})
    Repo.insert_all MyApp.Tag, maps, on_conflict: :nothing
    Repo.all from t in MyApp.Tag, where: t.name in ^names
  end
end

While insert_and_get_all/1 is idempotent, allowing us to run it multiple times and get the same result back, it does not run inside a transaction, so any failure while attempting to modify the parent post struct would end-up creating tags that have no posts associated to them.

Let's fix the problem above by introducing using Ecto.Multi. Let's start by splitting the logic into both Post and Tag modules and keeping it free from side-effects such as database operations:

defmodule MyApp.Post do
  use Ecto.Schema

  schema "posts" do
    field :title
    field :body
    many_to_many :tags, MyApp.Tag,
      join_through: "posts_tags",
      on_replace: :delete
    timestamps()
  end

  def changeset(struct, tags, params) do
    struct
    |> Ecto.Changeset.cast(params, [:title, :body])
    |> Ecto.Changeset.put_assoc(:tags, tags)
  end
end

defmodule MyApp.Tag do
  use Ecto.Schema

  schema "tags" do
    field :name
    timestamps()
  end

  def parse(tags) do
    (tags || "")
    |> String.split(",")
    |> Enum.map(&String.trim/1)
    |> Enum.reject(& &1 == "")
  end
end

Now, whenever we need to introduce a post with tags, we can create a multi that wraps all operations and the repository access:

alias MyApp.Tag

def insert_or_update_post_with_tags(post, params) do
  Ecto.Multi.new()
  |> Ecto.Multi.run(:tags, fn _, changes ->
    insert_and_get_all_tags(changes, params)
  end)
  |> Ecto.Multi.run(:post, fn _, changes ->
    insert_or_update_post(changes, post, params)
  end)
  |> Repo.transaction()
end

defp insert_and_get_all_tags(_changes, params) do
  case MyApp.Tag.parse(params["tags"]) do
    [] ->
      {:ok, []}
    names ->
      timestamp = NaiveDateTime.utc_now() |> NaiveDateTime.truncate(:second)
      maps = Enum.map(names, &%{name: &1, inserted_at: timestamp, updated_at: timestamp})
      Repo.insert_all(Tag, maps, on_conflict: :nothing)
      query = from t in Tag, where: t.name in ^names
      {:ok, Repo.all(query)}
  end
end

defp insert_or_update_post(%{tags: tags}, post, params) do
  post = MyApp.Post.changeset(post, tags, params)
  Repo.insert_or_update post
end

In the example above we have used Ecto.Multi.run/3 twice, albeit for two different reasons.


	In Ecto.Multi.run(:tags, ...), we used run/3 because we need to perform both insert_all and all operations, and while the multi exposes Ecto.Multi.insert_all/4, it does not yet expose a Ecto.Multi.all/3. Whenever we need to perform a repository operation that is not supported by Ecto.Multi, we can always fallback to run/3 or run/5.



	In Ecto.Multi.run(:post, ...), we used run/3 because we need to access the value of a previous multi operation. The function given to run/3 receives, as second argument, a map with the results of the operations performed so far. To grab the tags returned in the previous step, we simply pattern match on %{tags: tags} on insert_or_update_post.





Note: The first argument received by the function given to run/3 is the repo in which the transaction is executing.



While run/3 is very handy when we need to go beyond the functionalities provided natively by Ecto.Multi, it has the downside that operations defined with Ecto.Multi.run/3 are opaque and therefore they cannot be inspected by functions such as Ecto.Multi.to_list/1. Still, Ecto.Multi allows us to greatly simplify control flow logic and remove boilerplate when working with transactions.

  
Constraints and Upserts

In this guide we will learn how to use constraints and upserts. To showcase those features, we will work on a practical scenario: which is by studying a many to many relationship between posts and tags.

put_assoc vs cast_assoc

Imagine we are building an application that has blog posts and such posts may have many tags. Not only that, a given tag may also belong to many posts. This is a classic scenario where we would use many_to_many associations. Our migrations would look like:

create table(:posts) do
  add :title, :string
  add :body, :text
  timestamps()
end

create table(:tags) do
  add :name, :string
  timestamps()
end

create unique_index(:tags, [:name])

create table(:posts_tags, primary_key: false) do
  add :post_id, references(:posts)
  add :tag_id, references(:tags)
end

Note we added a unique index to the tag name because we don't want to have duplicated tags in our database. It is important to add an index at the database level instead of using a validation since there is always a chance two tags with the same name would be validated and inserted simultaneously, passing the validation and leading to duplicated entries.

Now let's also imagine we want the user to input such tags as a list of words split by comma, such as: "elixir, erlang, ecto". Once this data is received in the server, we will break it apart into multiple tags and associate them to the post, creating any tag that does not yet exist in the database.

While the constraints above sound reasonable, that's exactly what put us in trouble with cast_assoc/3. The cast_assoc/3 changeset function was designed to receive external parameters and compare them with the associated data in our structs.To do so correctly, Ecto requires tags to be sent as a list of maps. We can see an example of this in Polymorphic associations with many to many. However, here we expect tags to be sent in a string separated by comma.

Furthermore, cast_assoc/3 relies on the primary key field for each tag sent in order to decide if it should be inserted, updated or deleted. Again, because the user is simply passing a string, we don't have the ID information at hand.

When we can't cope with cast_assoc/3, it is time to use put_assoc/4. In put_assoc/4, we give Ecto structs or changesets instead of parameters, giving us the ability to manipulate the data as we want. Let's define the schema and the changeset function for a post which may receive tags as a string:

defmodule MyApp.Post do
  use Ecto.Schema

  schema "posts" do
    field :title
    field :body

    many_to_many :tags, MyApp.Tag,
      join_through: "posts_tags",
      on_replace: :delete

    timestamps()
  end

  def changeset(struct, params \\ %{}) do
    struct
    |> Ecto.Changeset.cast(params, [:title, :body])
    |> Ecto.Changeset.put_assoc(:tags, parse_tags(params))
  end

  defp parse_tags(params)  do
    (params["tags"] || "")
    |> String.split(",")
    |> Enum.map(&String.trim/1)
    |> Enum.reject(& &1 == "")
    |> Enum.map(&get_or_insert_tag/1)
  end

  defp get_or_insert_tag(name) do
    Repo.get_by(MyApp.Tag, name: name) ||
      Repo.insert!(MyApp.Tag, %Tag{name: name})
  end
end

In the changeset function above, we moved all the handling of tags to a separate function, called parse_tags/1, which checks for the parameter, breaks each tag apart via String.split/2, then removes any left over whitespace with String.trim/1, rejects any empty string and finally checks if the tag exists in the database or not, creating one in case none exists.

The parse_tags/1 function is going to return a list of MyApp.Tag structs which are then passed to put_assoc/4. By calling put_assoc/4, we are telling Ecto those should be the tags associated to the post from now on. In case a previous tag was associated to the post and not given in put_assoc/4, Ecto will invoke the behaviour defined in the :on_replace option, which we have set to :delete. The :delete behaviour will remove the association between the post and the removed tag from the database.

And that's all we need to use many_to_many associations with put_assoc/4. put_assoc/4 is very useful when we want to have more explicit control over our associations and it also works with has_many, belongs_to and all others association types.

However, our code is not yet ready for production. Let's see why.

Constraints and race conditions

Remember we added a unique index to the tag :name column when creating the tags table. We did so to protect us from having duplicate tags in the database.

By adding the unique index and then using get_by  with a insert! to get or insert a tag, we introduced a potential error in our application. If two posts are submitted at the same time with a similar tag, there is a chance we will check if the tag exists at the same time, leading both submissions to believe there is no such tag in the database. When that happens, only one of the submissions will succeed while the other one will fail. That's a race condition: your code will error from time to time, only when certain conditions are met. And those conditions are time sensitive.

Luckily Ecto gives us a mechanism to handle constraint errors from the database.

Checking for constraint errors

Since our get_or_insert_tag(name) function fails when a tag already exists in the database, we need to handle such scenarios accordingly. Let's rewrite it taking race conditions into account:

defp get_or_insert_tag(name) do
  %Tag{}
  |> Ecto.Changeset.change(name: name)
  |> Ecto.Changeset.unique_constraint(:name)
  |> Repo.insert
  |> case do
    {:ok, tag} -> tag
    {:error, _} -> Repo.get_by!(MyApp.Tag, name: name)
  end
end

Instead of inserting the tag directly, we now build a changeset, which allows us to use the unique_constraint annotation. Now if the Repo.insert operation fails because the unique index for :name is violated, Ecto won't raise, but return an {:error, changeset} tuple. Therefore, if Repo.insert succeeds, it is because the tag was saved, otherwise the tag already exists, which we then fetch with Repo.get_by!.

While the mechanism above fixes the race condition, it is a quite expensive one: we need to perform two queries for every tag that already exists in the database: the (failed) insert and then the repository lookup. Given that's the most common scenario, we may want to rewrite it to the following:

defp get_or_insert_tag(name) do
  Repo.get_by(MyApp.Tag, name: name) ||
    maybe_insert_tag(name)
end

defp maybe_insert_tag(name) do
  %Tag{}
  |> Ecto.Changeset.change(name: name)
  |> Ecto.Changeset.unique_constraint(:name)
  |> Repo.insert
  |> case do
    {:ok, tag} -> tag
    {:error, _} -> Repo.get_by!(MyApp.Tag, name: name)
  end
end

The above performs 1 query for every tag that already exists, 2 queries for every new tag and possibly 3 queries in the case of race conditions. While the above would perform slightly better on average, Ecto has a better option in stock.

Upserts

Ecto supports the so-called "upsert" command which is an abbreviation for "update or insert". The idea is that we try to insert a record and in case it conflicts with an existing entry, for example due to a unique index, we can choose how we want the database to act by either raising an error (the default behaviour), ignoring the insert (no error) or by updating the conflicting database entries.

"upsert" in Ecto is done with the :on_conflict option. Let's rewrite get_or_insert_tag(name) once more but this time using the :on_conflict option. Remember that "upsert" is a new feature in PostgreSQL 9.5, so make sure you are up to date.

Your first try in using :on_conflict may be by setting it to :nothing, as below:

defp get_or_insert_tag(name) do
  Repo.insert!(
    %MyApp.Tag{name: name},
    on_conflict: :nothing
  )
end

While the above won't raise an error in case of conflicts, it also won't update the struct given, so it will return a tag without ID. One solution is to force an update to happen in case of conflicts, even if the update is about setting the tag name to its current name. In such cases, PostgreSQL also requires the :conflict_target option to be given, which is the column (or a list of columns) we are expecting the conflict to happen:

defp get_or_insert_tag(name) do
  Repo.insert!(
    %MyApp.Tag{name: name},
    on_conflict: [set: [name: name]],
    conflict_target: :name
  )
end

And that's it! We try to insert a tag with the given name and if such tag already exists, we tell Ecto to update its name to the current value, updating the tag and fetching its id. While the above is certainly a step up from all solutions so far, it still performs one query per tag. If 10 tags are sent, we will perform 10 queries. Can we further improve this?

Upserts and insert_all

Ecto accepts the :on_conflict option not only in Ecto.Repo.insert/2 but also in the Ecto.Repo.insert_all/3 function. This means we can build one query that attempts to insert all missing tags and then another query that fetches all of them at once. Let's see how our Post schema will look like after those changes:

defmodule MyApp.Post do
  use Ecto.Schema

  # Schema is the same
  schema "posts" do
    add :title
    add :body

    many_to_many :tags, MyApp.Tag,
      join_through: "posts_tags",
      on_replace: :delete

    timestamps()
  end

  # Changeset is the same
  def changeset(struct, params \\ %{}) do
    struct
    |> Ecto.Changeset.cast(params, [:title, :body])
    |> Ecto.Changeset.put_assoc(:tags, parse_tags(params))
  end

  # Parse tags has slightly changed
  defp parse_tags(params)  do
    (params["tags"] || "")
    |> String.split(",")
    |> Enum.map(&String.trim/1)
    |> Enum.reject(& &1 == "")
    |> insert_and_get_all()
  end

  defp insert_and_get_all([]) do
    []
  end
  defp insert_and_get_all(names) do
    timestamp = NaiveDateTime.utc_now() |> NaiveDateTime.truncate(:second)
    maps = Enum.map(names, &%{name: &1, inserted_at: timestamp, updated_at: timestamp})
    Repo.insert_all MyApp.Tag, maps, on_conflict: :nothing
    Repo.all from t in MyApp.Tag, where: t.name in ^names
  end
end

Instead of attempting to get and insert each tag individually, the code above work on all tags at once, first by building a list of maps which is given to insert_all and then by looking up all tags with the existing names. Therefore, regardless of how many tags are sent, we will perform only 2 queries (unless no tag is sent, in which we return an empty list back promptly). This solution is only possible thanks to the :on_conflict option, which guarantees insert_all won't fail in case a unique index is violated, such as duplicate tag names. Remember, insert_all won't autogenerate values like timestamps. That's why we must include inserted_at and updated_at manually.

Finally, keep in mind that we haven't used transactions in any of the examples so far. That decision was deliberate as we relied on the fact that getting or inserting tags is an idempotent operation, i.e. we can repeat it many times for a given input and it will always give us the same result back. Therefore, even if we fail to introduce the post to the database due to a validation error, the user will be free to resubmit the form and we will just attempt to get or insert the same tags once again. The downside of this approach is that tags will be created even if creating the post fails, which means some tags may not have posts associated to them. In case that's not desired, the whole operation could be wrapped in a transaction or modeled with the Ecto.Multi.

  
Data mapping and validation

We will take a look at the role schemas play when validating and casting data through changesets. As we will see, sometimes the best solution is not to completely avoid schemas, but break a large schema into smaller ones. Maybe one for reading data, another for writing. Maybe one for your database, another for your forms.

Schemas are mappers

The Ecto.Schema moduledoc says:

An Ecto schema is used to map any data source into an Elixir struct.



We put emphasis on any because it is a common misconception to think Ecto schemas map only to your database tables.

For instance, when you write a web application using Phoenix and you use Ecto to receive external changes and apply such changes to your database, we have this mapping:

Database <-> Ecto schema <-> Forms / API

Although there is a single Ecto schema mapping to both your database and your API, in many situations it is better to break this mapping in two. Let's see some practical examples.

Imagine you are working with a client that wants the "Sign Up" form to contain the fields "First name", "Last name" along side "E-mail" and other information. You know there are a couple problems with this approach.

First of all, not everyone has a first and last name. Although your client is decided on presenting both fields, they are a UI concern, and you don't want the UI to dictate the shape of your data. Furthermore, you know it would be useful to break the "Sign Up" information across two tables, the "accounts" and "profiles" tables.

Given the requirements above, how would we implement the Sign Up feature in the backend?

One approach would be to have two schemas, Account and Profile, with virtual fields such as first_name and last_name, and use associations along side nested forms to tie the schemas to your UI. One of such schemas would be:

defmodule Profile do
  use Ecto.Schema

  schema "profiles" do
    field :name
    field :first_name, :string, virtual: true
    field :last_name, :string, virtual: true
    ...
  end
end

It is not hard to see how we are polluting our Profile schema with UI requirements by adding fields such first_name and last_name. If the Profile schema is used for both reading and writing data, it may end-up in an awkward place where it is not useful for any, as it contains fields that map just to one or the other operation.

One alternative solution is to break the "Database <-> Ecto schema <-> Forms / API" mapping in two parts. The first will cast and validate the external data with its own structure which you then transform and write to the database. For such, let's define a schema named Registration that will take care of casting and validating the form data exclusively, mapping directly to the UI fields:

defmodule Registration do
  use Ecto.Schema

  embedded_schema do
    field :first_name
    field :last_name
    field :email
  end
end

We used embedded_schema because it is not our intent to persist it anywhere. With the schema in hand, we can use Ecto changesets and validations to process the data:

fields = [:first_name, :last_name, :email]

changeset =
  %Registration{}
  |> Ecto.Changeset.cast(params["sign_up"], fields)
  |> validate_required(...)
  |> validate_length(...)

Now that the registration changes are mapped and validated, we can check if the resulting changeset is valid and act accordingly:

if changeset.valid? do
  # Get the modified registration struct from changeset
  registration = Ecto.Changeset.apply_changes(changeset)
  account = Registration.to_account(registration)
  profile = Registration.to_profile(registration)

  MyApp.Repo.transaction fn ->
    MyApp.Repo.insert_all "accounts", [account]
    MyApp.Repo.insert_all "profiles", [profile]
  end

  {:ok, registration}
else
  # Annotate the action so the UI shows errors
  changeset = %{changeset | action: :registration}
  {:error, changeset}
end

The to_account/1 and to_profile/1 functions in Registration would receive the registration struct and split the attributes apart accordingly:

def to_account(registration) do
  Map.take(registration, [:email])
end

def to_profile(%{first_name: first, last_name: last}) do
  %{name: "#{first} #{last}"}
end

In the example above, by breaking apart the mapping between the database and Elixir and between Elixir and the UI, our code becomes clearer and our data structures simpler.

Note we have used MyApp.Repo.insert_all/2 to add data to both "accounts" and "profiles" tables directly. We have chosen to bypass schemas altogether. However, there is nothing stopping you from also defining both Account and Profile schemas and changing to_account/1 and to_profile/1 to respectively return %Account{} and %Profile{} structs. Once structs are returned, they could be inserted through the usual MyApp.Repo.insert/2 operation. Doing so can be especially useful if there are uniqueness or other constraints that you want to check during insertion.

Schemaless changesets

Although we chose to define a Registration schema to use in the changeset, Ecto also allows developers to use changesets without schemas. We can dynamically define the data and their types. Let's rewrite the registration changeset above to bypass schemas:

data = %{}
types = %{name: :string, email: :string}

# The data+types tuple is equivalent to %Registration{}
changeset =
  {data, types}
  |> Ecto.Changeset.cast(params["sign_up"], Map.keys(types))
  |> validate_required(...)
  |> validate_length(...)

You can use this technique to validate API endpoints, search forms, and other sources of data. The choice of using schemas depends mostly if you want to use the same mapping in different places or if you desire the compile-time guarantees Elixir structs gives you. Otherwise, you can bypass schemas altogether, be it when using changesets or interacting with the repository.

However, the most important lesson in this guide is not when to use or not to use schemas, but rather understand when a big problem can be broken into smaller problems that can be solved independently leading to an overall cleaner solution. The choice of using schemas or not above didn't affect the solution as much as the choice of breaking the registration problem apart.

  
Dynamic queries

Ecto was designed from the ground up to have an expressive query API that leverages Elixir syntax to write queries that are pre-compiled for performance and safety. When building queries, we may use the keywords syntax

import Ecto.Query

from p in Post,
  where: p.author == "José" and p.category == "Elixir",
  where: p.published_at > ^minimum_date,
  order_by: [desc: p.published_at]

or the pipe-based one

import Ecto.Query

Post
|> where([p], p.author == "José" and p.category == "Elixir")
|> where([p], p.published_at > ^minimum_date)
|> order_by([p], desc: p.published_at)

While many developers prefer the pipe-based syntax, having to repeat the binding p made it quite verbose compared to the keyword one.

Another problem with the pre-compiled query syntax is that it has limited options to compose the queries dynamically. Imagine for example a web application that provides search functionality on top of existing posts. The user should be able to specify multiple criteria, such as the author name, the post category, publishing interval, etc.

To solve those problems, Ecto also provides a data-structure centric API to build queries as well as a very powerful mechanism for dynamic queries. Let's take a look.

Focusing on data structures

Ecto provides a simpler API for both keyword and pipe based queries by making data structures first-class. Let's see an example:

from p in Post,
  where: [author: "José", category: "Elixir"],
  where: p.published_at > ^minimum_date,
  order_by: [desc: :published_at]

and

Post
|> where(author: "José", category: "Elixir")
|> where([p], p.published_at > ^minimum_date)
|> order_by(desc: :published_at)

Notice how we were able to ditch the p selector in most expressions. In Ecto, all constructs, from select and order_by to where and group_by, accept data structures as input. The data structure can be specified at compile-time, as above, and also dynamically at runtime, shown below:

where = [author: "José", category: "Elixir"]
order_by = [desc: :published_at]
Post
|> where(^where)
|> where([p], p.published_at > ^minimum_date)
|> order_by(^order_by)

While using data-structures already brings a good amount of flexibility to Ecto queries, not all expressions can be converted to data structures. For example, where converts a key-value to a key == value comparison, and therefore order-based comparisons such as p.published_at > ^minimum_date need to be written as before.

Dynamic fragments

For cases where we cannot rely on data structures but still desire to build queries dynamically, Ecto includes the Ecto.Query.dynamic/2 macro.

The dynamic macro allows us to conditionally build query fragments and interpolate them in the main query. For example, imagine that in the example above you may optionally filter posts by a date of publication. You could of course write it like this:

query =
  Post
  |> where(^where)
  |> order_by(^order_by)

query =
  if published_at = params["published_at"] do
    where(query, [p], p.published_at < ^published_at)
  else
    query
  end

But with dynamic fragments, you can also write it as:

where = [author: "José", category: "Elixir"]
order_by = [desc: :published_at]

filter_published_at =
  if published_at = params["published_at"] do
    dynamic([p], p.published_at < ^published_at)
  else
    true
  end

Post
|> where(^where)
|> where(^filter_published_at)
|> order_by(^order_by)

The dynamic macro allows us to build dynamic expressions that are later interpolated into the query. dynamic expressions can also be interpolated into dynamic expressions, allowing developers to build complex expressions dynamically without hassle.

By using dynamic fragments, we can decouple the processing of parameters from the query generation. Let's see a more complex example.

Building dynamic queries

Let's go back to the original problem. We want to build a search functionality where the user can configure how to traverse all posts in many different ways. For example, the user may choose how to order the data, filter by author and category, as well as select posts published after a certain date.

To tackle this in Ecto, we can break our problem into a bunch of small functions, that build either data structures or dynamic fragments, and then we interpolate it into the query:

def filter(params) do
  Post
  |> order_by(^filter_order_by(params["order_by"]))
  |> where(^filter_where(params))
end

def filter_order_by("published_at_desc"),
  do: dynamic([p], desc: p.published_at)

def filter_order_by("published_at"),
  do: dynamic([p], p.published_at)

def filter_order_by(_),
  do: []

def filter_where(params) do
  Enum.reduce(params, dynamic(true), fn
    {"author", value}, dynamic ->
      dynamic([p], ^dynamic and p.author == ^value)

    {"category", value}, dynamic ->
      dynamic([p], ^dynamic and p.category == ^value)

    {"published_at", value}, dynamic ->
      dynamic([p], ^dynamic and p.published_at > ^value)

    {_, _}, dynamic ->
      # Not a where parameter
      dynamic
  end)
end

Because we were able to break our problem into smaller functions that receive regular data structures, we can use all the tools available in Elixir to work with data. For handling the order_by parameter, it may be best to simply pattern match on the order_by parameter. For building the where clause, we can use reduce to start with an empty dynamic (that always returns true) and refine it with new conditions as we traverse the parameters.

Testing also becomes simpler as we can test each function in isolation, even when using dynamic queries:

test "filter published at based on the given date" do
  assert dynamic_match?(
           filter_where(%{}),
           "true"
         )

  assert dynamic_match?(
           filter_where(%{"published_at" => "2010-04-17"}),
           "true and q.published_at > ^\"2010-04-17\""
         )
end

defp dynamic_match?(dynamic, string) do
  inspect(dynamic) == "dynamic([q], #{string})"
end

In the example above, we created a small helper that allows us to assert on the dynamic contents by matching on the results of inspect(dynamic).

Dynamic and joins

Even query joins can be tackled dynamically. For example, let's do two modifications to the example above. Let's say we can also sort by author name ("author_name" and "author_name_desc") and at the same time let's say that authors are in a separate table, which means our authors filter in filter_where now need to go through the join table.

Our final solution would look like this:

def filter(params) do
  Post
  # 1. Add named join binding
  |> join([p], assoc(p, :authors), as: :authors)
  |> order_by(^filter_order_by(params["order_by"]))
  |> where(^filter_where(params))
end

# 2. Returned dynamic with join binding
def filter_order_by("published_at_desc"),
  do: dynamic([p], desc: p.published_at)

def filter_order_by("published_at"),
  do: dynamic([p], p.published_at)

def filter_order_by("author_name_desc"),
  do: dynamic([authors: a], desc: a.name)

def filter_order_by("author_name"),
  do: dynamic([authors: a], a.name)

def filter_order_by(_),
  do: []

# 3. Change the authors clause inside reduce
def filter_where(params) do
  Enum.reduce(params, dynamic(true), fn
    {"author", value}, dynamic ->
      dynamic([authors: a], ^dynamic and a.name == ^value)

    {"category", value}, dynamic ->
      dynamic([p], ^dynamic and p.category == ^value)

    {"published_at", value}, dynamic ->
      dynamic([p], ^dynamic and p.published_at > ^value)

    {_, _}, dynamic ->
      # Not a where parameter
      dynamic
  end)
end

Adding more filters in the future is simply a matter of adding more clauses to the Enum.reduce/3 call in filter_where.

  
Multi tenancy with query prefixes

With Ecto we can run queries in different prefixes using a single pool of database connections. For databases engines such as Postgres, Ecto's prefix maps to Postgres' DDL schemas. For MySQL, each prefix is a different database on its own.

Query prefixes may be useful in different scenarios. For example, multi tenant apps running on Postgres would define multiple prefixes, usually one per client, under a single database. The idea is that prefixes will provide data isolation between the different users of the application, guaranteeing either globally or at the data level that queries and commands act on a specific prefix.

Prefixes may also be useful on high-traffic applications where data is partitioned upfront. For example, a gaming platform may break game data into isolated partitions, each named after a different prefix. A partition for a given player is either chosen at random or calculated based on the player information.

While query prefixes were designed with the two scenarios above in mind, they may also be used in other circumstances, which we will explore throughout this guide. All the examples below assume you are using Postgres. Other databases engines may require slightly different solutions.

Connection prefixes

As a starting point, let's start with a simple scenario: your application must connect to a particular prefix when running in production. This may be due to infrastructure conditions, database administration rules or others.

Let's define a repository and a schema to get started:

# lib/repo.ex
defmodule MyApp.Repo do
  use Ecto.Repo,
    otp_app: :my_app,
    adapter: Ecto.Adapters.Postgres
end

# lib/sample.ex
defmodule MyApp.Sample do
  use Ecto.Schema

  schema "samples" do
    field :name
    timestamps
  end
end

Now let's configure the repository:

# config/config.exs
config :my_app, MyApp.Repo,
  username: "postgres",
  password: "postgres",
  database: "demo",
  hostname: "localhost",
  pool_size: 10

And define a migration:

# priv/repo/migrations/20160101000000_create_sample.exs
defmodule MyApp.Repo.Migrations.CreateSample do
  use Ecto.Migration

  def change do
    create table(:samples) do
      add :name, :string
      timestamps()
    end
  end
end

Now let's create the database, migrate it and then start an IEx session:

$ mix ecto.create
$ mix ecto.migrate
$ iex -S mix
Interactive Elixir - press Ctrl+C to exit
iex(1)> MyApp.Repo.all MyApp.Sample
[]

We haven't done anything unusual so far. We created our database instance, made it up to date by running migrations and then successfully made a query against the "samples" table, which returned an empty list.

By default, connections to Postgres' databases run on the "public" prefix. When we run migrations and queries, they are all running against the "public" prefix. However imagine your application has a requirement to run on a particular prefix in production, let's call it "connection_prefix".

Luckily Postgres allows us to change the prefix our database connections run on by setting the "schema search path". The best moment to change the search path is right after we setup the database connection, ensuring all of our queries will run on that particular prefix, throughout the connection life-cycle.

To do so, let's change our database configuration in "config/config.exs" and specify an :after_connect option. :after_connect expects a tuple with module, function and arguments it will invoke with the connection process, as soon as a database connection is established:

query_args = ["SET search_path TO connection_prefix", []]

config :my_app, MyApp.Repo,
  username: "postgres",
  password: "postgres",
  database: "demo_dev",
  hostname: "localhost",
  pool_size: 10,
  after_connect: {Postgrex, :query!, query_args}

Now let's try to run the same query as before:

$ iex -S mix
Interactive Elixir - press Ctrl+C to exit
iex(1)> MyApp.Repo.all MyApp.Sample
** (Postgrex.Error) ERROR (undefined_table):
   relation "samples" does not exist

Our previously successful query now fails because there is no table "samples" under the new prefix. Let's try to fix that by running migrations:

$ mix ecto.migrate
** (Postgrex.Error) ERROR (invalid_schema_name):
   no schema has been selected to create in

Oops. Now migration says there is no such schema name. That's because Postgres automatically creates the "public" prefix every time we create a new database. If we want to use a different prefix, we must explicitly create it on the database we are running on:

$ psql -d demo_dev -c "CREATE SCHEMA connection_prefix"

Now we are ready to migrate and run our queries:

$ mix ecto.migrate
$ iex -S mix
Interactive Elixir - press Ctrl+C to exit
iex(1)> MyApp.Repo.all MyApp.Sample
[]

Data in different prefixes are isolated. Writing to the "samples" table in one prefix cannot be accessed by the other unless we change the prefix in the connection or use the Ecto conveniences we will discuss next.

Schema prefixes

Ecto also allows you to set a particular schema to run on a specific prefix. Imagine you are building a multi-tenant application. Each client data belongs to a particular prefix, such as "client_foo", "client_bar" and so forth. Yet your application may still rely on a set of tables that are shared across all clients. One of such tables may be exactly the table that maps the Client ID to its database prefix. Let's assume we want to store this data in a prefix named "main":

defmodule MyApp.Mapping do
  use Ecto.Schema

  @schema_prefix "main"
  schema "mappings" do
    field :client_id, :integer
    field :db_prefix
    timestamps
  end
end

Now running MyApp.Repo.all MyApp.Mapping will by default run on the "main" prefix, regardless of the value configured for the connection on the :after_connect callback. Similar will happen to insert, update, and similar operations, the @schema_prefix is used unless the :prefix is explicitly changed via Ecto.put_meta/2 or by passing the :prefix option to the repository operation.

Per-query and per-struct prefixes

Now, suppose that while still configured to connect to the "connection_prefix" on :after_connect, we run the following queries:

iex(1) alias MyApp.Sample
MyApp.Sample
iex(2) MyApp.Repo.all Sample
[]
iex(3) MyApp.Repo.insert %Sample{name: "mary"}
{:ok, %MyApp.Sample{...}}
iex(4) MyApp.Repo.all Sample
[%MyApp.Sample{...}]

The operations above ran on the "connection_prefix". So what happens if we try to run the sample query on the "public" prefix? All Ecto repository operations support the :prefix option. So let's set it to public.

iex(7)> MyApp.Repo.all Sample
[%MyApp.Sample{...}]
iex(8)> MyApp.Repo.all Sample, prefix: "public"
[]

Notice how we were able to change the prefix the query runs on. Back in the default "public" prefix, there is no data.

One interesting aspect of prefixes in Ecto is that the prefix information is carried along each struct returned by a query:

iex(9) [sample] = MyApp.Repo.all Sample
[%MyApp.Sample{}]
iex(10)> Ecto.get_meta(sample, :prefix)
nil

The example above returned nil, which means no prefix was specified by Ecto, and therefore the database connection default will be used. In this case, "connection_prefix" will be used because of the :after_connect callback we added at the beginning of this guide.

Since the prefix data is carried in the struct, we can use such to copy data from one prefix to the other. Let's copy the sample above from the "connection_prefix" to the "public" one:

iex(11)> new_sample = Ecto.put_meta(sample, prefix: "public")
%MyApp.Sample{}
iex(12)> MyApp.Repo.insert new_sample
{:ok, %MyApp.Sample{}}
iex(13)> [sample] = MyApp.Repo.all Sample, prefix: "public"
[%MyApp.Sample{}]
iex(14)> Ecto.get_meta(sample, :prefix)
"public"

Now we have data inserted in both prefixes.

Prefixes in queries and structs always cascade. For example, if you run MyApp.Repo.preload(post, [:comments]), the association will be queried for and loaded in the same prefix as the post struct. If post has associations and you call MyApp.Repo.insert(post) or MyApp.Repo.update(post), the associated data will also be inserted/updated in the same prefix as post. That's by design to facilitate working with groups of data in the same prefix, and especially because data in different prefixes must be kept isolated.

Per from/join prefixes

Finally, Ecto allows you to set the prefix individually for each from and join expression. Here's an example:

from p in Post, prefix: "foo",
  join: c in Comment, prefix: "bar"

Those will take precedence over all other prefixes we have defined so far. For each join/from in the query, the prefix used will be determined by the following order:


	If the prefix option is given exclusively to join/from


	If the @schema_prefix is set in the related schema


	If the :prefix field given to the repo operation (i.e. Repo.all query, prefix: prefix)


	The connection prefix




Migration prefixes

When the connection prefix is set, it also changes the prefix migrations run on. However it is also possible to set the prefix through the command line or per table in the migration itself.

For example, imagine you are a gaming company where the game is broken in 128 partitions, named "prefix_1", "prefix_2", "prefix_3" up to "prefix_128". Now, whenever you need to migrate data, you need to migrate data on all different 128 prefixes. There are two ways of achieve that.

The first mechanism is to invoke mix ecto.migrate multiple times, once per prefix, passing the --prefix option:

$ mix ecto.migrate --prefix "prefix_1"
$ mix ecto.migrate --prefix "prefix_2"
$ mix ecto.migrate --prefix "prefix_3"
...
$ mix ecto.migrate --prefix "prefix_128"

The other approach is by changing each desired migration to run across multiple prefixes. For example:

defmodule MyApp.Repo.Migrations.CreateSample do
  use Ecto.Migration

  def change do
    for i <- 1..128 do
      prefix = "prefix_#{i}"
      create table(:samples, prefix: prefix) do
        add :name, :string
        timestamps()
      end

      # Execute the commands on the current prefix
      # before moving on to the next prefix
      flush()
    end
  end
end

Summing up

Ecto provides many conveniences for working with querying prefixes. Those conveniences allow developers to configure prefixes with different precedence, starting with the highest one:


	from/join prefixes


	query/struct prefixes


	schema prefixes


	connection prefixes




This way developers can tackle different scenarios, from production requirements to multi-tenant applications.

  
Polymorphic associations with many to many

Besides belongs_to, has_many, has_one and :through associations, Ecto also includes many_to_many. many_to_many relationships, as the name says, allows a record from table X to have many associated entries from table Y and vice-versa. Although many_to_many associations can be written as has_many :through, using many_to_many may considerably simplify some workflows.

In this guide, we will talk about polymorphic associations and how many_to_many can remove boilerplate from certain approaches compared to has_many :through.

Todo lists v65131

The internet has seen its share of todo list applications. But that won't stop us from creating our own!

In our case, there is one aspect of todo list applications we are interested in, which is the relationship where the todo list has many todo items. This exact scenario is explored in detail in a post about nested associations and embeds from Dashbit's blog. Let's recap the important points.

Our todo list app has two schemas, Todo.List and Todo.Item:

defmodule MyApp.TodoList do
  use Ecto.Schema

  schema "todo_lists" do
    field :title
    has_many :todo_items, MyApp.TodoItem
    timestamps()
  end
end

defmodule MyApp.TodoItem do
  use Ecto.Schema

  schema "todo_items" do
    field :description
    timestamps()
  end
end

One of the ways to introduce a todo list with multiple items into the database is to couple our UI representation to our schemas. That's the approach we took in the blog post with Phoenix. Roughly:

<%= form_for @todo_list_changeset,
             todo_list_path(@conn, :create),
             fn f -> %>
  <%= text_input f, :title %>
  <%= inputs_for f, :todo_items, fn i -> %>
    ...
  <% end %>
<% end %>

When such a form is submitted in Phoenix, it will send parameters with the following shape:

%{
  "todo_list" => %{
    "title" => "shipping list",
    "todo_items" => %{
      0 => %{"description" => "bread"},
      1 => %{"description" => "eggs"}
    }
  }
}

We could then retrieve those parameters and pass it to an Ecto changeset and Ecto would automatically figure out what to do:

# In MyApp.TodoList
def changeset(struct, params \\ %{}) do
  struct
  |> Ecto.Changeset.cast(params, [:title])
  |> Ecto.Changeset.cast_assoc(:todo_items, required: true)
end

# And then in MyApp.TodoItem
def changeset(struct, params \\ %{}) do
  struct
  |> Ecto.Changeset.cast(params, [:description])
end

By calling Ecto.Changeset.cast_assoc/3, Ecto will look for a "todo_items" key inside the parameters given on cast, and compare those parameters with the items stored in the todo list struct. Ecto will automatically generate instructions to insert, update or delete todo items such that:


	if a todo item sent as parameter has an ID and it matches an existing associated todo item, we consider that todo item should be updated


	if a todo item sent as parameter does not have an ID (nor a matching ID), we consider that todo item should be inserted


	if a todo item is currently associated but its ID was not sent as parameter, we consider the todo item is being replaced and we act according to the :on_replace callback. By default :on_replace will raise so you choose a behaviour between replacing, deleting, ignoring or nilifying the association




The advantage of using cast_assoc/3 is that Ecto is able to do all of the hard work of keeping the entries associated, as long as we pass the data exactly in the format that Ecto expects. However, such approach is not always preferrable and in many situations it is better to design our associations differently or decouple our UIs from our database representation.

Polymorphic todo items

To show an example of where using cast_assoc/3 is just too complicated to be worth it, let's imagine you want your "todo items" to be polymorphic. For example, you want to be able to add todo items not only to "todo lists" but to many other parts of your application, such as projects, milestones, you name it.

First of all, it is important to remember Ecto does not provide the same type of polymorphic associations available in frameworks such as Rails and Laravel. In such frameworks, a polymorphic association uses two columns, the parent_id and parent_type. For example, one todo item would have parent_id of 1 with parent_type of "TodoList" while another would have parent_id of 1 with parent_type of "Project".

The issue with the design above is that it breaks database references. The database is no longer capable of guaranteeing the item you associate to exists or will continue to exist in the future. This leads to an inconsistent database which end-up pushing workarounds to your application.

The design above is also extremely inefficient, especially if you're working with large tables. Bear in mind that if that's your case, you might be forced to remove such polymorphic references in the future when frequent polymorphic queries start grinding the database to a halt even after adding indexes and optimizing the database.

Luckily, the documentation for the Ecto.Schema.belongs_to/3 macro includes a section named "Polymorphic associations" with some examples on how to design sane and performant associations. One of those approaches consists in using many join tables. Besides the "todo_lists" and "projects" tables and the "todo_items" table, we would create "todo_list_items" and "project_items" to associate todo items to todo lists and todo items to projects respectively. In terms of migrations, we are looking at the following:

create table(:todo_lists)  do
  add :title
  timestamps()
end

create table(:projects)  do
  add :name
  timestamps()
end

create table(:todo_items)  do
  add :description
  timestamps()
end

create table(:todo_lists_items) do
  add :todo_item_id, references(:todo_items)
  add :todo_list_id, references(:todo_lists)
  timestamps()
end

create table(:projects_items) do
  add :todo_item_id, references(:todo_items)
  add :project_id, references(:projects)
  timestamps()
end

By adding one table per association pair, we keep database references and can efficiently perform queries that relies on indexes.

First let's see how implement this functionality in Ecto using a has_many :through and then use many_to_many to remove a lot of the boilerplate we were forced to introduce.

Polymorphism with has_many :through

Given we want our todo items to be polymorphic, we can no longer associate a todo list to todo items directly. Instead we will create an intermediate schema to tie MyApp.TodoList and MyApp.TodoItem together.

defmodule MyApp.TodoList do
  use Ecto.Schema

  schema "todo_lists" do
    field :title
    has_many :todo_list_items, MyApp.TodoListItem
    has_many :todo_items,
      through: [:todo_list_items, :todo_item]
    timestamps()
  end
end

defmodule MyApp.TodoListItem do
  use Ecto.Schema

  schema "todo_list_items" do
    belongs_to :todo_list, MyApp.TodoList
    belongs_to :todo_item, MyApp.TodoItem
    timestamps()
  end
end

defmodule MyApp.TodoItem do
  use Ecto.Schema

  schema "todo_items" do
    field :description
    timestamps()
  end
end

Although we introduced MyApp.TodoListItem as an intermediate schema, has_many :through allows us to access all todo items for any todo list transparently:

todo_lists |> Repo.preload(:todo_items)

The trouble is that :through associations are read-only since Ecto does not have enough information to fill in the intermediate schema. This means that, if we still want to use cast_assoc to insert a todo list with many todo items directly from the UI, we cannot use the :through association and instead must go step by step. We would need to first cast_assoc(:todo_list_items) from TodoList and then call cast_assoc(:todo_item) from the TodoListItem schema:

# In MyApp.TodoList
def changeset(struct, params \\ %{}) do
  struct
  |> Ecto.Changeset.cast(params, [:title])
  |> Ecto.Changeset.cast_assoc(
    :todo_list_items,
    required: true
  )
end

# And then in the MyApp.TodoListItem
def changeset(struct, params \\ %{}) do
  struct
  |> Ecto.Changeset.cast_assoc(:todo_item, required: true)
end

# And then in MyApp.TodoItem
def changeset(struct, params \\ %{}) do
  struct
  |> Ecto.Changeset.cast(params, [:description])
end

To further complicate things, remember cast_assoc expects a particular shape of data that reflects your associations. In this case, because of the intermediate schema, the data sent through your forms in Phoenix would have to look as follows:

%{"todo_list" => %{
  "title" => "shipping list",
  "todo_list_items" => %{
    0 => %{"todo_item" => %{"description" => "bread"}},
    1 => %{"todo_item" => %{"description" => "eggs"}},
  }
}}

To make matters worse, you would have to duplicate this logic for every intermediate schema, and introduce MyApp.TodoListItem for todo lists, MyApp.ProjectItem for projects, etc.

Luckily, many_to_many allows us to remove all of this boilerplate.

Polymorphism with many_to_many

In a way, the idea behind many_to_many associations is that it allows us to associate two schemas via an intermediate schema while automatically taking care of all details about the intermediate schema. Let's rewrite the schemas above to use many_to_many:

defmodule MyApp.TodoList do
  use Ecto.Schema

  schema "todo_lists" do
    field :title
    many_to_many :todo_items, MyApp.TodoItem,
      join_through: MyApp.TodoListItem
    timestamps()
  end
end

defmodule MyApp.TodoListItem do
  use Ecto.Schema

  schema "todo_list_items" do
    belongs_to :todo_list, MyApp.TodoList
    belongs_to :todo_item, MyApp.TodoItem
    timestamps()
  end
end

defmodule MyApp.TodoItem do
  use Ecto.Schema

  schema "todo_items" do
    field :description
    timestamps()
  end
end

Notice MyApp.TodoList no longer needs to define a has_many association pointing to the MyApp.TodoListItem schema and instead we can just associate to :todo_items using many_to_many.

Differently from has_many :through, many_to_many associations are also writeable. This means we can send data through our forms exactly as we did at the beginning of this guide:

%{"todo_list" => %{
  "title" => "shipping list",
  "todo_items" => %{
    0 => %{"description" => "bread"},
    1 => %{"description" => "eggs"},
  }
}}

And we no longer need to define a changeset function in the intermediate schema:

# In MyApp.TodoList
def changeset(struct, params \\ %{}) do
  struct
  |> Ecto.Changeset.cast(params, [:title])
  |> Ecto.Changeset.cast_assoc(:todo_items, required: true)
end

# And then in MyApp.TodoItem
def changeset(struct, params \\ %{}) do
  struct
  |> Ecto.Changeset.cast(params, [:description])
end

In other words, we can use exactly the same code we had in the "todo lists has_many todo items" case. So even when external constraints require us to use a join table, many_to_many associations can automatically manage them for us. Everything you know about associations will just work with many_to_many associations as well.

Finally, even though we have specified a schema as the :join_through option in many_to_many, many_to_many can also work without intermediate schemas altogether by simply giving it a table name:

defmodule MyApp.TodoList do
  use Ecto.Schema

  schema "todo_lists" do
    field :title
    many_to_many :todo_items, MyApp.TodoItem,
      join_through: "todo_list_items"
    timestamps()
  end
end

In this case, you can completely remove the MyApp.TodoListItem schema from your application and the code above will still work. The only difference is that when using tables, any autogenerated value that is filled by Ecto schema, such as timestamps, won't be filled as we no longer have a schema. To solve this, you can either drop those fields from your migrations or set a default at the database level.

Summary

In this guide we used many_to_many associations to drastically improve a polymorphic association design that relied on has_many :through. Our goal was to allow "todo_items" to associate to different entities in our code base, such as "todo_lists" and "projects". We have done this by creating intermediate tables and by using many_to_many associations to automatically manage those join tables.

At the end, our schemas may look like:

defmodule MyApp.TodoList do
  use Ecto.Schema

  schema "todo_lists" do
    field :title
    many_to_many :todo_items, MyApp.TodoItem,
      join_through: "todo_list_items"
    timestamps()
  end

  def changeset(struct, params \\ %{}) do
    struct
    |> Ecto.Changeset.cast(params, [:title])
    |> Ecto.Changeset.cast_assoc(
      :todo_items,
      required: true
    )
  end
end

defmodule MyApp.Project do
  use Ecto.Schema

  schema "projects" do
    field :name
    many_to_many :todo_items, MyApp.TodoItem,
      join_through: "project_items"
    timestamps()
  end

  def changeset(struct, params \\ %{}) do
    struct
    |> Ecto.Changeset.cast(params, [:name])
    |> Ecto.Changeset.cast_assoc(
      :todo_items,
      required: true
    )
  end
end

defmodule MyApp.TodoItem do
  use Ecto.Schema

  schema "todo_items" do
    field :description
    timestamps()
  end

  def changeset(struct, params \\ %{}) do
    struct
    |> Ecto.Changeset.cast(params, [:description])
  end
end

And the database migration:

create table("todo_lists")  do
  add :title
  timestamps()
end

create table("projects")  do
  add :name
  timestamps()
end

create table("todo_items")  do
  add :description
  timestamps()
end

# Primary key and timestamps are not required if
# using many_to_many without schemas
create table("todo_lists_items", primary_key: false) do
  add :todo_item_id, references(:todo_items)
  add :todo_list_id, references(:todo_lists)
  # timestamps()
end

# Primary key and timestamps are not required if
# using many_to_many without schemas
create table("projects_items", primary_key: false) do
  add :todo_item_id, references(:todo_items)
  add :project_id, references(:projects)
  # timestamps()
end

Overall our code looks structurally the same as has_many would, although at the database level our relationships are expressed with join tables.

While in this guide we changed our code to cope with the parameter format required by cast_assoc, in Constraints and Upserts we drop cast_assoc altogether and use put_assoc which brings more flexibilities when working with associations.

  
Replicas and dynamic repositories

When applications reach a certain scale, a single database may not be enough to sustain the required throughput. In such scenarios, it is very common to introduce read replicas: all write operations are sent to primary database and most of the read operations are performed against the replicas. The credentials of the primary and replica databases are typically known upfront by the time the code is compiled.

In other cases, you may need a single Ecto repository to interact with different database instances which are not known upfront. For instance, you may need to communicate with hundreds of database very sporadically, so instead of opening up a connection to each of those hundreds of database when your application starts, you want to quickly start connection, perform some queries, and then shut down, while still leveraging Ecto's APIs as a whole.

This guide will cover how to tackle both approaches.

Primary and Replicas

Since the credentials of the primary and replicas databases are known upfront, adding support for primary and replica databases in your Ecto application is relatively straightforward. Imagine you have a MyApp.Repo and you want to add four read replicas. This could be done in three steps.

First, define the primary and replicas repositories in lib/my_app/repo.ex:

defmodule MyApp.Repo do
  use Ecto.Repo,
    otp_app: :my_app,
    adapter: Ecto.Adapters.Postgres

  @replicas [
    MyApp.Repo.Replica1,
    MyApp.Repo.Replica2,
    MyApp.Repo.Replica3,
    MyApp.Repo.Replica4
  ]

  def replica do
    Enum.random(@replicas)
  end

  for repo <- @replicas do
    defmodule repo do
      use Ecto.Repo,
        otp_app: :my_app,
        adapter: Ecto.Adapters.Postgres,
        read_only: true
    end
  end
end

The code above defines a regular MyApp.Repo and four replicas, called MyApp.Repo.Replica1 up to MyApp.Repo.Replica4. We pass the :read_only option to the replica repositories, so operations such as insert, update and friends are not made accessible. We also define a function called replica with the purpose of returning a random replica.

Next we need to make sure both primary and replicas are configured properly in your config/config.exs files. In development and test, you can likely use the same database credentials for all repositories, all pointing to the same database address:

replicas = [
  MyApp.Repo,
  MyApp.Repo.Replica1,
  MyApp.Repo.Replica2,
  MyApp.Repo.Replica3,
  MyApp.Repo.Replica4
]

for repo <- replicas do
  config :my_app, repo,
    username: "postgres",
    password: "postgres",
    database: "my_app_prod",
    hostname: "localhost",
    pool_size: 10
end

In production, you want each database to connect to a different hostname:

repos = %{
  MyApp.Repo => "prod-primary",
  MyApp.Repo.Replica1 => "prod-replica-1",
  MyApp.Repo.Replica2 => "prod-replica-2",
  MyApp.Repo.Replica3 => "prod-replica-3",
  MyApp.Repo.Replica4 => "prod-replica-4"
}

for {repo, hostname} <- repos do
  config :my_app, repo,
    username: "postgres",
    password: "postgres",
    database: "my_app_prod",
    hostname: hostname,
    pool_size: 10
end

Finally, make sure to start all repositories in your supervision tree:

children = [
  MyApp.Repo,
  MyApp.Repo.Replica1,
  MyApp.Repo.Replica2,
  MyApp.Repo.Replica3,
  MyApp.Repo.Replica4
]

Now that all repositories are configured, we can safely use them in your application code. Every time you are performing a read operation, you can the replica/0 function that we have added to return a random replica we will send the query to:

MyApp.Repo.replica().all(query)

And now you are ready to work with primary and replicas, no hacks or complex dependencies required!

Testing replicas

While all of the work we have done so far should fully work in development and production, it may not be enough for tests. Most developers testing Ecto applications are using a sandbox, such as the Ecto SQL Sandbox.

When using a sandbox, each of your tests run in an isolated and independent transaction. Once the test is done, the transaction is rolled back. Which means we can trivially revert all of the changes done in a test in a very performant way.

Unfortunately, even if you configure your primary and replicas to have the same credentials and point to the same hostname, each Ecto repository will open up their own pool of database connections. This means that, once you move to a primary + replicas setup, a simple test like this one won't pass:

user = Repo.insert!(%User{name: "jane doe"})
assert Repo.replica().get!(User, user.id)

That's because Repo.insert! will write to one database connection and the repository returned by Repo.replica() will perform the read in another connection. Since the write is done in transaction, its contents won't be available to other connections until the transaction commits, which will never happen for test connections.

There are two options to tackle this problem: one is to change replicas and the other is to use dynamic repos.

A custom replica definition

One simple solution to the problem above is to use a custom replica implementation during tests that always return the primary repository, like this:

if Mix.env() == :test do
  def replica, do: __MODULE__
else
  def replica, do: Enum.random(@replicas)
end

Now during tests, the replica will always return the repository primary repository itself. While this approach works fine, it has the downside that, if you accidentally invoke a write function in in a replica, the test will pass, since the replica function is returning the primary repo, while the code will fail in production.

Using :default_dynamic_repo

Another approach to testing is to set the :default_dynamic_repo option when defining the repository. Let's see what we mean by that.

When you list a repository in your supervision tree, such as MyApp.Repo, behind the scenes it will start a supervision tree with a process named MyApp.Repo. By default, the process has the same name as the repository module itself. Now every time you invoke a function in MyApp.Repo, such as MyApp.Repo.insert/2, Ecto will use the connection pool from the process named MyApp.Repo.

From v3.0, Ecto has the ability to start multiple processes from the same repository. The only requirement is that they must have different process names, like this:

children = [
  MyApp.Repo,
  {MyApp.Repo, name: :another_instance_of_repo}
]

While the particular example doesn't make much sense (we will cover an actual use case for this feature next), the idea is that now you have two repositories running: one is named MyApp.Repo and the other one is named :another_instance_of_repo. Each of those processes have their own connection pool. You can tell Ecto which process you want to use in your repo operations by calling:

MyApp.Repo.put_dynamic_repo(MyApp.Repo)
MyApp.Repo.put_dynamic_repo(:another_instance_of_repo)

Once you call MyApp.Repo.put_dynamic_repo(name), all invocations made on MyApp.Repo will use the connection pool denoted by name.

How can this help with our replica tests? If we look back to the supervision tree we defined earlier in this guide, you will find this:

children = [
  MyApp.Repo,
  MyApp.Repo.Replica1,
  MyApp.Repo.Replica2,
  MyApp.Repo.Replica3,
  MyApp.Repo.Replica4
]

We are starting five different repositories and five different connection pools. Since we want the replica repositories to use the MyApp.Repo, we can achieve this by doing the following on the setup of each test:

@replicas [
  MyApp.Repo.Replica1,
  MyApp.Repo.Replica2,
  MyApp.Repo.Replica3,
  MyApp.Repo.Replica4
]

setup do
  for replica <- @replicas do
    replica.put_dynamic_repo(MyApp.Repo)
  end

  :ok
end

There is even a better way! We can pass a :default_dynamic_repo option when we define the repository. In this case, we want to set the :default_dynamic_repo to MyApp.Repo only during the test environment. In your lib/my_app/repo.ex, do this:

  for repo <- @replicas do
    default_dynamic_repo =
      if Mix.env() == :test do
        MyApp.Repo
      else
        repo
      end

    defmodule repo do
      use Ecto.Repo,
        otp_app: :my_app,
        adapter: Ecto.Adapters.Postgres,
        read_only: true,
        default_dynamic_repo: default_dynamic_repo
    end
  end

And now your tests should work as before, while still being able to detect if you accidentally perform a write operation in a replica.

Dynamic repositories

At this point, we have learned that Ecto allows you to start multiple connections based on the same repository. This is typically useful when you have to connect multiple databases or perform short-lived database connections.

For example, you can start a repository with a given set of credentials dynamically, like this:

MyApp.Repo.start_link(
  name: :some_client,
  hostname: "client.example.com",
  username: "...",
  password: "...",
  pool_size: 1
)

In other words, start_link accepts the same options as the database configuration. Now let's do a query on the dynamically started repository. If you attempt to simply perform MyApp.Repo.all(Post), it may fail, as by default it will try to use a process named MyApp.Repo, which may or may not be running. So don't forget to call put_dynamic_repo/1 before:

MyApp.Repo.put_dynamic_repo(:some_client)
MyApp.Repo.all(Post)

Ecto also allows you to start a repository with no name (just like that famous horse). In such cases, you need to explicitly pass name: nil and match on the result of MyApp.Repo.start_link/1 to retrieve the PID, which should be given to put_dynamic_repo. Let's also use this opportunity and perform proper database clean-up, by shutting up the new repository and reverting the value of put_dynamic_repo:

default_dynamic_repo = MyApp.Repo.get_dynamic_repo()

{:ok, repo} =
  MyApp.Repo.start_link(
    name: nil,
    hostname: "client.example.com",
    username: "...",
    password: "...",
    pool_size: 1
  )

try do
  MyApp.Repo.put_dynamic_repo(repo)
  MyApp.Repo.all(Post)
after
  MyApp.Repo.put_dynamic_repo(default_dynamic_repo)
  Supervisor.stop(repo)
end

We can encapsulate all of this in a function too, which you could define in your repository:

defmodule MyApp.Repo do
  use Ecto.Repo, ...

  def with_dynamic_repo(credentials, callback) do
    default_dynamic_repo = get_dynamic_repo()
    start_opts = [name: nil, pool_size: 1] ++ credentials
    {:ok, repo} = MyApp.Repo.start_link(start_opts)

    try do
      MyApp.Repo.put_dynamic_repo(repo)
      callback.()
    after
      MyApp.Repo.put_dynamic_repo(default_dynamic_repo)
      Supervisor.stop(repo)
    end
  end
end

And now use it as:

credentials = [
  hostname: "client.example.com",
  username: "...",
  password: "..."
]

MyApp.Repo.with_dynamic_repo(credentials, fn ->
  MyApp.Repo.all(Post)
end)

And that's it! Now you can have dynamic connections, all properly encapsulated in a single function and built on top of the dynamic repo API.

  
Schemaless queries

Most queries in Ecto are written using schemas. For example, to retrieve all posts in a database, one may write:

MyApp.Repo.all(Post)

In the construct above, Ecto knows all fields and their types in the schema, rewriting the query above to:

query =
  from p in Post,
    select: %Post{title: p.title, body: p.body, ...}

MyApp.Repo.all(query)

Although you might use schemas for most of your queries, Ecto also adds the ability to write regular schemaless queries when prefered.

One example is this ability to select all desired fields without duplication:

from "posts", select: [:title, :body]

When a list of fields is given, Ecto will automatically convert the list of fields to a map or a struct.

Support for passing a list of fields or keyword lists is available to almost all query constructs. For example, we can use an update query to change the title of a given post without a schema:

def update_title(post, new_title) do
  query =
    from "posts",
      where: [id: ^post.id],
      update: [set: [title: ^new_title]]

  MyApp.Repo.update_all(query)
end

The Ecto.Query.update/3 construct supports four commands:


	:set - sets the given column to the given values


	:inc - increments the given column by the given value


	:push - pushes (appends) the given value to the end of an array column


	:pull - pulls (removes) the given value from an array column




For example, we can increment a column atomically by using the :inc command, with or without schemas:

def increment_page_views(post) do
  query =
    from "posts",
      where: [id: ^post.id],
      update: [inc: [page_views: 1]]

  MyApp.Repo.update_all(query)
end

Let's take a look at another example. Imagine you are writing a reporting view, it may be counter-productive to think how your existing application schemas relate to the report being generated. It is often simpler to write a query that returns only the data you need, without trying to fit the data into existing schemas:

import Ecto.Query

def running_activities(start_at, end_at)
  query =
    from u in "users",
      join: a in "activities",
      on: a.user_id == u.id,
      where:
        a.start_at > type(^start_at, :naive_datetime) and
          a.end_at < type(^end_at, :naive_datetime),
      group_by: a.user_id,
      select: %{
        user_id: a.user_id,
        interval: a.end_at - a.start_at,
        count: count(u.id)
      }

  MyApp.Repo.all(query)
end

The function above does not rely on schemas. It returns only the data that matters for building the report. Notice how we use the type/2 function to specify what is the expected type of the argument we are interpolating, benefiting from the same type casting guarantees a schema would give.

By allowing regular data structures to be given to most query operations, Ecto makes queries with and without schemas more accessible. Not only that, it also enables developers to write dynamic queries, where fields, filters, ordering cannot be specified upfront.

insert_all, update_all and delete_all

Ecto allows all database operations to be expressed without a schema. One of the functions provided is Ecto.Repo.insert_all/3. With insert_all, developers can insert multiple entries at once into a repository:

MyApp.Repo.insert_all(
  Post,
  [
    [title: "hello", body: "world"],
    [title: "another", body: "post"]
  ]
)

Updates and deletes can also be done without schemas via Ecto.Repo.update_all/3 and Ecto.Repo.delete_all/2 respectively:

# Use the ID to trigger updates
post = from p in "posts", where: [id: ^id]

# Update the title for all matchihg posts
{1, _} =
  MyApp.Repo.update_all post, set: [title: "new title"]

# Delete all matching posts
{1, _} =
  MyApp.Repo.delete_all post

It is not hard to see how these operations directly map to their SQL variants, keeping the database at your fingertips without the need to intermediate all operations through schemas.

  
Test factories

Many projects depend on external libraries to build their test data. Some of those libraries are called factories because they provide convenience functions for producing different groups of data. However, given Ecto is able to manage complex data trees, we can implement such functionality without relying on third-party projects.

To get started, let's create a file at "test/support/factory.ex" with the following contents:

defmodule MyApp.Factory do
  alias MyApp.Repo

  # Factories

  def build(:post) do
    %MyApp.Post{title: "hello world"}
  end

  def build(:comment) do
    %MyApp.Comment{body: "good post"}
  end

  def build(:post_with_comments) do
    %MyApp.Post{
      title: "hello with comments",
      comments: [
        build(:comment, body: "first"),
        build(:comment, body: "second")
      ]
    }
  end

  def build(:user) do
    %MyApp.User{
      email: "hello#{System.unique_integer()}",
      username: "hello#{System.unique_integer()}"
    }
  end

  # Convenience API

  def build(factory_name, attributes) do
    factory_name |> build() |> struct(attributes)
  end

  def insert!(factory_name, attributes \\ []) do
    Repo.insert! build(factory_name, attributes)
  end
end

Our factory module defines four "factories" as different clauses to the build function: :post, :comment, :post_with_comments and :user. Each clause defines structs with the fields that are required by the database. In certain cases, the generated struct also needs to generate unique fields, such as the user's email and username. We did so by calling Elixir's System.unique_integer() - you could call System.unique_integer([:positive]) if you need a strictly positive number.

At the end, we defined two functions, build/2 and insert!/2, which are conveniences for building structs with specific attributes and for inserting data directly in the repository respectively.

That's literally all that is necessary for building our factories. We are now ready to use them in our tests. First, open up your "mix.exs" and make sure the "test/support/factory.ex" file is compiled:

def project do
  [...,
   elixirc_paths: elixirc_paths(Mix.env),
   ...]
end

defp elixirc_paths(:test), do: ["lib", "test/support"]
defp elixirc_paths(_), do: ["lib"]

Now in any of the tests that need to generate data, we can import the MyApp.Factory module and use its functions:

import MyApp.Factory

build(:post)
#=> %MyApp.Post{id: nil, title: "hello world", ...}

build(:post, title: "custom title")
#=> %MyApp.Post{id: nil, title: "custom title", ...}

insert!(:post, title: "custom title")
#=> %MyApp.Post{id: ..., title: "custom title"}

By building the functionality we need on top of Ecto capabilities, we are able to extend and improve our factories on whatever way we desire, without being constrained to third-party limitations.

  
    
Ecto    



      
Ecto is split into 4 main components:


	Ecto.Repo - repositories are wrappers around the data store.
Via the repository, we can create, update, destroy and query existing entries.
A repository needs an adapter and credentials to communicate to the database



	Ecto.Schema - schemas are used to map any data source into an Elixir
struct. We will often use them to map tables into Elixir data but that's
one of their use cases and not a requirement for using Ecto



	Ecto.Changeset - changesets provide a way for developers to filter
and cast external parameters, as well as a mechanism to track and
validate changes before they are applied to your data



	Ecto.Query - written in Elixir syntax, queries are used to retrieve
information from a given repository. Queries in Ecto are secure, avoiding
common problems like SQL Injection, while still being composable, allowing
developers to build queries piece by piece instead of all at once





Besides the four components above, most developers use Ecto to interact
with SQL databases, such as Postgres and MySQL via the
ecto_sql project. ecto_sql provides many
conveniences for working with SQL databases as well as the ability to version
how your database changes through time via
database migrations.

If you want to quickly check a sample application using Ecto, please check
the getting started guide and
the accompanying sample application. Ecto's README
also links to other resources.

In the following sections, we will provide an overview of those components and
how they interact with each other. Feel free to access their respective module
documentation for more specific examples, options and configuration.

Repositories

Ecto.Repo is a wrapper around the database. We can define a
repository as follows:

defmodule Repo do
  use Ecto.Repo,
    otp_app: :my_app,
    adapter: Ecto.Adapters.Postgres
end

Where the configuration for the Repo must be in your application
environment, usually defined in your config/config.exs:

config :my_app, Repo,
  database: "ecto_simple",
  username: "postgres",
  password: "postgres",
  hostname: "localhost",
  # OR use a URL to connect instead
  url: "postgres://postgres:postgres@localhost/ecto_simple"

Each repository in Ecto defines a start_link/0 function that needs to be invoked
before using the repository. In general, this function is not called directly,
but used as part of your application supervision tree.

If your application was generated with a supervisor (by passing --sup to mix new)
you will have a lib/my_app/application.ex file containing the application start
callback that defines and starts your supervisor.  You just need to edit the start/2
function to start the repo as a supervisor on your application's supervisor:

def start(_type, _args) do
  children = [
    {MyApp.Repo, []}
  ]

  opts = [strategy: :one_for_one, name: MyApp.Supervisor]
  Supervisor.start_link(children, opts)
end

Schema

Schemas allow developers to define the shape of their data.
Let's see an example:

defmodule Weather do
  use Ecto.Schema

  # weather is the DB table
  schema "weather" do
    field :city,    :string
    field :temp_lo, :integer
    field :temp_hi, :integer
    field :prcp,    :float, default: 0.0
  end
end

By defining a schema, Ecto automatically defines a struct with
the schema fields:

iex> weather = %Weather{temp_lo: 30}
iex> weather.temp_lo
30

The schema also allows us to interact with a repository:

iex> weather = %Weather{temp_lo: 0, temp_hi: 23}
iex> Repo.insert!(weather)
%Weather{...}

After persisting weather to the database, it will return a new copy of
%Weather{} with the primary key (the id) set. We can use this value
to read a struct back from the repository:

# Get the struct back
iex> weather = Repo.get Weather, 1
%Weather{id: 1, ...}

# Delete it
iex> Repo.delete!(weather)
%Weather{...}

NOTE: by using Ecto.Schema, an :id field with type :id (:id means :integer) is
generated by default, which is the primary key of the Schema. If you want
to use a different primary key, you can declare custom @primary_key
before the schema/2 call. Consult the Ecto.Schema documentation
for more information.



Notice how the storage (repository) and the data are decoupled. This provides
two main benefits:


	By having structs as data, we guarantee they are light-weight,
serializable structures. In many languages, the data is often represented
by large, complex objects, with entwined state transactions, which makes
serialization, maintenance and understanding hard;



	You do not need to define schemas in order to interact with repositories,
operations like all, insert_all and so on allow developers to directly
access and modify the data, keeping the database at your fingertips when
necessary;





Changesets

Although in the example above we have directly inserted and deleted the
struct in the repository, operations on top of schemas are done through
changesets so Ecto can efficiently track changes.

Changesets allow developers to filter, cast, and validate changes before
we apply them to the data. Imagine the given schema:

defmodule User do
  use Ecto.Schema

  import Ecto.Changeset

  schema "users" do
    field :name
    field :email
    field :age, :integer
  end

  def changeset(user, params \\ %{}) do
    user
    |> cast(params, [:name, :email, :age])
    |> validate_required([:name, :email])
    |> validate_format(:email, ~r/@/)
    |> validate_inclusion(:age, 18..100)
  end
end

The changeset/2 function first invokes Ecto.Changeset.cast/4 with
the struct, the parameters and a list of allowed fields; this returns a changeset.
The parameters is a map with binary keys and values that will be cast based
on the type defined on the schema.

Any parameter that was not explicitly listed in the fields list will be ignored.

After casting, the changeset is given to many Ecto.Changeset.validate_*
functions that validate only the changed fields. In other words:
if a field was not given as a parameter, it won't be validated at all.
For example, if the params map contain only the "name" and "email" keys,
the "age" validation won't run.

Once a changeset is built, it can be given to functions like insert and
update in the repository that will return an :ok or :error tuple:

case Repo.update(changeset) do
  {:ok, user} ->
    # user updated
  {:error, changeset} ->
    # an error occurred
end

The benefit of having explicit changesets is that we can easily provide
different changesets for different use cases. For example, one
could easily provide specific changesets for registering and updating
users:

def registration_changeset(user, params) do
  # Changeset on create
end

def update_changeset(user, params) do
  # Changeset on update
end

Changesets are also capable of transforming database constraints,
like unique indexes and foreign key checks, into errors. Allowing
developers to keep their database consistent while still providing
proper feedback to end users. Check Ecto.Changeset.unique_constraint/3
for some examples as well as the other _constraint functions.

Query

Last but not least, Ecto allows you to write queries in Elixir and send
them to the repository, which translates them to the underlying database.
Let's see an example:

import Ecto.Query, only: [from: 2]

query = from u in User,
          where: u.age > 18 or is_nil(u.email),
          select: u

# Returns %User{} structs matching the query
Repo.all(query)

In the example above we relied on our schema but queries can also be
made directly against a table by giving the table name as a string. In
such cases, the data to be fetched must be explicitly outlined:

query = from u in "users",
          where: u.age > 18 or is_nil(u.email),
          select: %{name: u.name, age: u.age}

# Returns maps as defined in select
Repo.all(query)

Queries are defined and extended with the from macro. The supported
keywords are:


	:distinct


	:where


	:order_by


	:offset


	:limit


	:lock


	:group_by


	:having


	:join


	:select


	:preload




Examples and detailed documentation for each of those are available
in the Ecto.Query module. Functions supported in queries are listed
in Ecto.Query.API.

When writing a query, you are inside Ecto's query syntax. In order to
access params values or invoke Elixir functions, you need to use the ^
operator, which is overloaded by Ecto:

def min_age(min) do
  from u in User, where: u.age > ^min
end

Besides Repo.all/1 which returns all entries, repositories also
provide Repo.one/1 which returns one entry or nil, Repo.one!/1
which returns one entry or raises, Repo.get/2 which fetches
entries for a particular ID and more.

Finally, if you need an escape hatch, Ecto provides fragments
(see Ecto.Query.API.fragment/1) to inject SQL (and non-SQL)
fragments into queries. Also, most adapters provide direct
APIs for queries, like Ecto.Adapters.SQL.query/4, allowing
developers to completely bypass Ecto queries.

Other topics

Associations

Ecto supports defining associations on schemas:

defmodule Post do
  use Ecto.Schema

  schema "posts" do
    has_many :comments, Comment
  end
end

defmodule Comment do
  use Ecto.Schema

  schema "comments" do
    field :title, :string
    belongs_to :post, Post
  end
end

When an association is defined, Ecto also defines a field in the schema
with the association name. By default, associations are not loaded into
this field:

iex> post = Repo.get(Post, 42)
iex> post.comments
#Ecto.Association.NotLoaded<...>

However, developers can use the preload functionality in queries to
automatically pre-populate the field:

Repo.all from p in Post, preload: [:comments]

Preloading can also be done with a pre-defined join value:

Repo.all from p in Post,
          join: c in assoc(p, :comments),
          where: c.votes > p.votes,
          preload: [comments: c]

Finally, for the simple cases, preloading can also be done after
a collection was fetched:

posts = Repo.all(Post) |> Repo.preload(:comments)

The Ecto module also provides conveniences for working
with associations. For example, Ecto.assoc/2 returns a query
with all associated data to a given struct:

import Ecto

# Get all comments for the given post
Repo.all assoc(post, :comments)

# Or build a query on top of the associated comments
query = from c in assoc(post, :comments), where: not is_nil(c.title)
Repo.all(query)

Another function in Ecto is build_assoc/3, which allows
someone to build an associated struct with the proper fields:

Repo.transaction fn ->
  post = Repo.insert!(%Post{title: "Hello", body: "world"})

  # Build a comment from post
  comment = Ecto.build_assoc(post, :comments, body: "Excellent!")

  Repo.insert!(comment)
end

In the example above, Ecto.build_assoc/3 is equivalent to:

%Comment{post_id: post.id, body: "Excellent!"}

You can find more information about defining associations and each
respective association module in Ecto.Schema docs.

NOTE: Ecto does not lazy load associations. While lazily loading
associations may sound convenient at first, in the long run it
becomes a source of confusion and performance issues.



Embeds

Ecto also supports embeds. While associations keep parent and child
entries in different tables, embeds stores the child along side the
parent.

Databases like MongoDB have native support for embeds. Databases
like PostgreSQL uses a mixture of JSONB (embeds_one/3) and ARRAY
columns to provide this functionality.

Check Ecto.Schema.embeds_one/3 and Ecto.Schema.embeds_many/3
for more information.

Mix tasks and generators

Ecto provides many tasks to help your workflow as well as code generators.
You can find all available tasks by typing mix help inside a project
with Ecto listed as a dependency.

Ecto generators will automatically open the generated files if you have
ECTO_EDITOR set in your environment variable.

Repo resolution

Ecto requires developers to specify the key :ecto_repos in their
application configuration before using tasks like ecto.create and
ecto.migrate. For example:

config :my_app, :ecto_repos, [MyApp.Repo]

config :my_app, MyApp.Repo,
  database: "ecto_simple",
  username: "postgres",
  password: "postgres",
  hostname: "localhost"

In the example below, we get all comments associated to the given
post:

post = Repo.get Post, 1
Repo.all Ecto.assoc(post, :comments)

assoc/2 can also receive a list of posts, as long as the posts are
not empty:

posts = Repo.all from p in Post, where: is_nil(p.published_at)
Repo.all Ecto.assoc(posts, :comments)

This function can also be used to dynamically load through associations
by giving it a list. For example, to get all authors for all comments for
the given posts, do:

posts = Repo.all from p in Post, where: is_nil(p.published_at)
Repo.all Ecto.assoc(posts, [:comments, :author])

  


  
    
      
      Link to this function

       View Source
     
  

  
Checks if an association is loaded.


  
  Examples



iex> post = Repo.get(Post, 1)
iex> Ecto.assoc_loaded?(post.comments)
false
iex> post = post |> Repo.preload(:comments)
iex> Ecto.assoc_loaded?(post.comments)
true

Builds a struct from the given assoc in struct.

If the relationship is a has_one or has_many and
the key is set in the given struct, the key will automatically
be set in the built association:

iex> post = Repo.get(Post, 13)
%Post{id: 13}
iex> build_assoc(post, :comments)
%Comment{id: nil, post_id: 13}

Note though it doesn't happen with belongs_to cases, as the
key is often the primary key and such is usually generated
dynamically:

iex> comment = Repo.get(Comment, 13)
%Comment{id: 13, post_id: 25}
iex> build_assoc(comment, :post)
%Post{id: nil}

You can also pass the attributes, which can be a map or
a keyword list, to set the struct's fields except the
association key.

iex> build_assoc(post, :comments, text: "cool")
%Comment{id: nil, post_id: 13, text: "cool"}

iex> build_assoc(post, :comments, %{text: "cool"})
%Comment{id: nil, post_id: 13, text: "cool"}

iex> build_assoc(post, :comments, post_id: 1)
%Comment{id: nil, post_id: 13}

It is possible to set:


	:source - changes the struct query source


	:prefix - changes the struct query prefix


	:context - changes the struct meta context


	:state - changes the struct state




Please refer to the Ecto.Schema.Metadata module for more information.

Changesets allow filtering, casting, validation and
definition of constraints when manipulating structs.

There is an example of working with changesets in the introductory
documentation in the Ecto module. The functions cast/4 and
change/2 are the usual entry points for creating changesets.
The first one is used to cast and validate external parameters,
such as parameters sent through a form, API, command line, etc.
The second one is used to change data directly from your application.

The remaining functions in this module, such as validations,
constraints, association handling, are about manipulating
changesets. Let's discuss some of this extra functionality.

External vs internal data

Changesets allow working with both kinds of data:


	internal to the application - for example programmatically generated,
or coming from other subsystems. This use case is primarily covered
by the change/2 and put_change/3 functions.



	external to the application - for example data provided by the user in
a form that needs to be type-converted and properly validated. This
use case is primarily covered by the cast/4 function.





Validations and constraints

Ecto changesets provide both validations and constraints which
are ultimately turned into errors in case something goes wrong.

The difference between them is that most validations can be
executed without a need to interact with the database and, therefore,
are always executed before attempting to insert or update the entry
in the database. Some validations may happen against the database but
they are inherently unsafe. Those validations start with a unsafe_
prefix, such as unsafe_validate_unique/3.

On the other hand, constraints rely on the database and are always safe.
As a consequence, validations are always checked before constraints.
Constraints won't even be checked in case validations failed.

Let's see an example:

defmodule User do
  use Ecto.Schema
  import Ecto.Changeset

  schema "users" do
    field :name
    field :email
    field :age, :integer
  end

  def changeset(user, params \\ %{}) do
    user
    |> cast(params, [:name, :email, :age])
    |> validate_required([:name, :email])
    |> validate_format(:email, ~r/@/)
    |> validate_inclusion(:age, 18..100)
    |> unique_constraint(:email)
  end
end

In the changeset/2 function above, we define three validations.
They check that name and email fields are present in the
changeset, the e-mail is of the specified format, and the age is
between 18 and 100 - as well as a unique constraint in the email
field.

Let's suppose the e-mail is given but the age is invalid. The
changeset would have the following errors:

changeset = User.changeset(%User{}, %{age: 0, email: "mary@example.com"})
{:error, changeset} = Repo.insert(changeset)
changeset.errors #=> [age: {"is invalid", []}, name: {"can't be blank", []}]

In this case, we haven't checked the unique constraint in the
e-mail field because the data did not validate. Let's fix the
age and the name, and assume that the e-mail already exists in the
database:

changeset = User.changeset(%User{}, %{age: 42, name: "Mary", email: "mary@example.com"})
{:error, changeset} = Repo.insert(changeset)
changeset.errors #=> [email: {"has already been taken", []}]

Validations and constraints define an explicit boundary when the check
happens. By moving constraints to the database, we also provide a safe,
correct and data-race free means of checking the user input.

Deferred constraints

Some databases support deferred constraints, i.e., constraints which are
checked at the end of the transaction rather than at the end of each statement.

Changesets do not support this type of constraints. When working with deferred
constraints, a violation while invoking Repo.insert/2 or Repo.update/2 won't
return {:error, changeset}, but rather raise an error at the end of the
transaction.

Empty values

Many times, the data given on cast needs to be further pruned, specially
regarding empty values. For example, if you are gathering data to be
cast from the command line or through an HTML form or any other text-based
format, it is likely those means cannot express nil values. For
those reasons, changesets include the concept of empty values, which are
values that will be automatically converted to the field's default value
on cast/4. Those values are stored in the changeset empty_values field
and default to [""].

Associations, embeds and on replace

Using changesets you can work with associations as well as with embedded
structs. There are two primary APIs:


	cast_assoc/3 and cast_embed/3 - those functions are used when
working with external data. In particular, they allow you to change
associations and embeds alongside the parent struct, all at once.



	put_assoc/4 and put_embed/4 - it allows you to replace the
association or embed as a whole. This can be used to move associated
data from one entry to another, to completely remove or replace
existing entries.





See the documentation for those functions for more information.

The :on_replace option

When using any of those APIs, you may run into situations where Ecto sees
data is being replaced. For example, imagine a Post has many Comments where
the comments have IDs 1, 2 and 3. If you call cast_assoc/3 passing only
the IDs 1 and 2, Ecto will consider 3 is being "replaced" and it will raise
by default. Such behaviour can be changed when defining the relation by
setting :on_replace option when defining your association/embed according
to the values below:


	:raise (default) - do not allow removing association or embedded
data via parent changesets


	:mark_as_invalid - if attempting to remove the association or
embedded data via parent changeset - an error will be added to the parent
changeset, and it will be marked as invalid


	:nilify - sets owner reference column to nil (available only for
associations). Use this on a belongs_to column to allow the association
to be cleared out so that it can be set to a new value. Will set action
on associated changesets to :replace


	:update - updates the association, available only for has_one and belongs_to.
This option will update all the fields given to the changeset including the id
for the association


	:delete - removes the association or related data from the database.
This option has to be used carefully (see below). Will set action on associated
changesets to :replace




The :delete option in particular must be used carefully as it would allow
users to delete any associated data by simply not sending any data for a given
field. If you need deletion, it is often preferred to add a separate boolean
virtual field to the changeset function that will allow you to manually mark
it for deletion, as in the example below:

defmodule Comment do
  use Ecto.Schema
  import Ecto.Changeset

  schema "comments" do
    field :body, :string
    field :delete, :boolean, virtual: true
  end

  def changeset(comment, params) do
    cast(comment, params, [:body, :delete])
    |> maybe_mark_for_deletion
  end

  defp maybe_mark_for_deletion(changeset) do
    if get_change(changeset, :delete) do
      %{changeset | action: :delete}
    else
      changeset
    end
  end
end

Schemaless changesets

In the changeset examples so far, we have always used changesets to validate
and cast data contained in a struct defined by an Ecto schema, such as the %User{}
struct defined by the User module.

However, changesets can also be used with "regular" structs too by passing a tuple
with the data and its types:

user = %User{}
types = %{first_name: :string, last_name: :string, email: :string}
changeset =
  {user, types}
  |> Ecto.Changeset.cast(params["name"], Map.keys(types))
  |> Ecto.Changeset.validate_required(...)
  |> Ecto.Changeset.validate_length(...)

where the user struct refers to the definition in the following module:

defmodule User do
  defstruct [:name, :age]
end

Changesets can also be used with data in a plain map, by following the same API:

data  = %{}
types = %{name: :string}
params = %{name: "Callum"}
changeset =
  {data, types}
  |> Ecto.Changeset.cast(params, Map.keys(types))
  |> Ecto.Changeset.validate_required(...)
  |> Ecto.Changeset.validate_length(...)

Such functionality makes Ecto extremely useful to cast, validate and prune data even
if it is not meant to be persisted to the database.

Changeset actions

Changesets have an action field which is usually set by Ecto.Repo
whenever one of the operations such as insert or update is called:

changeset = User.changeset(%User{}, %{age: 42, email: "mary@example.com"})
{:error, changeset} = Repo.insert(changeset)
changeset.action
#=> :insert

This means that when working with changesets that are not meant to be
persisted to the database, such as schemaless changesets, you may need
to explicitly set the action to one specific value. Frameworks such as
Phoenix use the action value to define how HTML forms should act.

Instead of setting the action manually, you may use apply_action/2 that
emulates operations such as Repo.insert. apply_action/2 will return
{:ok, changes} if the changeset is valid or {:error, changeset}, with
the given action set in the changeset in case of errors.

The Ecto.Changeset struct

The public fields are:


	valid?       - Stores if the changeset is valid


	data         - The changeset source data, for example, a struct


	params       - The parameters as given on changeset creation


	changes      - The changes from parameters that were approved in casting


	errors       - All errors from validations


	required     - All required fields as a list of atoms


	action       - The action to be performed with the changeset


	types        - Cache of the data's field types


	empty_values - A list of values to be considered empty


	repo         - The repository applying the changeset (only set after a Repo function is called)


	repo_opts    - A keyword list of options given to the underlying repository operation




The following fields are private and must not be accessed directly.


	validations


	constraints


	filters


	prepare

    assoc_constraint(changeset, assoc, opts \\ [])
  

    Checks the associated field exists.





  
    cast(data, params, permitted, opts \\ [])
  

    Applies the given params as changes for the given data according to
the given set of permitted keys. Returns a changeset.





  
    cast_assoc(changeset, name, opts \\ [])
  

    Casts the given association with the changeset parameters.





  
    cast_embed(changeset, name, opts \\ [])
  

    Casts the given embed with the changeset parameters.





  
    change(data, changes \\ %{})
  

    Wraps the given data in a changeset or adds changes to a changeset.





  
    check_constraint(changeset, field, opts \\ [])
  

    Checks for a check constraint in the given field.





  
    constraints(changeset)
  

    Returns all constraints in a changeset.





  
    delete_change(changeset, key)
  

    Deletes a change with the given key.





  
    exclusion_constraint(changeset, field, opts \\ [])
  

    Checks for an exclusion constraint in the given field.





  
    fetch_change(changeset, key)
  

    Fetches a change from the given changeset.





  
    fetch_change!(changeset, key)
  

    Same as fetch_change/2 but returns the value or raises if the given key was not found.





  
    fetch_field(changeset, key)
  

    Fetches the given field from changes or from the data.





  
    fetch_field!(changeset, key)
  

    Same as fetch_field/2 but returns the value or raises if the given key was not found.





  
    force_change(changeset, key, value)
  

    Forces a change on the given key with value.





  
    foreign_key_constraint(changeset, field, opts \\ [])
  

    Checks for foreign key constraint in the given field.





  
    get_change(changeset, key, default \\ nil)
  

    Gets a change or returns a default value.





  
    get_field(changeset, key, default \\ nil)
  

    Gets a field from changes or from the data.





  
    merge(changeset1, changeset2)
  

    Merges two changesets.





  
    no_assoc_constraint(changeset, assoc, opts \\ [])
  

    Checks the associated field does not exist.





  
    optimistic_lock(data_or_changeset, field, incrementer \\ &increment_with_rollover/1)
  

    Applies optimistic locking to the changeset.





  
    prepare_changes(changeset, function)
  

    Provides a function executed by the repository on insert/update/delete.





  
    put_assoc(changeset, name, value, opts \\ [])
  

    Puts the given association entry or entries as a change in the changeset.





  
    put_change(changeset, key, value)
  

    Puts a change on the given key with value.





  
    put_embed(changeset, name, value, opts \\ [])
  

    Puts the given embed entry or entries as a change in the changeset.





  
    traverse_errors(changeset, msg_func)
  

    Traverses changeset errors and applies the given function to error messages.





  
    unique_constraint(changeset, field, opts \\ [])
  

    Checks for a unique constraint in the given field.





  
    unsafe_validate_unique(changeset, fields, repo, opts \\ [])
  

    Validates that no existing record with a different primary key
has the same values for these fields.





  
    update_change(changeset, key, function)
  

    Updates a change.





  
    validate_acceptance(changeset, field, opts \\ [])
  

    Validates the given parameter is true.





  
    validate_change(changeset, field, validator)
  

    Validates the given field change.





  
    validate_change(changeset, field, metadata, validator)
  

    Stores the validation metadata and validates the given field change.





  
    validate_confirmation(changeset, field, opts \\ [])
  

    Validates that the given parameter matches its confirmation.





  
    validate_exclusion(changeset, field, data, opts \\ [])
  

    Validates a change is not included in the given enumerable.





  
    validate_format(changeset, field, format, opts \\ [])
  

    Validates a change has the given format.





  
    validate_inclusion(changeset, field, data, opts \\ [])
  

    Validates a change is included in the given enumerable.





  
    validate_length(changeset, field, opts)
  

    Validates a change is a string or list of the given length.





  
    validate_number(changeset, field, opts)
  

    Validates the properties of a number.





  
    validate_required(changeset, fields, opts \\ [])
  

    Validates that one or more fields are present in the changeset.





  
    validate_subset(changeset, field, data, opts \\ [])
  

    Validates a change, of type enum, is a subset of the given enumerable. Like
validate_inclusion/4 for lists.





  
    validations(changeset)
  

    Returns a keyword list of the validations for this changeset.




  

      

      
        
          
            
            Anchor for this section
          
Types        

        

  
    
      
      Link to this type
    
    action()

      
       
       View Source
     
      
          action() :: nil | :insert | :update | :delete | :replace | :ignore | atom()

      

  

  
  


  
    
      
      Link to this type
    
    constraint()

      
       
       View Source
     
      
          constraint() :: %{
  type: :check | :exclusion | :foreign_key | :unique,
  constraint: String.t(),
  match: :exact | :suffix | :prefix,
  field: atom(),
  error_message: String.t(),
  error_type: atom()
}

      

  

  
  


  
    
      
      Link to this type
    
    data()

      
       
       View Source
     
      
          data() :: map()

      

  

  
  


  
    
      
      Link to this type
    
    error()

      
       
       View Source
     
      
          error() :: {String.t(), Keyword.t()}

      

  

  
  


  
    
      
      Link to this type
    
    t()

      
       
       View Source
     
      
          t() :: t(Ecto.Schema.t() | map() | nil)

      

  

  
  


  
    
      
      Link to this type
    
    t(data_type)

      
       
       View Source
     
      
          t(data_type) :: %Ecto.Changeset{
  action: action(),
  changes: %{required(atom()) => term()},
  constraints: [constraint()],
  data: data_type,
  empty_values: term(),
  errors: [{atom(), error()}],
  filters: %{required(atom()) => term()},
  params: %{required(String.t()) => term()} | nil,
  prepare: [(t() -> t())],
  repo: atom() | nil,
  repo_changes: %{required(atom()) => term()},
  repo_opts: Keyword.t(),
  required: [atom()],
  types: nil | %{required(atom()) => Ecto.Type.t()},
  valid?: boolean(),
  validations: [{atom(), term()}]
}

      

  

  
  


  
    
      
      Link to this type
    
    types()

      
       
       View Source
     
      
          types() :: map()

      

  

  
  

        

      
      
        
          
            
            Anchor for this section
          
Functions        

        

    
  
    
      
      Link to this function
    
    add_error(changeset, key, message, keys \\ [])

      
       
       View Source
     
      
          add_error(t(), atom(), String.t(), Keyword.t()) :: t()

      

  

  
Adds an error to the changeset.

An additional keyword list keys can be passed to provide additional
contextual information for the error. This is useful when using
traverse_errors/2


  
  Examples



iex> changeset = change(%Post{}, %{title: ""})
iex> changeset = add_error(changeset, :title, "empty")
iex> changeset.errors
[title: {"empty", []}]
iex> changeset.valid?
false

iex> changeset = change(%Post{}, %{title: ""})
iex> changeset = add_error(changeset, :title, "empty", additional: "info")
iex> changeset.errors
[title: {"empty", [additional: "info"]}]
iex> changeset.valid?
false

  


  
    
      
      Link to this function
    
    apply_action(changeset, action)

      
       
       View Source
     
      
          apply_action(t(), atom()) ::
  {:ok, Ecto.Schema.t() | data()} | {:error, t()}

      

  

  
Applies the changeset action only if the changes are valid.

If the changes are valid, all changes are applied to the changeset data.
If the changes are invalid, no changes are applied, and an error tuple
is returned with the changeset containing the action that was attempted
to be applied.

The action may be any atom.


  
  Examples



iex> {:ok, data} = apply_action(changeset, :update)

iex> {:error, changeset} = apply_action(changeset, :update)
%Ecto.Changeset{action: :update}

  


  
    
      
      Link to this function
    
    apply_action!(changeset, action)

      
       
       View Source
     
      
          apply_action!(t(), atom()) :: Ecto.Schema.t() | data()

      

  

  
Applies the changeset action if the changes are valid or raises an error.


  
  Examples



iex> changeset = change(%Post{author: "bar"}, %{title: "foo"})
iex> apply_action!(changeset, :update)
%Post{author: "bar", title: "foo"}

iex> changeset = change(%Post{author: "bar"}, %{title: :bad})
iex> apply_action!(changeset, :update)
** (Ecto.InvalidChangesetError) could not perform update because changeset is invalid.

See apply_action/2 for more information.

  


  
    
      
      Link to this function
    
    apply_changes(changeset)

      
       
       View Source
     
      
          apply_changes(t()) :: Ecto.Schema.t() | data()

      

  

  
Applies the changeset changes to the changeset data.

This operation will return the underlying data with changes
regardless if the changeset is valid or not.


  
  Examples



iex> changeset = change(%Post{author: "bar"}, %{title: "foo"})
iex> apply_changes(changeset)
%Post{author: "bar", title: "foo"}

  


    
  
    
      
      Link to this function
    
    assoc_constraint(changeset, assoc, opts \\ [])

      
       
       View Source
     
      
          assoc_constraint(t(), atom(), Keyword.t()) :: t()

      

  

  
Checks the associated field exists.

This is similar to foreign_key_constraint/3 except that the
field is inferred from the association definition. This is useful
to guarantee that a child will only be created if the parent exists
in the database too. Therefore, it only applies to belongs_to
associations.

As the name says, a constraint is required in the database for
this function to work. Such constraint is often added as a
reference to the child table:

create table(:comments) do
  add :post_id, references(:posts)
end

Now, when inserting a comment, it is possible to forbid any
comment to be added if the associated post does not exist:

comment
|> Ecto.Changeset.cast(params, [:post_id])
|> Ecto.Changeset.assoc_constraint(:post)
|> Repo.insert


  
  Options




	:message - the message in case the constraint check fails,
defaults to "does not exist"


	:name - the constraint name. By default, the constraint
name is inferred from the table + association field.
May be required explicitly for complex cases




  


    
  
    
      
      Link to this function
    
    cast(data, params, permitted, opts \\ [])

      
       
       View Source
     
      
          cast(
  Ecto.Schema.t() | t() | {data(), types()},
  %{required(binary()) => term()} | %{required(atom()) => term()} | :invalid,
  [atom()],
  Keyword.t()
) :: t()

      

  

  
Applies the given params as changes for the given data according to
the given set of permitted keys. Returns a changeset.

The given data may be either a changeset, a schema struct or a {data, types}
tuple. The second argument is a map of params that are cast according
to the type information from data. params is a map with string keys
or a map with atom keys containing potentially invalid data.

During casting, all permitted parameters whose values match the specified
type information will have their key name converted to an atom and stored
together with the value as a change in the :changes field of the changeset.
All parameters that are not explicitly permitted are ignored.

If casting of all fields is successful, the changeset is returned as valid.

Note that cast/4 validates the types in the params, but not in the given
data.


  
  Options




	:empty_values - a list of values to be considered as empty when casting.
Defaults to the changeset value, which defaults to [""]





  
  Examples



iex> changeset = cast(post, params, [:title])
iex> if changeset.valid? do
...>   Repo.update!(changeset)
...> end

Passing a changeset as the first argument:

iex> changeset = cast(post, %{title: "Hello"}, [:title])
iex> new_changeset = cast(changeset, %{title: "Foo", body: "World"}, [:body])
iex> new_changeset.params
%{"title" => "Hello", "body" => "World"}

Or creating a changeset from a simple map with types:

iex> data = %{title: "hello"}
iex> types = %{title: :string}
iex> changeset = cast({data, types}, %{title: "world"}, [:title])
iex> apply_changes(changeset)
%{title: "world"}


  
  Composing casts



cast/4 also accepts a changeset as its first argument. In such cases, all
the effects caused by the call to cast/4 (additional errors and changes)
are simply added to the ones already present in the argument changeset.
Parameters are merged (not deep-merged) and the ones passed to cast/4
take precedence over the ones already in the changeset.

  


    
  
    
      
      Link to this function
    
    cast_assoc(changeset, name, opts \\ [])

      
       
       View Source
     
  

  
Casts the given association with the changeset parameters.

This function should be used when working with the entire association at
once (and not a single element of a many-style association) and using data
external to the application.

cast_assoc/3 works matching the records extracted from the database (preload)
and compares it with the parameters provided from an external source.

For example, imagine a user has many addresses relationship where
post data is sent as follows

%{"name" => "john doe", "addresses" => [
  %{"street" => "somewhere", "country" => "brazil", "id" => 1},
  %{"street" => "elsewhere", "country" => "poland"},
]}

and then

User
|> Repo.get!(id)
|> Repo.preload(:addresses) # Only required when updating data
|> Ecto.Changeset.cast(params, [])
|> Ecto.Changeset.cast_assoc(:addresses, with: &MyApp.Address.changeset/2)

The parameters for the given association will be retrieved
from changeset.params. Those parameters are expected to be
a map with attributes, similar to the ones passed to cast/4.
Once parameters are retrieved, cast_assoc/3 will match those
parameters with the associations already in the changeset record.

Once cast_assoc/3 is called, Ecto will compare each parameter
with the user's already preloaded addresses and act as follows:


	If the parameter does not contain an ID, the parameter data
will be passed to MyApp.Address.changeset/2 with a new struct
and become an insert operation


	If the parameter contains an ID and there is no associated child
with such ID, the parameter data will be passed to
MyApp.Address.changeset/2 with a new struct and become an insert
operation


	If the parameter contains an ID and there is an associated child
with such ID, the parameter data will be passed to
MyApp.Address.changeset/2 with the existing struct and become an
update operation


	If there is an associated child with an ID and its ID is not given
as parameter, the :on_replace callback for that association will
be invoked (see the "On replace" section on the module documentation)




Every time the MyApp.Address.changeset/2 function is invoked, it must
return a changeset. This changeset will be applied to your Repo with
the proper action accordingly.

Note developers are allowed to explicitly set the :action field of a
changeset to instruct Ecto how to act in certain situations. Let's suppose
that, if one of the associations has only empty fields, you want to ignore
the entry altogether instead of showing an error. The changeset function could
be written like this:

def changeset(struct, params) do
  struct
  |> cast(params, [:title, :body])
  |> validate_required([:title, :body])
  |> case do
    %{valid?: false, changes: changes} = changeset when changes == %{} ->
      # If the changeset is invalid and has no changes, it is
      # because all required fields are missing, so we ignore it.
      %{changeset | action: :ignore}
    changeset ->
      changeset
  end
end


  
  Partial changes for many-style associations



By preloading an association using a custom query you can confine the behavior
of cast_assoc/3. This opens up the possibility to work on a subset of the data,
instead of all associations in the database.

Taking the initial example of users having addresses imagine those addresses
are set up to belong to a country. If you want to allow users to bulk edit all
addresses that belong to a single country, you can do so by changing the preload
query:

query = from MyApp.Address, where: [country: ^edit_country]

User
|> Repo.get!(id)
|> Repo.preload(addresses: query)
|> Ecto.Changeset.cast(params, [])
|> Ecto.Changeset.cast_assoc(:addresses)

This will allow you to cast and update only the association for the given country.
The important point for partial changes is that any addresses, which were not
preloaded won't be changed.


  
  Options




	:required - if the association is a required field


	:required_message - the message on failure, defaults to "can't be blank"


	:invalid_message - the message on failure, defaults to "is invalid"


	:force_update_on_change - force the parent record to be updated in the
repository if there is a change, defaults to true


	:with - the function to build the changeset from params. Defaults to the
changeset/2 function of the associated module. It can be changed by passing
an anonymous function or an MFA tuple.  If using an MFA, the default changeset
and parameters arguments will be prepended to the given args. For example,
using with: {Author, :special_changeset, ["hello"]} will be invoked as
Author.special_changeset(changeset, params, "hello")




  


    
  
    
      
      Link to this function
    
    cast_embed(changeset, name, opts \\ [])

      
       
       View Source
     
  

  
Casts the given embed with the changeset parameters.

The parameters for the given embed will be retrieved
from changeset.params. Those parameters are expected to be
a map with attributes, similar to the ones passed to cast/4.
Once parameters are retrieved, cast_embed/3 will match those
parameters with the embeds already in the changeset record.
See cast_assoc/3 for an example of working with casts and
associations which would also apply for embeds.

The changeset must have been previously cast using
cast/4 before this function is invoked.


  
  Options




	:required - if the embed is a required field


	:required_message - the message on failure, defaults to "can't be blank"


	:invalid_message - the message on failure, defaults to "is invalid"


	:force_update_on_change - force the parent record to be updated in the
repository if there is a change, defaults to true


	:with - the function to build the changeset from params. Defaults to the
changeset/2 function of the embedded module. It can be changed by passing
an anonymous function or an MFA tuple.  If using an MFA, the default changeset
and parameters arguments will be prepended to the given args. For example,
using with: {Author, :special_changeset, ["hello"]} will be invoked as
Author.special_changeset(changeset, params, "hello")




  


    
  
    
      
      Link to this function
    
    change(data, changes \\ %{})

      
       
       View Source
     
      
          change(
  Ecto.Schema.t() | t() | {data(), types()},
  %{required(atom()) => term()} | Keyword.t()
) :: t()

      

  

  
Wraps the given data in a changeset or adds changes to a changeset.

changes is a map or keyword where the key is an atom representing a
field, association or embed and the value is a term. Note the value is
directly stored in the changeset with no validation whatsoever. For this
reason, this function is meant for working with data internal to the
application.

When changing embeds and associations, see put_assoc/4 for a complete
reference on the accepted values.

This function is useful for:


	wrapping a struct inside a changeset


	directly changing a struct without performing castings nor validations


	directly bulk-adding changes to a changeset




Changed attributes will only be added if the change does not have the
same value as the field in the data.

When a changeset is passed as the first argument, the changes passed as the
second argument are merged over the changes already in the changeset if they
differ from the values in the struct.

When a {data, types} is passed as the first argument, a changeset is
created with the given data and types and marked as valid.

See cast/4 if you'd prefer to cast and validate external parameters.


  
  Examples



iex> changeset = change(%Post{})
%Ecto.Changeset{...}
iex> changeset.valid?
true
iex> changeset.changes
%{}

iex> changeset = change(%Post{author: "bar"}, title: "title")
iex> changeset.changes
%{title: "title"}

iex> changeset = change(%Post{title: "title"}, title: "title")
iex> changeset.changes
%{}

iex> changeset = change(changeset, %{title: "new title", body: "body"})
iex> changeset.changes.title
"new title"
iex> changeset.changes.body
"body"

  


    
  
    
      
      Link to this function
    
    check_constraint(changeset, field, opts \\ [])

      
       
       View Source
     
  

  
Checks for a check constraint in the given field.

The check constraint works by relying on the database to check
if the check constraint has been violated or not and, if so,
Ecto converts it into a changeset error.

In order to use the check constraint, the first step is
to define the check constraint in a migration:

create constraint("users", :price_must_be_positive, check: "price > 0")

Now that a constraint exists, when modifying users, we could
annotate the changeset with a check constraint so Ecto knows
how to convert it into an error message:

cast(user, params, [:price])
|> check_constraint(:price, name: :price_must_be_positive)

Now, when invoking Repo.insert/2 or Repo.update/2, if the
price is not positive, it will be converted into an error and
{:error, changeset} returned by the repository. Note that the error
will occur only after hitting the database so it will not be visible
until all other validations pass.


  
  Options




	:message - the message in case the constraint check fails.
Defaults to "is invalid"


	:name - the name of the constraint. Required.


	:match - how the changeset constraint name is matched against the
repo constraint, may be :exact, :suffix or :prefix. Defaults to :exact.
:suffix matches any repo constraint which ends_with? :name
 to this changeset constraint.
:prefix matches any repo constraint which starts_with? :name
 to this changeset constraint.




  


  
    
      
      Link to this function
    
    constraints(changeset)

      
       
       View Source
     
      
          constraints(t()) :: [constraint()]

      

  

  
Returns all constraints in a changeset.

A constraint is a map with the following fields:


	:type - the type of the constraint that will be checked in the database,
such as :check, :unique, etc


	:constraint - the database constraint name as a string


	:match - the type of match Ecto will perform on a violated constraint
against the :constraint value. It is :exact, :suffix or :prefix


	:field - the field a violated constraint will apply the error to


	:error_message - the error message in case of violated constraints


	:error_type - the type of error that identifies the error message




  


  
    
      
      Link to this function
    
    delete_change(changeset, key)

      
       
       View Source
     
      
          delete_change(t(), atom()) :: t()

      

  

  
Deletes a change with the given key.


  
  Examples



iex> changeset = change(%Post{}, %{title: "foo"})
iex> changeset = delete_change(changeset, :title)
iex> get_change(changeset, :title)
nil

  


    
  
    
      
      Link to this function
    
    exclusion_constraint(changeset, field, opts \\ [])

      
       
       View Source
     
  

  
Checks for an exclusion constraint in the given field.

The exclusion constraint works by relying on the database to check
if the exclusion constraint has been violated or not and, if so,
Ecto converts it into a changeset error.


  
  Options




	:message - the message in case the constraint check fails,
defaults to "violates an exclusion constraint"


	:name - the constraint name. By default, the constraint
name is inferred from the table + field. May be required
explicitly for complex cases


	:match - how the changeset constraint name is matched against the
repo constraint, may be :exact or :suffix. Defaults to :exact.
:suffix matches any repo constraint which ends_with? :name
 to this changeset constraint.




  


  
    
      
      Link to this function
    
    fetch_change(changeset, key)

      
       
       View Source
     
      
          fetch_change(t(), atom()) :: {:ok, term()} | :error

      

  

  
Fetches a change from the given changeset.

This function only looks at the :changes field of the given changeset and
returns {:ok, value} if the change is present or :error if it's not.


  
  Examples



iex> changeset = change(%Post{body: "foo"}, %{title: "bar"})
iex> fetch_change(changeset, :title)
{:ok, "bar"}
iex> fetch_change(changeset, :body)
:error

  


  
    
      
      Link to this function
    
    fetch_change!(changeset, key)

      
       
       View Source
     
      
          fetch_change!(t(), atom()) :: term()

      

  

  
Same as fetch_change/2 but returns the value or raises if the given key was not found.


  
  Examples



iex> changeset = change(%Post{body: "foo"}, %{title: "bar"})
iex> fetch_change!(changeset, :title)
"bar"
iex> fetch_change!(changeset, :body)
** (KeyError) key :body not found in: %{title: "bar"}

  


  
    
      
      Link to this function
    
    fetch_field(changeset, key)

      
       
       View Source
     
      
          fetch_field(t(), atom()) :: {:changes, term()} | {:data, term()} | :error

      

  

  
Fetches the given field from changes or from the data.

While fetch_change/2 only looks at the current changes
to retrieve a value, this function looks at the changes and
then falls back on the data, finally returning :error if
no value is available.

For relations, these functions will return the changeset
original data with changes applied. To retrieve raw changesets,
please use fetch_change/2.


  
  Examples



iex> post = %Post{title: "Foo", body: "Bar baz bong"}
iex> changeset = change(post, %{title: "New title"})
iex> fetch_field(changeset, :title)
{:changes, "New title"}
iex> fetch_field(changeset, :body)
{:data, "Bar baz bong"}
iex> fetch_field(changeset, :not_a_field)
:error

  


  
    
      
      Link to this function
    
    fetch_field!(changeset, key)

      
       
       View Source
     
      
          fetch_field!(t(), atom()) :: term()

      

  

  
Same as fetch_field/2 but returns the value or raises if the given key was not found.


  
  Examples



iex> post = %Post{title: "Foo", body: "Bar baz bong"}
iex> changeset = change(post, %{title: "New title"})
iex> fetch_field!(changeset, :title)
"New title"
iex> fetch_field!(changeset, :other)
** (KeyError) key :other not found in: %Post{...}

  


  
    
      
      Link to this function
    
    force_change(changeset, key, value)

      
       
       View Source
     
      
          force_change(t(), atom(), term()) :: t()

      

  

  
Forces a change on the given key with value.

If the change is already present, it is overridden with
the new value.


  
  Examples



iex> changeset = change(%Post{author: "bar"}, %{title: "foo"})
iex> changeset = force_change(changeset, :title, "bar")
iex> changeset.changes
%{title: "bar"}

iex> changeset = force_change(changeset, :author, "bar")
iex> changeset.changes
%{title: "bar", author: "bar"}

  


    
  
    
      
      Link to this function
    
    foreign_key_constraint(changeset, field, opts \\ [])

      
       
       View Source
     
      
          foreign_key_constraint(t(), atom(), Keyword.t()) :: t()

      

  

  
Checks for foreign key constraint in the given field.

The foreign key constraint works by relying on the database to
check if the associated data exists or not. This is useful to
guarantee that a child will only be created if the parent exists
in the database too.

In order to use the foreign key constraint the first step is
to define the foreign key in a migration. This is often done
with references. For example, imagine you are creating a
comments table that belongs to posts. One would have:

create table(:comments) do
  add :post_id, references(:posts)
end

By default, Ecto will generate a foreign key constraint with
name "comments_post_id_fkey" (the name is configurable).

Now that a constraint exists, when creating comments, we could
annotate the changeset with foreign key constraint so Ecto knows
how to convert it into an error message:

cast(comment, params, [:post_id])
|> foreign_key_constraint(:post_id)

Now, when invoking Repo.insert/2 or Repo.update/2, if the
associated post does not exist, it will be converted into an
error and {:error, changeset} returned by the repository.


  
  Options




	:message - the message in case the constraint check fails,
defaults to "does not exist"


	:name - the constraint name. By default, the constraint
name is inferred from the table + field. May be required
explicitly for complex cases




  


    
  
    
      
      Link to this function
    
    get_change(changeset, key, default \\ nil)

      
       
       View Source
     
      
          get_change(t(), atom(), term()) :: term()

      

  

  
Gets a change or returns a default value.


  
  Examples



iex> changeset = change(%Post{body: "foo"}, %{title: "bar"})
iex> get_change(changeset, :title)
"bar"
iex> get_change(changeset, :body)
nil

  


    
  
    
      
      Link to this function
    
    get_field(changeset, key, default \\ nil)

      
       
       View Source
     
      
          get_field(t(), atom(), term()) :: term()

      

  

  
Gets a field from changes or from the data.

While get_change/3 only looks at the current changes
to retrieve a value, this function looks at the changes and
then falls back on the data, finally returning default if
no value is available.

For relations, these functions will return the changeset data
with changes applied. To retrieve raw changesets, please use get_change/3.

iex> post = %Post{title: "A title", body: "My body is a cage"}
iex> changeset = change(post, %{title: "A new title"})
iex> get_field(changeset, :title)
"A new title"
iex> get_field(changeset, :not_a_field, "Told you, not a field!")
"Told you, not a field!"

  


  
    
      
      Link to this function
    
    merge(changeset1, changeset2)

      
       
       View Source
     
      
          merge(t(), t()) :: t()

      

  

  
Merges two changesets.

This function merges two changesets provided they have been applied to the
same data (their :data field is equal); if the data differs, an
ArgumentError exception is raised. If one of the changesets has a :repo
field which is not nil, then the value of that field is used as the :repo
field of the resulting changeset; if both changesets have a non-nil and
different :repo field, an ArgumentError exception is raised.

The other fields are merged with the following criteria:


	params - params are merged (not deep-merged) giving precedence to the
params of changeset2 in case of a conflict. If both changesets have their
:params fields set to nil, the resulting changeset will have its params
set to nil too.


	changes - changes are merged giving precedence to the changeset2
changes.


	errors and validations - they are simply concatenated.


	required - required fields are merged; all the fields that appear
in the required list of both changesets are moved to the required
list of the resulting changeset.





  
  Examples



iex> changeset1 = cast(%Post{}, %{title: "Title"}, [:title])
iex> changeset2 = cast(%Post{}, %{title: "New title", body: "Body"}, [:title, :body])
iex> changeset = merge(changeset1, changeset2)
iex> changeset.changes
%{body: "Body", title: "New title"}

iex> changeset1 = cast(%Post{body: "Body"}, %{title: "Title"}, [:title])
iex> changeset2 = cast(%Post{}, %{title: "New title"}, [:title])
iex> merge(changeset1, changeset2)
** (ArgumentError) different :data when merging changesets

  


    
  
    
      
      Link to this function
    
    no_assoc_constraint(changeset, assoc, opts \\ [])

      
       
       View Source
     
      
          no_assoc_constraint(t(), atom(), Keyword.t()) :: t()

      

  

  
Checks the associated field does not exist.

This is similar to foreign_key_constraint/3 except that the
field is inferred from the association definition. This is useful
to guarantee that parent can only be deleted (or have its primary
key changed) if no child exists in the database. Therefore, it only
applies to has_* associations.

As the name says, a constraint is required in the database for
this function to work. Such constraint is often added as a
reference to the child table:

create table(:comments) do
  add :post_id, references(:posts)
end

Now, when deleting the post, it is possible to forbid any post to
be deleted if they still have comments attached to it:

post
|> Ecto.Changeset.change
|> Ecto.Changeset.no_assoc_constraint(:comments)
|> Repo.delete


  
  Options




	:message - the message in case the constraint check fails,
defaults to "is still associated with this entry" (for has_one)
and "are still associated with this entry" (for has_many)


	:name - the constraint name. By default, the constraint
name is inferred from the association table + association
field. May be required explicitly for complex cases




  


    
  
    
      
      Link to this function
    
    optimistic_lock(data_or_changeset, field, incrementer \\ &increment_with_rollover/1)

      
       
       View Source
     
      
          optimistic_lock(Ecto.Schema.t() | t(), atom(), (term() -> term())) :: t()

      

  

  
Applies optimistic locking to the changeset.

Optimistic
locking (or
optimistic concurrency control) is a technique that allows concurrent edits
on a single record. While pessimistic locking works by locking a resource for
an entire transaction, optimistic locking only checks if the resource changed
before updating it.

This is done by regularly fetching the record from the database, then checking
whether another user has made changes to the record only when updating the
record. This behaviour is ideal in situations where the chances of concurrent
updates to the same record are low; if they're not, pessimistic locking or
other concurrency patterns may be more suited.


  
  Usage



Optimistic locking works by keeping a "version" counter for each record; this
counter gets incremented each time a modification is made to a record. Hence,
in order to use optimistic locking, a field must exist in your schema for
versioning purpose. Such field is usually an integer but other types are
supported.


  
  Examples



Assuming we have a Post schema (stored in the posts table), the first step
is to add a version column to the posts table:

alter table(:posts) do
  add :lock_version, :integer, default: 1
end

The column name is arbitrary and doesn't need to be :lock_version. Now add
a field to the schema too:

defmodule Post do
  use Ecto.Schema

  schema "posts" do
    field :title, :string
    field :lock_version, :integer, default: 1
  end

  def changeset(:update, struct, params \\ %{}) do
    struct
    |> Ecto.Changeset.cast(params, [:title])
    |> Ecto.Changeset.optimistic_lock(:lock_version)
  end
end

Now let's take optimistic locking for a spin:

iex> post = Repo.insert!(%Post{title: "foo"})
%Post{id: 1, title: "foo", lock_version: 1}
iex> valid_change = Post.changeset(:update, post, %{title: "bar"})
iex> stale_change = Post.changeset(:update, post, %{title: "baz"})
iex> Repo.update!(valid_change)
%Post{id: 1, title: "bar", lock_version: 2}
iex> Repo.update!(stale_change)
** (Ecto.StaleEntryError) attempted to update a stale entry:

%Post{id: 1, title: "baz", lock_version: 1}

When a conflict happens (a record which has been previously fetched is
being updated, but that same record has been modified since it was
fetched), an Ecto.StaleEntryError exception is raised.

Optimistic locking also works with delete operations. Just call the
optimistic_lock/3 function with the data before delete:

iex> changeset = Ecto.Changeset.optimistic_lock(post, :lock_version)
iex> Repo.delete(changeset)

optimistic_lock/3 by default assumes the field
being used as a lock is an integer. If you want to use another type,
you need to pass the third argument customizing how the next value
is generated:

iex> Ecto.Changeset.optimistic_lock(post, :lock_uuid, fn _ -> Ecto.UUID.generate end)

  


  
    
      
      Link to this function
    
    prepare_changes(changeset, function)

      
       
       View Source
     
      
          prepare_changes(t(), (t() -> t())) :: t()

      

  

  
Provides a function executed by the repository on insert/update/delete.

If the changeset given to the repository is valid, the function given to
prepare_changes/2 will be called with the changeset and must return a
changeset, allowing developers to do final adjustments to the changeset or
to issue data consistency commands. The repository itself can be accessed
inside the function under the repo field in the changeset. If the
changeset given to the repository is invalid, the function will not be
invoked.

The given function is guaranteed to run inside the same transaction
as the changeset operation for databases that do support transactions.


  
  Example



A common use case is updating a counter cache, in this case updating a post's
comment count when a comment is created:

def create_comment(comment, params) do
  comment
  |> cast(params, [:body, :post_id])
  |> prepare_changes(fn changeset ->
       if post_id = get_change(changeset, :post_id) do
         query = from Post, where: [id: ^post_id]
         changeset.repo.update_all(query, inc: [comment_count: 1])
       end
       changeset
     end)
end

We retrieve the repo from the comment changeset itself and use
update_all to update the counter cache in one query. Finally, the original
changeset must be returned.

  


    
  
    
      
      Link to this function
    
    put_assoc(changeset, name, value, opts \\ [])

      
       
       View Source
     
  

  
Puts the given association entry or entries as a change in the changeset.

This function is used to work with associations as a whole. For example,
if a Post has many Comments, it allows you to add, remove or change all
comments at once. If your goal is to simply add a new comment to a post,
then it is preferred to do so manually, as we will describe later in the
"Example: Adding a comment to a post" section.

This function requires the associated data to have been preloaded, except
when the parent changeset has been newly built and not yet persisted.
Missing data will invoke the :on_replace behaviour defined on the
association.

For associations with cardinality one, nil can be used to remove the existing
entry. For associations with many entries, an empty list may be given instead.

If the association has no changes, it will be skipped. If the association is
invalid, the changeset will be marked as invalid. If the given value is not any
of values below, it will raise.

The associated data may be given in different formats:


	a map or a keyword list representing changes to be applied to the
associated data. A map or keyword list can be given to update the
associated data as long as they have matching primary keys.
For example, put_assoc(changeset, :comments, [%{id: 1, title: "changed"}])
will locate the comment with :id of 1 and update its title.
If no comment with such id exists, one is created on the fly.
Since only a single comment was given, any other associated comment
will be replaced. On all cases, it is expected the keys to be atoms.
This API is mostly used in scripts and tests, to make it straight-
forward to create schemas with associations at once, such as:

Ecto.Changeset.change(
  %Post{},
  title: "foo",
  comments: [
    %{body: "first"},
    %{body: "second"}
  ]
)



	changesets or structs - when a changeset or struct is given, they
are treated as the canonical data and the associated data currently
stored in the association is ignored. For instance, the operation
put_assoc(changeset, :comments, [%Comment{id: 1, title: "changed"}])
will send the Comment as is to the database, ignoring any comment
currently associated, even if a matching ID is found. If the comment
is already persisted to the database, then put_assoc/4 only takes
care of guaranteeing that the comments and the parent data are associated.
This extremely useful when associating existing data, as we will see
in the "Example: Adding tags to a post" section.





Note, however, that put_assoc/4 always expects all data currently associated to
be given. In both examples above, if the changeset has any other comment besides
the comment with id equal to 1, all of them will be considered as replaced,
invoking the relevant :on_replace callback which may potentially remove the
data. In other words, if only a comment with a id equal to 1 is given, it will
be the only one kept. Therefore, put_assoc/4 always works with the whole data,
which may be undesired in some cases. Let's see an example.


  
  Example: Adding a comment to a post



Imagine a relationship where Post has many comments and you want to add a
new comment to an existing post. While it is possible to use put_assoc/4
for this, it would be unecessarily complex. Let's see an example.

First, let's fetch the post with all existing comments:

post = Post |> Repo.get!(1) |> Repo.preload(:comments)

The following approach is wrong:

post
|> Ecto.Changeset.change()
|> Ecto.Changeset.put_assoc(:comments, [%Comment{body: "bad example!"}])
|> Repo.update!()

The reason why the example above is wrong is because put_assoc/4 always
works with the full data. So the example above will effectively erase
all previous comments and only keep the comment you are currently adding.
Instead, you could try:

post
|> Ecto.Changeset.change()
|> Ecto.Changeset.put_assoc(:comments, [%Comment{body: "so-so example!"} | post.comments])
|> Repo.update!()

In this example, we prepend the new comment to the list of existing comments.
Ecto will diff the list of comments currently in post with the list of comments
given, and correctly insert the new comment to the database. Note, however,
Ecto is doing a lot of work just to figure out something we knew since the
beginning, which is that there is only one new comment.

In cases like above, when you want to work only on a single entry, it is
much easier to simply work on the associated directly. For example, we
could instead set the post association in the comment:

%Comment{body: "better example"}
|> Ecto.Changeset.change()
|> Ecto.Changeset.put_assoc(:post, post)
|> Repo.insert!()

Alternatively, we can make sure that when we create a comment, it is already
associated to the post:

Ecto.build_assoc(post, :comments)
|> Ecto.Changeset.change(body: "great example!")
|> Repo.insert!()

Or we can simply set the post_id in the comment itself:

%Comment{body: "better example", post_id: post.id}
|> Repo.insert!()

In other words, when you find yourself wanting to work only with a subset
of the data, then using put_assoc/4 is most likely unnecessary. Instead,
you want to work on the other side of the association.

Let's see an example where using put_assoc/4 is a good fit.


  
  Example: Adding tags to a post



Imagine you are receiving a set of tags you want to associate to a post.
Let's imagine that those tags exist upfront and are all persisted to the
database. Imagine we get the data in this format:

params = %{"title" => "new post", "tags" => ["learner"]}

Now, since the tags already exist, we will bring all of them from the
database and put them directly in the post:

tags = Repo.all(from t in Tag, where: t.name in ^params["tags"])

post
|> Repo.preload(:tags)
|> Ecto.Changeset.cast(params, [:title]) # No need to allow :tags as we put them directly
|> Ecto.Changeset.put_assoc(:tags, tags) # Explicitly set the tags

Since in this case we always require the user to pass all tags
directly, using put_assoc/4 is a great fit. It will automatically
remove any tag not given and properly associate all of the given
tags with the post.

Furthermore, since the tag information is given as structs read directly
from the database, Ecto will treat the data as correct and only do the
minimum necessary to guarantee that posts and tags are associated,
without trying to update or diff any of the fields in the tag struct.

Although it accepts an opts argument, there are no options currently
supported by put_assoc/4.

  


  
    
      
      Link to this function
    
    put_change(changeset, key, value)

      
       
       View Source
     
      
          put_change(t(), atom(), term()) :: t()

      

  

  
Puts a change on the given key with value.

key is an atom that represents any field, embed or
association in the changeset. Note the value is directly
stored in the changeset with no validation whatsoever.
For this reason, this function is meant for working with
data internal to the application.

If the change is already present, it is overridden with
the new value. If the change has the same value as in the
changeset data, it is not added to the list of changes.

When changing embeds and associations, see put_assoc/4
for a complete reference on the accepted values.


  
  Examples



iex> changeset = change(%Post{author: "bar"}, %{title: "foo"})
iex> changeset = put_change(changeset, :title, "bar")
iex> changeset.changes
%{title: "bar"}

iex> changeset = put_change(changeset, :author, "bar")
iex> changeset.changes
%{title: "bar"}

  


    
  
    
      
      Link to this function
    
    put_embed(changeset, name, value, opts \\ [])

      
       
       View Source
     
  

  
Puts the given embed entry or entries as a change in the changeset.

This function is used to work with embeds as a whole. For embeds with
cardinality one, nil can be used to remove the existing entry. For
embeds with many entries, an empty list may be given instead.

If the embed has no changes, it will be skipped. If the embed is
invalid, the changeset will be marked as invalid.

The list of supported values and their behaviour is described in
put_assoc/4. If the given value is not any of values listed there,
it will raise.

Although this function accepts an opts argument, there are no options
currently supported by put_embed/4.

  


  
    
      
      Link to this function
    
    traverse_errors(changeset, msg_func)

      
       
       View Source
     
      
          traverse_errors(
  t(),
  (error() -> String.t()) | (t(), atom(), error() -> String.t())
) :: %{required(atom()) => [String.t()]}

      

  

  
Traverses changeset errors and applies the given function to error messages.

This function is particularly useful when associations and embeds
are cast in the changeset as it will traverse all associations and
embeds and place all errors in a series of nested maps.

A changeset is supplied along with a function to apply to each
error message as the changeset is traversed. The error message
function receives an error tuple {msg, opts}, for example:

{"should be at least %{count} characters", [count: 3, validation: :length, min: 3]}


  
  Examples



iex> traverse_errors(changeset, fn {msg, opts} ->
...>   Enum.reduce(opts, msg, fn {key, value}, acc ->
...>     String.replace(acc, "%{#{key}}", to_string(value))
...>   end)
...> end)
%{title: ["should be at least 3 characters"]}

Optionally function can accept three arguments: changeset, field
and error tuple {msg, opts}. It is useful whenever you want to extract
validations rules from changeset.validations to build detailed error
description.

  


    
  
    
      
      Link to this function
    
    unique_constraint(changeset, field, opts \\ [])

      
       
       View Source
     
      
          unique_constraint(t(), atom(), Keyword.t()) :: t()

      

  

  
Checks for a unique constraint in the given field.

The unique constraint works by relying on the database to check
if the unique constraint has been violated or not and, if so,
Ecto converts it into a changeset error.

In order to use the uniqueness constraint, the first step is
to define the unique index in a migration:

create unique_index(:users, [:email])

Now that a constraint exists, when modifying users, we could
annotate the changeset with a unique constraint so Ecto knows
how to convert it into an error message:

cast(user, params, [:email])
|> unique_constraint(:email)

Now, when invoking Repo.insert/2 or Repo.update/2, if the
email already exists, it will be converted into an error and
{:error, changeset} returned by the repository. Note that the error
will occur only after hitting the database so it will not be visible
until all other validations pass.


  
  Options




	:message - the message in case the constraint check fails,
defaults to "has already been taken"


	:name - the constraint name. By default, the constraint
name is inferred from the table + field. May be required
explicitly for complex cases


	:match - how the changeset constraint name is matched against the
repo constraint, may be :exact or :suffix. Defaults to :exact.
:suffix matches any repo constraint which ends_with? :name
 to this changeset constraint.





  
  Complex constraints



Because the constraint logic is in the database, we can leverage
all the database functionality when defining them. For example,
let's suppose the e-mails are scoped by company id. We would write
in a migration:

create unique_index(:users, [:email, :company_id])

Because such indexes have usually more complex names, we need
to explicitly tell the changeset which constraint name to use (here we're
using the naming convention that unique_index uses):

cast(user, params, [:email])
|> unique_constraint(:email, name: :users_email_company_id_index)

Notice that the first param is just one of the unique index fields, this will
be used as the error key to the changeset errors keyword list. For example,
the above unique_constraint/3 would generate something like:

Repo.insert!(%User{email: "john@elixir.org", company_id: 1})
changeset = User.changeset(%User{}, %{email: "john@elixir.org", company_id: 1})
{:error, changeset} = Repo.insert(changeset)
changeset.errors #=> [email: {"has already been taken", []}]

Alternatively, you can give both unique_index and unique_constraint
the same name:

# In the migration
create unique_index(:users, [:email, :company_id], name: :users_email_company_id_index)

# In the changeset function
cast(user, params, [:email])
|> unique_constraint(:email, name: :users_email_company_id_index)


  
  Partitioning



If your table is partitioned, then your unique index might look different
per partition - Postgres adds p to the middle of your key, like:
  
    
    Ecto.Multi - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Multi    



      
Ecto.Multi is a data structure for grouping multiple Repo operations.

Ecto.Multi makes it possible to pack operations that should be
performed in a single database transaction and gives a way to introspect
the queued operations without actually performing them. Each operation
is given a name that is unique and will identify its result in case of
success or failure.

All operations will be executed in the order they were added.

The Ecto.Multi structure should be considered opaque. You can use
%Ecto.Multi{} to pattern match the type, but accessing fields or
directly modifying them is not advised.

Ecto.Multi.to_list/1 returns a canonical representation of the
structure that can be used for introspection.

Changesets

If multi contains operations that accept changesets (like insert/4,
update/4 or delete/4) they will be checked before starting the
transaction. If any changeset has errors, the transaction won't even
be started and the error will be immediately returned.

Note: insert/4, update/4, insert_or_update/4, and delete/4
variants that accept a function are not performing such checks since
the functions are executed after the transaction has started.

Run

Multi allows you to run arbitrary functions as part of your transaction
via run/3 and run/5. This is especially useful when an operation
depends on the value of a previous operation. For this reason, the
function given as a callback to run/3 and run/5 will receive the repo
as the first argument, and all changes performed by the multi so far as a
map for the second argument.

The function given to run must return {:ok, value} or {:error, value}
as its result. Returning an error will abort any further operations
and make the whole multi fail.

Example

Let's look at an example definition and usage. The use case we'll be
looking into is resetting a password. We need to update the account
with proper information, log the request and remove all current sessions:

defmodule PasswordManager do
  alias Ecto.Multi

  def reset(account, params) do
    Multi.new()
    |> Multi.update(:account, Account.password_reset_changeset(account, params))
    |> Multi.insert(:log, Log.password_reset_changeset(account, params))
    |> Multi.delete_all(:sessions, Ecto.assoc(account, :sessions))
  end
end

We can later execute it in the integration layer using Repo:

Repo.transaction(PasswordManager.reset(account, params))

By pattern matching on the result we can differentiate different conditions:

case result do
  {:ok, %{account: account, log: log, sessions: sessions}} ->
    # Operation was successful, we can access results (exactly the same
    # we would get from running corresponding Repo functions) under keys
    # we used for naming the operations.
  {:error, failed_operation, failed_value, changes_so_far} ->
    # One of the operations failed. We can access the operation's failure
    # value (like changeset for operations on changesets) to prepare a
    # proper response. We also get access to the results of any operations
    # that succeeded before the indicated operation failed. However, any
    # successful operations would have been rolled back.
end

We can also easily unit test our transaction without actually running it.
Since changesets can use in-memory-data, we can use an account that is
constructed in memory as well (without persisting it to the database):

test "dry run password reset" do
  account = %Account{password: "letmein"}
  multi = PasswordManager.reset(account, params)

  assert [
    {:account, {:update, account_changeset, []}},
    {:log, {:insert, log_changeset, []}},
    {:sessions, {:delete_all, query, []}}
  ] = Ecto.Multi.to_list(multi)

  # We can introspect changesets and query to see if everything
  # is as expected, for example:
  assert account_changeset.valid?
  assert log_changeset.valid?
  assert inspect(query) == "#Ecto.Query<from a in Session>"
end

The name of each operation does not have to be an atom. This can be particularly
useful when you wish to update a collection of changesets at once, and track their
errors individually:

accounts = [%Account{id: 1}, %Account{id: 2}]

Enum.reduce(accounts, Multi.new(), fn account, multi ->
  Multi.update(
    multi,
    {:account, account.id},
    Account.password_reset_changeset(account, params)
  )
end)

    append(lhs, rhs)
  

    Appends the second multi to the first one.





  
    delete(multi, name, changeset_or_struct_fun, opts \\ [])
  

    Adds a delete operation to the multi.





  
    delete_all(multi, name, queryable, opts \\ [])
  

    Adds a delete_all operation to the multi.





  
    error(multi, name, value)
  

    Causes the multi to fail with the given value.





  
    insert(multi, name, changeset_or_struct_or_fun, opts \\ [])
  

    Adds an insert operation to the multi.





  
    insert_all(multi, name, schema_or_source, entries, opts \\ [])
  

    Adds an insert_all operation to the multi.





  
    insert_or_update(multi, name, changeset, opts \\ [])
  

    Inserts or updates a changeset depending on whether the changeset was persisted or not.





  
    merge(multi, merge)
  

    Merges a multi returned dynamically by an anonymous function.





  
    merge(multi, mod, fun, args)
  

    Merges a multi returned dynamically by calling module and function with args.





  
    new()
  

    Returns an empty Ecto.Multi struct.





  
    prepend(lhs, rhs)
  

    Prepends the second multi to the first one.





  
    run(multi, name, run)
  

    Adds a function to run as part of the multi.





  
    run(multi, name, mod, fun, args)
  

    Adds a function to run as part of the multi.





  
    to_list(multi)
  

    Returns the list of operations stored in multi.





  
    update(multi, name, changeset_or_fun, opts \\ [])
  

    Adds an update operation to the multi.





  
    update_all(multi, name, queryable, updates, opts \\ [])
  

    Adds an update_all operation to the multi.




  

      

      
        
          
            
            Anchor for this section
          
Types        

        

  
    
      
      Link to this type
    
    changes()

      
       
       View Source
     
      
          changes() :: map()

      

  

  
  


  
    
      
      Link to this type
    
    fun(result)

      
       
       View Source
     
      
          fun(result) :: (changes() -> result)

      

  

  
  


  
    
      
      Link to this type
    
    merge()

      
       
       View Source
     
      
          merge() :: (changes() -> t()) | {module(), atom(), [any()]}

      

  

  
  


  
    
      
      Link to this type
    
    name()

      
       
       View Source
     
      
          name() :: any()

      

  

  
  


  
    
      
      Link to this type
    
    run()

      
       
       View Source
     
      
          run() ::
  (Ecto.Repo.t(), changes() -> {:ok | :error, any()})
  | {module(), atom(), [any()]}

      

  

  
  


  
    
      
      Link to this type
    
    t()

      
       
       View Source
     
      
          t() :: %Ecto.Multi{names: names(), operations: operations()}

      

  

  
  

        

      
      
        
          
            
            Anchor for this section
          
Functions        

        

  
    
      
      Link to this function
    
    append(lhs, rhs)

      
       
       View Source
     
      
          append(t(), t()) :: t()

      

  

  
Appends the second multi to the first one.

All names must be unique between both structures.


  
  Example



iex> lhs = Ecto.Multi.new() |> Ecto.Multi.run(:left, fn _, changes -> {:ok, changes} end)
iex> rhs = Ecto.Multi.new() |> Ecto.Multi.run(:right, fn _, changes -> {:error, changes} end)
iex> Ecto.Multi.append(lhs, rhs) |> Ecto.Multi.to_list |> Keyword.keys
[:left, :right]

  


    
  
    
      
      Link to this function
    
    delete(multi, name, changeset_or_struct_fun, opts \\ [])

      
       
       View Source
     
      
          delete(
  t(),
  name(),
  Ecto.Changeset.t()
  | Ecto.Schema.t()
  | fun(Ecto.Changeset.t() | Ecto.Schema.t()),
  Keyword.t()
) :: t()

      

  

  
Adds a delete operation to the multi.

Accepts the same arguments and options as Ecto.Repo.delete/2 does.


  
  Example



post = MyApp.Repo.get!(Post, 1)
Ecto.Multi.new()
|> Ecto.Multi.delete(:delete, post)
|> MyApp.Repo.transaction()

Ecto.Multi.new()
|> Ecto.Multi.run(:post, fn repo, _changes ->
     case repo.get(Post, 1) do
       nil -> {:error, :not_found}
       post -> {:ok, post}
     end
   end)
|> Ecto.Multi.delete(:delete, fn %{post: post} ->
     # Others validations
     post
   end)
|> MyApp.Repo.transaction()

  


    
  
    
      
      Link to this function
    
    delete_all(multi, name, queryable, opts \\ [])

      
       
       View Source
     
      
          delete_all(t(), name(), Ecto.Queryable.t(), Keyword.t()) :: t()

      

  

  
Adds a delete_all operation to the multi.

Accepts the same arguments and options as Ecto.Repo.delete_all/2 does.


  
  Example



queryable = from(p in Post, where: p.id < 5)
Ecto.Multi.new()
|> Ecto.Multi.delete_all(:delete_all, queryable)
|> MyApp.Repo.transaction()

  


  
    
      
      Link to this function
    
    error(multi, name, value)

      
       
       View Source
     
      
          error(t(), name(), error :: term()) :: t()

      

  

  
Causes the multi to fail with the given value.

Running the multi in a transaction will execute
no previous steps and returns the value of the first
error added.

  


    
  
    
      
      Link to this function
    
    insert(multi, name, changeset_or_struct_or_fun, opts \\ [])

      
       
       View Source
     
      
          insert(
  t(),
  name(),
  Ecto.Changeset.t()
  | Ecto.Schema.t()
  | fun(Ecto.Changeset.t() | Ecto.Schema.t()),
  Keyword.t()
) :: t()

      

  

  
Adds an insert operation to the multi.

Accepts the same arguments and options as Ecto.Repo.insert/2 does.


  
  Example



Ecto.Multi.new()
|> Ecto.Multi.insert(:insert, %Post{title: "first"})
|> MyApp.Repo.transaction()

Ecto.Multi.new()
|> Ecto.Multi.insert(:post, %Post{title: "first"})
|> Ecto.Multi.insert(:comment, fn %{post: post} ->
     Ecto.build_assoc(post, :comments)
   end)
|> MyApp.Repo.transaction()

  


    
  
    
      
      Link to this function
    
    insert_all(multi, name, schema_or_source, entries, opts \\ [])

      
       
       View Source
     
      
          insert_all(
  t(),
  name(),
  schema_or_source(),
  [map() | Keyword.t()],
  Keyword.t()
) :: t()

      

  

  
Adds an insert_all operation to the multi.

Accepts the same arguments and options as Ecto.Repo.insert_all/3 does.


  
  Example



posts = [%{title: "My first post"}, %{title: "My second post"}]
Ecto.Multi.new()
|> Ecto.Multi.insert_all(:insert_all, Post, posts)
|> MyApp.Repo.transaction()

  


    
  
    
      
      Link to this function
    
    insert_or_update(multi, name, changeset, opts \\ [])

      
       
       View Source
     
      
          insert_or_update(
  t(),
  name(),
  Ecto.Changeset.t() | fun(Ecto.Changeset.t()),
  Keyword.t()
) :: t()

      

  

  
Inserts or updates a changeset depending on whether the changeset was persisted or not.

Accepts the same arguments and options as Ecto.Repo.insert_or_update/2 does.


  
  Example



changeset = Post.changeset(%Post{}, %{title: "New title"})
Ecto.Multi.new()
|> Ecto.Multi.insert_or_update(:insert_or_update, changeset)
|> MyApp.Repo.transaction()

Ecto.Multi.new()
|> Ecto.Multi.run(:post, fn repo, _changes ->
     {:ok, repo.get(Post, 1) || %Post{}}
   end)
|> Ecto.Multi.insert_or_update(:update, fn %{post: post} ->
     Ecto.Changeset.change(post, title: "New title")
   end)
|> MyApp.Repo.transaction()

  


  
    
      
      Link to this function
    
    merge(multi, merge)

      
       
       View Source
     
      
          merge(t(), (changes() -> t())) :: t()

      

  

  
Merges a multi returned dynamically by an anonymous function.

This function is useful when the multi to be merged requires information
from the original multi. Hence the second argument is an anonymous function
that receives the multi changes so far. The anonymous function must return
another multi.

If you would prefer to simply merge two multis together, see append/2 or
prepend/2.

Duplicated operations are not allowed.


  
  Example



multi =
  Ecto.Multi.new()
  |> Ecto.Multi.insert(:post, %Post{title: "first"})

multi
|> Ecto.Multi.merge(fn %{post: post} ->
  Ecto.Multi.new()
  |> Ecto.Multi.insert(:comment, Ecto.build_assoc(post, :comments))
end)
|> MyApp.Repo.transaction()

  


  
    
      
      Link to this function
    
    merge(multi, mod, fun, args)

      
       
       View Source
     
      
          merge(t(), module(), function, args) :: t()
when function: atom(), args: [any()]

      

  

  
Merges a multi returned dynamically by calling module and function with args.

Similar to merge/2, but allows to pass module name, function and arguments.
The function should return an Ecto.Multi, and receives changes so far
as the first argument (prepended to those passed in the call to the function).

Duplicated operations are not allowed.

  


  
    
      
      Link to this function
    
    new()

      
       
       View Source
     
      
          new() :: t()

      

  

  
Returns an empty Ecto.Multi struct.


  
  Example



iex> Ecto.Multi.new() |> Ecto.Multi.to_list()
[]

  


  
    
      
      Link to this function
    
    prepend(lhs, rhs)

      
       
       View Source
     
      
          prepend(t(), t()) :: t()

      

  

  
Prepends the second multi to the first one.

All names must be unique between both structures.


  
  Example



iex> lhs = Ecto.Multi.new() |> Ecto.Multi.run(:left, fn _, changes -> {:ok, changes} end)
iex> rhs = Ecto.Multi.new() |> Ecto.Multi.run(:right, fn _, changes -> {:error, changes} end)
iex> Ecto.Multi.prepend(lhs, rhs) |> Ecto.Multi.to_list |> Keyword.keys
[:right, :left]

  


  
    
      
      Link to this function
    
    run(multi, name, run)

      
       
       View Source
     
      
          run(t(), name(), run()) :: t()

      

  

  
Adds a function to run as part of the multi.

The function should return either {:ok, value} or {:error, value},
and receives the repo as the first argument, and the changes so far
as the second argument.


  
  Example



Ecto.Multi.run(multi, :write, fn _repo, %{image: image} ->
  with :ok <- File.write(image.name, image.contents) do
    {:ok, nil}
  end
end)

  


  
    
      
      Link to this function
    
    run(multi, name, mod, fun, args)

      
       
       View Source
     
      
          run(t(), name(), module(), function, args) :: t()
when function: atom(), args: [any()]

      

  

  
Adds a function to run as part of the multi.

Similar to run/3, but allows to pass module name, function and arguments.
The function should return either {:ok, value} or {:error, value}, and
receives the repo as the first argument, and the changes so far as the
second argument (prepended to those passed in the call to the function).

  


  
    
      
      Link to this function
    
    to_list(multi)

      
       
       View Source
     
      
          to_list(t()) :: [{name(), term()}]

      

  

  
Returns the list of operations stored in multi.

Always use this function when you need to access the operations you
have defined in Ecto.Multi. Inspecting the Ecto.Multi struct internals
directly is discouraged.

  


    
  
    
      
      Link to this function
    
    update(multi, name, changeset_or_fun, opts \\ [])

      
       
       View Source
     
      
          update(
  t(),
  name(),
  Ecto.Changeset.t() | fun(Ecto.Changeset.t()),
  Keyword.t()
) :: t()

      

  

  
Adds an update operation to the multi.

Accepts the same arguments and options as Ecto.Repo.update/2 does.


  
  Example



post = MyApp.Repo.get!(Post, 1)
changeset = Ecto.Changeset.change(post, title: "New title")
Ecto.Multi.new()
|> Ecto.Multi.update(:update, changeset)
|> MyApp.Repo.transaction()

Ecto.Multi.new()
|> Ecto.Multi.insert(:post, %Post{title: "first"})
|> Ecto.Multi.update(:fun, fn %{post: post} ->
     Ecto.Changeset.change(post, title: "New title")
   end)
|> MyApp.Repo.transaction()

  


    
  
    
      
      Link to this function
    
    update_all(multi, name, queryable, updates, opts \\ [])

      
       
       View Source
     
      
          update_all(t(), name(), Ecto.Queryable.t(), Keyword.t(), Keyword.t()) ::
  t()

      

  

  
Adds an update_all operation to the multi.

Accepts the same arguments and options as Ecto.Repo.update_all/3 does.


  
  Example



Ecto.Multi.new()
|> Ecto.Multi.update_all(:update_all, Post, set: [title: "New title"])
|> MyApp.Repo.transaction()

  

        

      
  

  
    
    Ecto.Query - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Query    



      
Provides the Query DSL.

Queries are used to retrieve and manipulate data from a repository
(see Ecto.Repo). Ecto queries come in two flavors: keyword-based
and macro-based. Most examples will use the keyword-based syntax,
the macro one will be explored in later sections.

Let's see a sample query:

# Imports only from/2 of Ecto.Query
import Ecto.Query, only: [from: 2]

# Create a query
query = from u in "users",
          where: u.age > 18,
          select: u.name

# Send the query to the repository
Repo.all(query)

In the example above, we are directly querying the "users" table
from the database.

Query expressions

Ecto allows a limited set of expressions inside queries. In the
query below, for example, we use u.age to access a field, the
> comparison operator and the literal 0:

query = from u in "users", where: u.age > 0, select: u.name

You can find the full list of operations in Ecto.Query.API.
Besides the operations listed there, the following literals are
supported in queries:


	Integers: 1, 2, 3


	Floats: 1.0, 2.0, 3.0


	Booleans: true, false


	Binaries: <<1, 2, 3>>


	Strings: "foo bar", ~s(this is a string)


	Arrays: [1, 2, 3], ~w(interpolate words)




All other types and dynamic values must be passed as a parameter using
interpolation as explained below.

Interpolation and casting

External values and Elixir expressions can be injected into a query
expression with ^:

def with_minimum(age, height_ft) do
  from u in "users",
    where: u.age > ^age and u.height > ^(height_ft * 3.28),
    select: u.name
end

with_minimum(18, 5.0)

When interpolating values, you may want to explicitly tell Ecto
what is the expected type of the value being interpolated:

age = "18"
Repo.all(from u in "users",
          where: u.age > type(^age, :integer),
          select: u.name)

In the example above, Ecto will cast the age to type integer. When
a value cannot be cast, Ecto.Query.CastError is raised.

To avoid the repetition of always specifying the types, you may define
an Ecto.Schema. In such cases, Ecto will analyze your queries and
automatically cast the interpolated "age" when compared to the u.age
field, as long as the age field is defined with type :integer in
your schema:

age = "18"
Repo.all(from u in User, where: u.age > ^age, select: u.name)

Another advantage of using schemas is that we no longer need to specify
the select option in queries, as by default Ecto will retrieve all
fields specified in the schema:

age = "18"
Repo.all(from u in User, where: u.age > ^age)

For this reason, we will use schemas on the remaining examples but
remember Ecto does not require them in order to write queries.

nil comparison

nil comparison in filters, such as where and having, is forbidden
and it will raise an error:

# Raises if age is nil
from u in User, where: u.age == ^age

This is done as a security measure to avoid attacks that attempt
to traverse entries with nil columns. To check that value is nil,
use is_nil/1 instead:

from u in User, where: is_nil(u.age)

Composition

Ecto queries are composable. For example, the query above can
actually be defined in two parts:

# Create a query
query = from u in User, where: u.age > 18

# Extend the query
query = from u in query, select: u.name

Composing queries uses the same syntax as creating a query.
The difference is that, instead of passing a schema like User
on the right side of in, we passed the query itself.

Any value can be used on the right-side of in as long as it implements
the Ecto.Queryable protocol. For now, we know the protocol is
implemented for both atoms (like User) and strings (like "users").

In any case, regardless if a schema has been given or not, Ecto
queries are always composable thanks to its binding system.

Positional bindings

On the left side of in we specify the query bindings.  This is
done inside from and join clauses.  In the query below u is a
binding and u.age is a field access using this binding.

query = from u in User, where: u.age > 18

Bindings are not exposed from the query.  When composing queries, you
must specify bindings again for each refinement query.  For example,
to further narrow down the above query, we again need to tell Ecto what
bindings to expect:

query = from u in query, select: u.city

Bindings in Ecto are positional, and the names do not have to be
consistent between input and refinement queries. For example, the
query above could also be written as:

query = from q in query, select: q.city

It would make no difference to Ecto. This is important because
it allows developers to compose queries without caring about
the bindings used in the initial query.

When using joins, the bindings should be matched in the order they
are specified:

# Create a query
query = from p in Post,
          join: c in Comment, on: c.post_id == p.id

# Extend the query
query = from [p, c] in query,
          select: {p.title, c.body}

You are not required to specify all bindings when composing.
For example, if we would like to order the results above by
post insertion date, we could further extend it as:

query = from q in query, order_by: q.inserted_at

The example above will work if the input query has 1 or 10
bindings. As long as the number of bindings is less than the
number of from + joins, Ecto will match only what you have
specified. The first binding always matches the source given
in from.

Similarly, if you are interested only on the last binding
(or the last bindings) in a query, you can use ... to
specify "all bindings before" and match on the last one.

For instance, imagine you wrote:

posts_with_comments =
  from p in query, join: c in Comment, on: c.post_id == p.id

And now we want to make sure to return both the post title
and the comment body. Although we may not know how many
bindings there are in the query, we are sure posts is the
first binding and comments are the last one, so we can write:

from [p, ..., c] in posts_with_comments, select: {p.title, c.body}

In other words, ... will include all the binding between the
first and the last, which may be no binding at all, one or many.

Named bindings

Another option for flexibly building queries with joins are
named bindings. Coming back to the previous example, provided
we bind a join to a concrete name:

posts_with_comments =
  from p in query,
    join: c in Comment, as: :comment, on: c.post_id == p.id

We can refer to it by that name using a following form of
bindings list:

from [p, comment: c] in posts_with_comments, select: {p.title, c.body}

This approach lets us not worry about keeping track of the position
of the bindings when composing the query.

What's more, a name can be assigned to the first binding as well:

from p in Post, as: :post

Only atoms are accepted for binding names. Named binding references
are expected to be placed in the tail position of the bindings list.

Bindingless operations

Although bindings are extremely useful when working with joins,
they are not necessary when the query has only the from clause.
For such cases, Ecto supports a way for building queries
without specifying the binding:

from Post,
  where: [category: "fresh and new"],
  order_by: [desc: :published_at],
  select: [:id, :title, :body]

The query above will select all posts with category "fresh and new",
order by the most recently published, and return Post structs with
only the id, title and body fields set. It is equivalent to:

from p in Post,
  where: p.category == "fresh and new",
  order_by: [desc: p.published_at],
  select: struct(p, [:id, :title, :body])

One advantage of bindingless queries is that they are data-driven
and therefore useful for dynamically building queries. For example,
the query above could also be written as:

where = [category: "fresh and new"]
order_by = [desc: :published_at]
select = [:id, :title, :body]
from Post, where: ^where, order_by: ^order_by, select: ^select

This feature is very useful when queries need to be built based
on some user input, like web search forms, CLIs and so on.

Fragments

If you need an escape hatch, Ecto provides fragments
(see Ecto.Query.API.fragment/1) to inject SQL (and non-SQL)
fragments into queries.

For example, to get all posts while running the "lower(?)"
function in the database where p.title is interpolated
in place of ?, one can write:

from p in Post,
  where: is_nil(p.published_at) and
         fragment("lower(?)", p.title) == ^title

Also, most adapters provide direct APIs for queries, like
Ecto.Adapters.SQL.query/4, allowing developers to
completely bypass Ecto queries.

Macro API

In all examples so far we have used the keywords query syntax to
create a query:

import Ecto.Query
from u in "users", where: u.age > 18, select: u.name

Due to the prevalence of the pipe operator in Elixir, Ecto also supports
a pipe-based syntax:

"users"
|> where([u], u.age > 18)
|> select([u], u.name)

The keyword-based and pipe-based examples are equivalent. The downside
of using macros is that the binding must be specified for every operation.
However, since keyword-based and pipe-based examples are equivalent, the
bindingless syntax also works for macros:

"users"
|> where([u], u.age > 18)
|> select([:name])

Such allows developers to write queries using bindings only in more
complex query expressions.

This module documents each of those macros, providing examples in
both the keywords query and pipe expression formats.

Query Prefix

It is possible to set a prefix for the queries. For Postgres users,
this will specify the schema where the table is located, while for
MySQL users this will specify the database where the table is
located.  When no prefix is set, Postgres queries are assumed to be
in the public schema, while MySQL queries are assumed to be in the
database set in the config for the repo.

The query prefix may be set either for the whole query or on each
individual from and join expression. If a prefix is not given
to a from or a join, the prefix of the schema given to the from
or join is used. The query prefix is used only if none of the above
are declared.

Let's see some examples. To see the query prefix globally, the simplest
mechanism is to pass an option to the repository operation:

results = Repo.all(query, prefix: "accounts")

You may also set the prefix for the whole query by setting the prefix field:

results =
  query # May be User or an Ecto.Query itself
  |> Ecto.Queryable.to_query
  |> Map.put(:prefix, "accounts")
  |> Repo.all()

Setting the prefix in the query changes the default prefix of all from
and join expressions. You can override the query prefix by either setting
the @schema_prefix in your schema definitions or by passing the prefix
option:

from u in User,
  prefix: "accounts",
  join: p in assoc(u, :posts),
  prefix: "public"

Overall, here is the prefix lookup precedence:


	The :prefix option given to from/join has the highest precedence


	Then it falls back to the @schema_prefix attribute declared in the schema
given to from/join


	Then it falls back to the query prefix




The prefixes set in the query will be preserved when loading data.

      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Types
    


  
    dynamic()
  




  
    t()
  



  

  
    
      Functions
    


  
    distinct(query, binding \\ [], expr)
  

    A distinct query expression.





  
    dynamic(binding \\ [], expr)
  

    Builds a dynamic query expression.





  
    except(query, other_query)
  

    An except (set difference) query expression.





  
    except_all(query, other_query)
  

    An except (set difference) query expression.





  
    exclude(query, field)
  

    Resets a previously set field on a query.





  
    first(queryable, order_by \\ nil)
  

    Restricts the query to return the first result ordered by primary key.





  
    from(expr, kw \\ [])
  

    Creates a query.





  
    group_by(query, binding \\ [], expr)
  

    A group by query expression.





  
    has_named_binding?(queryable, key)
  

    Returns true if query has binding with a given name, otherwise false.





  
    having(query, binding \\ [], expr)
  

    An AND having query expression.





  
    intersect(query, other_query)
  

    An intersect query expression.





  
    intersect_all(query, other_query)
  

    An intersect query expression.





  
    join(query, qual, binding \\ [], expr, opts \\ [])
  

    A join query expression.





  
    last(queryable, order_by \\ nil)
  

    Restricts the query to return the last result ordered by primary key.





  
    limit(query, binding \\ [], expr)
  

    A limit query expression.





  
    lock(query, expr)
  

    A lock query expression.





  
    offset(query, binding \\ [], expr)
  

    An offset query expression.





  
    or_having(query, binding \\ [], expr)
  

    An OR having query expression.





  
    or_where(query, binding \\ [], expr)
  

    An OR where query expression.





  
    order_by(query, binding \\ [], expr)
  

    An order by query expression.





  
    preload(query, bindings \\ [], expr)
  

    Preloads the associations into the result set.





  
    recursive_ctes(query, value)
  

    Enables or disables recursive mode for CTEs.





  
    reverse_order(query)
  

    Reverses the ordering of the query.





  
    select(query, binding \\ [], expr)
  

    A select query expression.





  
    select_merge(query, binding \\ [], expr)
  

    Mergeable select query expression.





  
    subquery(query, opts \\ [])
  

    Converts a query into a subquery.





  
    union(query, other_query)
  

    A union query expression.





  
    union_all(query, other_query)
  

    A union all query expression.





  
    update(query, binding \\ [], expr)
  

    An update query expression.





  
    where(query, binding \\ [], expr)
  

    An AND where query expression.





  
    windows(query, binding \\ [], expr)
  

    Defines windows which can be used with Ecto.Query.WindowAPI.





  
    with_cte(query, name, list)
  

    A common table expression (CTE) also known as WITH expression.

Types        

        

  
    
      
      Link to this opaque
    
    dynamic()

      
       
       View Source
     
      (opaque)
      
          dynamic()

      

  

  
  


  
    
      
      Link to this type
    
    t()

      
       
       View Source
     
      
          t() :: %Ecto.Query{
  aliases: term(),
  assocs: term(),
  combinations: term(),
  distinct: term(),
  from: term(),
  group_bys: term(),
  havings: term(),
  joins: term(),
  limit: term(),
  lock: term(),
  offset: term(),
  order_bys: term(),
  prefix: term(),
  preloads: term(),
  select: term(),
  sources: term(),
  updates: term(),
  wheres: term(),
  windows: term(),
  with_ctes: term()
}

      

  

  
  

        

      
      
        
          
            
            Anchor for this section
          
Functions        

        

    
  
    
      
      Link to this macro
    
    distinct(query, binding \\ [], expr)

      
       
       View Source
     
      (macro)
  

  
A distinct query expression.

When true, only keeps distinct values from the resulting
select expression.

If supported by your database, you can also pass query expressions
to distinct and it will generate a query with DISTINCT ON. In such
cases, distinct accepts exactly the same expressions as order_by
and any distinct expression will be automatically prepended to the
order_by expressions in case there is any order_by expression.


  
  Keywords examples



# Returns the list of different categories in the Post schema
from(p in Post, distinct: true, select: p.category)

# If your database supports DISTINCT ON(),
# you can pass expressions to distinct too
from(p in Post,
   distinct: p.category,
   order_by: [p.date])

# The DISTINCT ON() also supports ordering similar to ORDER BY.
from(p in Post,
   distinct: [desc: p.category],
   order_by: [p.date])

# Using atoms
from(p in Post, distinct: :category, order_by: :date)


  
  Expressions example



Post
|> distinct(true)
|> order_by([p], [p.category, p.author])

  


    
  
    
      
      Link to this macro
    
    dynamic(binding \\ [], expr)

      
       
       View Source
     
      (macro)
  

  
Builds a dynamic query expression.

Dynamic query expressions allows developers to build queries
expression bit by bit so they are later interpolated in a query.


  
  Examples



For example, imagine you have a set of conditions you want to
build your query on:

conditions = false

conditions =
  if params["is_public"] do
    dynamic([p], p.is_public or ^conditions)
  else
    conditions
  end

conditions =
  if params["allow_reviewers"] do
    dynamic([p, a], a.reviewer == true or ^conditions)
  else
    conditions
  end

from query, where: ^conditions

In the example above, we were able to build the query expressions
bit by bit, using different bindings, and later interpolate it all
at once inside the query.

A dynamic expression can always be interpolated inside another dynamic
expression and into the constructs described below.


  
  where, having and a join's `on'



The dynamic macro can be interpolated at the root of a where,
having or a join's on.

For example, assuming the conditions variable defined in the
previous section, the following is forbidden because it is not
at the root of a where:

from q in query, where: q.some_condition and ^conditions

Fortunately that's easily solvable by simply rewriting it to:

conditions = dynamic([q], q.some_condition and ^conditions)
from query, where: ^conditions


  
  order_by



Dynamics can be interpolated inside keyword lists at the root of
order_by. For example, you can write:

order_by = [
  asc: :some_field,
  desc: dynamic([p], fragment("?>>?", p.another_field, "json_key"))
]

from query, order_by: ^order_by

Dynamics are also supported in order_by/2 clauses inside windows/2.

As with where and friends, it is not possible to pass dynamics
outside of a root. For example, this won't work:

from query, order_by: [asc: ^dynamic(...)]

But this will:

from query, order_by: ^[asc: dynamic(...)]


  
  group_by



Dynamics can be interpolated inside keyword lists at the root of
group_by. For example, you can write:

group_by = [
  :some_field,
  dynamic([p], fragment("?>>?", p.another_field, "json_key"))
]

from query, group_by: ^group_by

Dynamics are also supported in partition_by/2 clauses inside windows/2.

As with where and friends, it is not possible to pass dynamics
outside of a root. For example, this won't work:

from query, group_by: [:some_field, ^dynamic(...)]

But this will:

from query, order_by: ^[:some_field, dynamic(...)]


  
  Updates



Dynamic is also supported inside updates, for example:

updates = [
  set: [average: dynamic([p], p.sum / p.count)]
]

from query, update: ^updates

  


  
    
      
      Link to this macro
    
    except(query, other_query)

      
       
       View Source
     
      (macro)
  

  
An except (set difference) query expression.

Takes the difference of the result sets of multiple queries. The
select of each query must be exactly the same, with the same
types in the same order.

Except expression returns only unique rows as if each query returned
distinct results. This may cause performance penalty. If you need
just to take the difference of multiple result sets without
removing duplicate rows consider using except_all/2.

Note that the operations order_by, limit and offset of the
current query apply to the result of the set difference.


  
  Keywords example



supplier_query = from s in Supplier, select: s.city
from c in Customer, select: c.city, except: ^supplier_query


  
  Expressions example



supplier_query = Supplier |> select([s], s.city)
Customer |> select([c], c.city) |> except(^supplier_query)

  


  
    
      
      Link to this macro
    
    except_all(query, other_query)

      
       
       View Source
     
      (macro)
  

  
An except (set difference) query expression.

Takes the difference of the result sets of multiple queries. The
select of each query must be exactly the same, with the same
types in the same order.

Note that the operations order_by, limit and offset of the
current query apply to the result of the set difference.


  
  Keywords example



supplier_query = from s in Supplier, select: s.city
from c in Customer, select: c.city, except_all: ^supplier_query


  
  Expressions example



supplier_query = Supplier |> select([s], s.city)
Customer |> select([c], c.city) |> except_all(^supplier_query)

  


  
    
      
      Link to this function
    
    exclude(query, field)

      
       
       View Source
     
  

  
Resets a previously set field on a query.

It can reset many fields except the query source (from). When excluding
a :join, it will remove all types of joins. If you prefer to remove a
single type of join, please see paragraph below.


  
  Examples



Ecto.Query.exclude(query, :join)
Ecto.Query.exclude(query, :where)
Ecto.Query.exclude(query, :order_by)
Ecto.Query.exclude(query, :group_by)
Ecto.Query.exclude(query, :having)
Ecto.Query.exclude(query, :distinct)
Ecto.Query.exclude(query, :select)
Ecto.Query.exclude(query, :combinations)
Ecto.Query.exclude(query, :with_ctes)
Ecto.Query.exclude(query, :limit)
Ecto.Query.exclude(query, :offset)
Ecto.Query.exclude(query, :lock)
Ecto.Query.exclude(query, :preload)

You can also remove specific joins as well such as left_join and
inner_join:

Ecto.Query.exclude(query, :inner_join)
Ecto.Query.exclude(query, :cross_join)
Ecto.Query.exclude(query, :left_join)
Ecto.Query.exclude(query, :right_join)
Ecto.Query.exclude(query, :full_join)
Ecto.Query.exclude(query, :inner_lateral_join)
Ecto.Query.exclude(query, :left_lateral_join)

However, keep in mind that if a join is removed and its bindings
were referenced elsewhere, the bindings won't be removed, leading
to a query that won't compile.

  


    
  
    
      
      Link to this function
    
    first(queryable, order_by \\ nil)

      
       
       View Source
     
  

  
Restricts the query to return the first result ordered by primary key.

The query will be automatically ordered by the primary key
unless order_by is given or order_by is set in the query.
Limit is always set to 1.


  
  Examples



Post |> first |> Repo.one
query |> first(:inserted_at) |> Repo.one

  


    
  
    
      
      Link to this macro
    
    from(expr, kw \\ [])

      
       
       View Source
     
      (macro)
  

  
Creates a query.

It can either be a keyword query or a query expression.

If it is a keyword query the first argument must be
either an in expression, or a value that implements
the Ecto.Queryable protocol. If the query needs a
reference to the data source in any other part of the
expression, then an in must be used to create a reference
variable. The second argument should be a keyword query
where the keys are expression types and the values are
expressions.

If it is a query expression the first argument must be
a value that implements the Ecto.Queryable protocol
and the second argument the expression.


  
  Keywords example



from(c in City, select: c)


  
  Expressions example



City |> select([c], c)


  
  Examples



def paginate(query, page, size) do
  from query,
    limit: ^size,
    offset: ^((page-1) * size)
end

The example above does not use in because limit and offset
do not require a reference to the data source. However, extending
the query with a where expression would require the use of in:

def published(query) do
  from p in query, where: not(is_nil(p.published_at))
end

Notice we have created a p variable to reference the query's
original data source. This assumes that the original query
only had one source. When the given query has more than one source,
positional or named bindings may be used to access the additional sources.

def published_multi(query) do
  from [p,o] in query,
  where: not(is_nil(p.published_at)) and not(is_nil(o.published_at))
end

Note the variables p and o can be named whatever you like
as they have no importance in the query sent to the database.

  


    
  
    
      
      Link to this macro
    
    group_by(query, binding \\ [], expr)

      
       
       View Source
     
      (macro)
  

  
A group by query expression.

Groups together rows from the schema that have the same values in the given
fields. Using group_by "groups" the query giving it different semantics
in the select expression. If a query is grouped, only fields that were
referenced in the group_by can be used in the select or if the field
is given as an argument to an aggregate function.

group_by also accepts a list of atoms where each atom refers to
a field in source. For more complicated queries you can access fields
directly instead of atoms.


  
  Keywords examples



# Returns the number of posts in each category
from(p in Post,
  group_by: p.category,
  select: {p.category, count(p.id)})

# Using atoms
from(p in Post, group_by: :category, select: {p.category, count(p.id)})

# Using direct fields access
from(p in Post,
  join: c in assoc(p, :category)
  group_by: [p.id, c.name])


  
  Expressions example



Post |> group_by([p], p.category) |> select([p], count(p.id))

  


  
    
      
      Link to this function
    
    has_named_binding?(queryable, key)

      
       
       View Source
     
  

  
Returns true if query has binding with a given name, otherwise false.

For more information on named bindings see "Named bindings" in this module doc.

  


    
  
    
      
      Link to this macro
    
    having(query, binding \\ [], expr)

      
       
       View Source
     
      (macro)
  

  
An AND having query expression.

Like where, having filters rows from the schema, but after the grouping is
performed giving it the same semantics as select for a grouped query
(see group_by/3). having groups the query even if the query has no
group_by expression.


  
  Keywords example



# Returns the number of posts in each category where the
# average number of comments is above ten
from(p in Post,
  group_by: p.category,
  having: avg(p.num_comments) > 10,
  select: {p.category, count(p.id)})


  
  Expressions example



Post
|> group_by([p], p.category)
|> having([p], avg(p.num_comments) > 10)
|> select([p], count(p.id))

  


  
    
      
      Link to this macro
    
    intersect(query, other_query)

      
       
       View Source
     
      (macro)
  

  
An intersect query expression.

Takes the overlap of the result sets of multiple queries. The
select of each query must be exactly the same, with the same
types in the same order.

Intersect expression returns only unique rows as if each query returned
distinct results. This may cause performance penalty. If you need
just to take the intersection of multiple result sets without
removing duplicate rows consider using intersect_all/2.

Note that the operations order_by, limit and offset of the
current query apply to the result of the set difference.


  
  Keywords example



supplier_query = from s in Supplier, select: s.city
from c in Customer, select: c.city, intersect: ^supplier_query


  
  Expressions example



supplier_query = Supplier |> select([s], s.city)
Customer |> select([c], c.city) |> intersect(^supplier_query)

  


  
    
      
      Link to this macro
    
    intersect_all(query, other_query)

      
       
       View Source
     
      (macro)
  

  
An intersect query expression.

Takes the overlap of the result sets of multiple queries. The
select of each query must be exactly the same, with the same
types in the same order.

Note that the operations order_by, limit and offset of the
current query apply to the result of the set difference.


  
  Keywords example



supplier_query = from s in Supplier, select: s.city
from c in Customer, select: c.city, intersect_all: ^supplier_query


  
  Expressions example



supplier_query = Supplier |> select([s], s.city)
Customer |> select([c], c.city) |> intersect_all(^supplier_query)

  


    
    
  
    
      
      Link to this macro
    
    join(query, qual, binding \\ [], expr, opts \\ [])

      
       
       View Source
     
      (macro)
  

  
A join query expression.

Receives a source that is to be joined to the query and a condition for
the join. The join condition can be any expression that evaluates
to a boolean value. The join is by default an inner join, the qualifier
can be changed by giving the atoms: :inner, :left, :right, :cross,
:full, :inner_lateral or :left_lateral. For a keyword query the :join
keyword can be changed to: :inner_join, :left_join, :right_join,
:cross_join, :full_join, :inner_lateral_join or :left_lateral_join.

Currently it is possible to join on:


	an Ecto.Schema, such as p in Post


	an interpolated Ecto query with zero or more where clauses,
such as c in ^(from "posts", where: [public: true])


	an association, such as c in assoc(post, :comments)


	a subquery, such as c in subquery(another_query)


	a query fragment, such as c in fragment("SOME COMPLEX QUERY"),
see "Joining with fragments" below.





  
  Options



Each join accepts the following options:


	:on - a query expression or keyword list to filter the join


	:as - a named binding for the join


	:prefix - the prefix to be used for the join when issuing a database query


	:hints - a string or a list of strings to be used as database hints




In the keyword query syntax, those options must be given immediately
after the join. In the expression syntax, the options are given as
the fifth argument.


  
  Keywords examples



from c in Comment,
  join: p in Post,
  on: p.id == c.post_id,
  select: {p.title, c.text}

from p in Post,
  left_join: c in assoc(p, :comments),
  select: {p, c}

Keywords can also be given or interpolated as part of on:

from c in Comment,
  join: p in Post,
  on: [id: c.post_id],
  select: {p.title, c.text}

Any key in on will apply to the currently joined expression.

It is also possible to interpolate an Ecto query on the right side
of in. For example, the query above can also be written as:

posts = Post
from c in Comment,
  join: p in ^posts,
  on: [id: c.post_id],
  select: {p.title, c.text}

The above is specially useful to dynamically join on existing
queries, for example, to dynamically choose a source, or by
choosing between public posts or posts that have been recently
published:

posts =
  if params["drafts"] do
    from p in Post, where: [drafts: true]
  else
    from p in Post, where: [public: true]
  end

from c in Comment,
  join: p in ^posts, on: [id: c.post_id],
  select: {p.title, c.text}

Only simple queries with where expressions can be interpolated
in join.


  
  Expressions examples



Comment
|> join(:inner, [c], p in Post, on: c.post_id == p.id)
|> select([c, p], {p.title, c.text})

Post
|> join(:left, [p], c in assoc(p, :comments))
|> select([p, c], {p, c})

Post
|> join(:left, [p], c in Comment, on: c.post_id == p.id and c.is_visible == true)
|> select([p, c], {p, c})


  
  Joining with fragments



When you need to join on a complex query, Ecto supports fragments in joins:

Comment
|> join(:inner, [c], p in fragment("SOME COMPLEX QUERY", c.id, ^some_param))

Although using fragments in joins is discouraged in favor of Ecto
Query syntax, they are necessary when writing lateral joins as
lateral joins require a subquery that refer to previous bindings:

Game
|> join(:inner_lateral, [g], gs in fragment("SELECT * FROM games_sold AS gs WHERE gs.game_id = ? ORDER BY gs.sold_on LIMIT 2", g.id))
|> select([g, gs], {g.name, gs.sold_on})

Note that the join does not automatically wrap the fragment in
parentheses, since some expressions require parens and others
require no parens. Therefore, in cases such as common table
expressions, you will have to explicitly wrap the fragment content
in parens.


  
  Hints



from and join also support index hints, as found in databases such as
MySQL and
MSSQL.

For example, a developer using MySQL may write:

from p in Post,
  join: c in Comment,
  hints: ["USE INDEX FOO", "USE INDEX BAR"],
  where: p.id == c.post_id,
  select: c

Keep in mind you want to use hints rarely, so don't forget to read the database
disclaimers about such functionality.

  


    
  
    
      
      Link to this function
    
    last(queryable, order_by \\ nil)

      
       
       View Source
     
  

  
Restricts the query to return the last result ordered by primary key.

The query ordering will be automatically reversed, with ASC
columns becoming DESC columns (and vice-versa) and limit is set
to 1. If there is no ordering, the query will be automatically
ordered decreasingly by primary key.


  
  Examples



Post |> last |> Repo.one
query |> last(:inserted_at) |> Repo.one

  


    
  
    
      
      Link to this macro
    
    limit(query, binding \\ [], expr)

      
       
       View Source
     
      (macro)
  

  
A limit query expression.

Limits the number of rows returned from the result. Can be any expression but
has to evaluate to an integer value and it can't include any field.

If limit is given twice, it overrides the previous value.


  
  Keywords example



from(u in User, where: u.id == ^current_user, limit: 1)


  
  Expressions example



User |> where([u], u.id == ^current_user) |> limit(1)

  


  
    
      
      Link to this macro
    
    lock(query, expr)

      
       
       View Source
     
      (macro)
  

  
A lock query expression.

Provides support for row-level pessimistic locking using
SELECT ... FOR UPDATE or other, database-specific, locking clauses.
expr can be any expression but has to evaluate to a boolean value or to a
string and it can't include any fields.

If lock is used more than once, the last one used takes precedence.

Ecto also supports optimistic
locking but not
through queries. For more information on optimistic locking, have a look at
the Ecto.Changeset.optimistic_lock/3 function


  
  Keywords example



from(u in User, where: u.id == ^current_user, lock: "FOR SHARE NOWAIT")


  
  Expressions example



User |> where(u.id == ^current_user) |> lock("FOR SHARE NOWAIT")

  


    
  
    
      
      Link to this macro
    
    offset(query, binding \\ [], expr)

      
       
       View Source
     
      (macro)
  

  
An offset query expression.

Offsets the number of rows selected from the result. Can be any expression
but it must evaluate to an integer value and it can't include any field.

If offset is given twice, it overrides the previous value.


  
  Keywords example



# Get all posts on page 4
from(p in Post, limit: 10, offset: 30)


  
  Expressions example



Post |> limit(10) |> offset(30)

  


    
  
    
      
      Link to this macro
    
    or_having(query, binding \\ [], expr)

      
       
       View Source
     
      (macro)
  

  
An OR having query expression.

Like having but combines with the previous expression by using
OR. or_having behaves for having the same way or_where
behaves for where.


  
  Keywords example



# Augment a previous group_by with a having condition.
from(p in query, or_having: avg(p.num_comments) > 10)


  
  Expressions example



# Augment a previous group_by with a having condition.
Post |> or_having([p], avg(p.num_comments) > 10)

  


    
  
    
      
      Link to this macro
    
    or_where(query, binding \\ [], expr)

      
       
       View Source
     
      (macro)
  

  
An OR where query expression.

Behaves exactly the same as where except it combines with any previous
expression by using an OR. All expressions have to evaluate to a boolean
value.

or_where also accepts a keyword list where each key is a field to be
compared with the given value. Each key-value pair will be combined
using AND, exactly as in where.


  
  Keywords example



from(c in City, where: [country: "Sweden"], or_where: [country: "Brazil"])

If interpolating keyword lists, the keyword list entries are combined
using ANDs and joined to any existing expression with an OR:

filters = [country: "USA", name: "New York"]
from(c in City, where: [country: "Sweden"], or_where: ^filters)

is equivalent to:

from c in City, where: (c.country == "Sweden") or
                       (c.country == "USA" and c.name == "New York")

The behaviour above is by design to keep the changes between where
and or_where minimal. Plus, if you have a keyword list and you
would like each pair to be combined using or, it can be easily done
with Enum.reduce/3:

filters = [country: "USA", is_tax_exempt: true]
Enum.reduce(filters, City, fn {key, value}, query ->
  from q in query, or_where: field(q, ^key) == ^value
end)

which will be equivalent to:

from c in City, or_where: (c.country == "USA"), or_where: c.is_tax_exempt == true


  
  Expressions example



City |> where([c], c.country == "Sweden") |> or_where([c], c.country == "Brazil")

  


    
  
    
      
      Link to this macro
    
    order_by(query, binding \\ [], expr)

      
       
       View Source
     
      (macro)
  

  
An order by query expression.

Orders the fields based on one or more fields. It accepts a single field
or a list of fields. The default direction is ascending (:asc) and can be
customized in a keyword list as one of the following:


	:asc


	:asc_nulls_last


	:asc_nulls_first


	:desc


	:desc_nulls_last


	:desc_nulls_first




The *_nulls_first and *_nulls_last variants are not supported by all
databases. While all databases default to ascending order, the choice of
"nulls first" or "nulls last" is specific to each database implementation.

order_by may be invoked or listed in a query many times. New expressions
are always appended to the previous ones.

order_by also accepts a list of atoms where each atom refers to a field in
source or a keyword list where the direction is given as key and the field
to order as value.


  
  Keywords examples



from(c in City, order_by: c.name, order_by: c.population)
from(c in City, order_by: [c.name, c.population])
from(c in City, order_by: [asc: c.name, desc: c.population])

from(c in City, order_by: [:name, :population])
from(c in City, order_by: [asc: :name, desc_nulls_first: :population])

A keyword list can also be interpolated:

values = [asc: :name, desc_nulls_first: :population]
from(c in City, order_by: ^values)

A fragment can also be used:

from c in City, order_by: [
  # a deterministic shuffled order
  fragment("? % ? DESC", c.id, ^modulus),
  desc: c.id,
]


  
  Expressions example



City |> order_by([c], asc: c.name, desc: c.population)
City |> order_by(asc: :name) # Sorts by the cities name

  


    
  
    
      
      Link to this macro
    
    preload(query, bindings \\ [], expr)

      
       
       View Source
     
      (macro)
  

  
Preloads the associations into the result set.

Imagine you have a schema Post with a has_many :comments
association and you execute the following query:

Repo.all from p in Post, preload: [:comments]

The example above will fetch all posts from the database and then do
a separate query returning all comments associated with the given posts.
The comments are then processed and associated to each returned post
under the comments field.

Often times, you may want posts and comments to be selected and
filtered in the same query. For such cases, you can explicitly tell
an existing join to be preloaded into the result set:

Repo.all from p in Post,
           join: c in assoc(p, :comments),
           where: c.published_at > p.updated_at,
           preload: [comments: c]

In the example above, instead of issuing a separate query to fetch
comments, Ecto will fetch posts and comments in a single query and
then do a separate pass associating each comment to its parent post.
Therefore, instead of returning number_of_posts * number_of_comments
results, like a join would, it returns only posts with the comments
fields properly filled in.

Nested associations can also be preloaded in both formats:

Repo.all from p in Post,
           preload: [comments: :likes]

Repo.all from p in Post,
           join: c in assoc(p, :comments),
           join: l in assoc(c, :likes),
           where: l.inserted_at > c.updated_at,
           preload: [comments: {c, likes: l}]


  
  Preload queries



Preload also allows queries to be given, allowing you to filter or
customize how the preloads are fetched:

comments_query = from c in Comment, order_by: c.published_at
Repo.all from p in Post, preload: [comments: ^comments_query]

The example above will issue two queries, one for loading posts and
then another for loading the comments associated with the posts.
Comments will be ordered by published_at.

Note: keep in mind operations like limit and offset in the preload
query will affect the whole result set and not each association. For
example, the query below:

comments_query = from c in Comment, order_by: c.popularity, limit: 5
Repo.all from p in Post, preload: [comments: ^comments_query]

won't bring the top of comments per post. Rather, it will only bring
the 5 top comments across all posts.


  
  Preload functions



Preload also allows functions to be given. In such cases, the function
receives the IDs of the parent association and it must return the associated
data. Ecto then will map this data and sort it by the relationship key:

comment_preloader = fn post_ids -> fetch_comments_by_post_ids(post_ids) end
Repo.all from p in Post, preload: [comments: ^comment_preloader]

This is useful when the whole dataset was already loaded or must be
explicitly fetched from elsewhere. The IDs received by the preloading
function and the result returned depends on the association type:


	For has_many and belongs_to - the function receives the IDs of
the parent association and it must return a list of maps or structs
with the associated entries. The associated map/struct must contain
the "foreign_key" field. For example, if a post has many comments,
when preloading the comments with a custom function, the function
will receive a list of "post_ids" as argument and it must return
maps or structs representing the comments. The maps/structs must
include the :post_id field



	For has_many :through - it behaves similarly to a regular has_many
but note that the IDs received are the ones from the closest
parent and not the furthest one. Imagine for example a post has
many comments and each comment has an author. Therefore, a post
may have many comments_authors, written as
has_many :comments_authors, through: [:comments, :author]. When
preloading authors with a custom function via :comments_authors,
the function will receive the IDs of the comments and not of the
posts. That's because through associations are still loaded step
by step



	For many_to_many -  the function receives the IDs of the parent
association and it must return a tuple with the parent id as first
element and the association map or struct as second. For example,
if a post has many tags, when preloading the tags with a custom
function, the function will receive a list of "post_ids" as argument
and it must return a tuple in the format of {post_id, tag}






  
  Keywords example



# Returns all posts, their associated comments, and the associated
# likes for those comments.
from(p in Post,
  preload: [comments: :likes],
  select: p)


  
  Expressions examples



Post |> preload(:comments) |> select([p], p)

Post
|> join(:left, [p], c in assoc(p, :comments))
|> preload([p, c], [:user, comments: c])
|> select([p], p)

  


  
    
      
      Link to this function
    
    recursive_ctes(query, value)

      
       
       View Source
     
  

  
Enables or disables recursive mode for CTEs.

According to the SQL standard it affects all CTEs in the query, not individual ones.

See with_cte/3 on example of how to build a query with a recursive CTE.

  


  
    
      
      Link to this function
    
    reverse_order(query)

      
       
       View Source
     
  

  
Reverses the ordering of the query.

ASC columns become DESC columns (and vice-versa). If the query
has no order_bys, it orders by the inverse of the primary key.


  
  Examples



query |> reverse_order |> Repo.one
Post |> order(asc: :id) |> reverse_order == Post |> order(desc: :id)

  


    
  
    
      
      Link to this macro
    
    select(query, binding \\ [], expr)

      
       
       View Source
     
      (macro)
  

  
A select query expression.

Selects which fields will be selected from the schema and any transformations
that should be performed on the fields. Any expression that is accepted in a
query can be a select field.

Select also allows each expression to be wrapped in lists, tuples or maps as
shown in the examples below. A full schema can also be selected.

There can only be one select expression in a query, if the select expression
is omitted, the query will by default select the full schema. If select is
given more than once, an error is raised. Use exclude/2 if you would like
to remove a previous select for overriding or see select_merge/3 for a
limited version of select that is composable and can be called multiple
times.

select also accepts a list of atoms where each atom refers to a field in
the source to be selected.


  
  Keywords examples



from(c in City, select: c) # returns the schema as a struct
from(c in City, select: {c.name, c.population})
from(c in City, select: [c.name, c.county])
from(c in City, select: %{n: c.name, answer: 42})
from(c in City, select: %{c | alternative_name: c.name})
from(c in City, select: %Data{name: c.name})

It is also possible to select a struct and limit the returned
fields at the same time:

from(City, select: [:name])

The syntax above is equivalent to:

from(city in City, select: struct(city, [:name]))

You can also write:

from(city in City, select: map(city, [:name]))

If you want a map with only the selected fields to be returned.
For more information, read the docs for Ecto.Query.API.struct/2
and Ecto.Query.API.map/2.


  
  Expressions examples



City |> select([c], c)
City |> select([c], {c.name, c.country})
City |> select([c], %{"name" => c.name})
City |> select([:name])
City |> select([c], struct(c, [:name]))
City |> select([c], map(c, [:name]))

  


    
  
    
      
      Link to this macro
    
    select_merge(query, binding \\ [], expr)

      
       
       View Source
     
      (macro)
  

  
Mergeable select query expression.

This macro is similar to select/3 except it may be specified
multiple times as long as every entry is a map. This is useful
for merging and composing selects. For example:

query = from p in Post, select: %{}

query =
  if include_title? do
    from p in query, select_merge: %{title: p.title}
  else
    query
  end

query =
  if include_visits? do
    from p in query, select_merge: %{visits: p.visits}
  else
    query
  end

In the example above, the query is built little by little by merging
into a final map. If both conditions above are true, the final query
would be equivalent to:

from p in Post, select: %{title: p.title, visits: p.visits}

If :select_merge is called and there is no value selected previously,
it will default to the source, p in the example above.

The argument given to :select_merge must always be a map. The value
being merged on must be a struct or a map. If it is a struct, the fields
merged later on must be part of the struct, otherwise an error is raised.

If the argument to :select_merge is a constructed struct (struct/2) or
map (map/2) where the source to struct/2 or map/2 may be a nil value
(as in an outer join), the source will be returned unmodified.

query =
  Post
  |> join(:left, [p], t in Post.Translation,
    on: t.post_id == p.id and t.locale == ^"en"
  )
  |> select_merge([_p, t], map(t, ^~w(title summary)a))

If there is no English translation for the post, the untranslated post
title will be returned and summary will be nil. If there is, both
title and summary will be the value from Post.Translation.

select_merge cannot be used to set fields in associations, as
associations are always loaded later, overriding any previous value.

  


    
  
    
      
      Link to this function
    
    subquery(query, opts \\ [])

      
       
       View Source
     
  

  
Converts a query into a subquery.

If a subquery is given, returns the subquery itself.
If any other value is given, it is converted to a query via
Ecto.Queryable and wrapped in the Ecto.SubQuery struct.

Subqueries are currently only supported in the from
and join fields.


  
  Examples



# Get the average salary of the top 10 highest salaries
query = from Employee, order_by: [desc: :salary], limit: 10
from e in subquery(query), select: avg(e.salary)

A prefix can be specified for a subquery, similar to standard repo operations:

query = from Employee, order_by: [desc: :salary], limit: 10
from e in subquery(query, prefix: "my_prefix"), select: avg(e.salary)

Although subqueries are not allowed in WHERE expressions,
most subqueries in WHERE expression can be rewritten as JOINs.
Imagine you want to write this query:

UPDATE posts
  SET sync_started_at = $1
  WHERE id IN (
    SELECT id FROM posts
      WHERE synced = false AND (sync_started_at IS NULL OR sync_started_at < $1)
      LIMIT $2
  )

If you attempt to write it as where: p.id in ^subquery(foo),
Ecto won't accept such query. However, the subquery above can be
written as a JOIN, which is supported by Ecto. The final Ecto
query will look like this:

subset_query = from(p in Post,
  where: p.synced == false and
           (is_nil(p.sync_started_at) or p.sync_started_at < ^min_sync_started_at),
  limit: ^batch_size
)

Repo.update_all(
  from(p in Post, join: s in subquery(subset_query), on: s.id == p.id),
  set: [sync_started_at: NaiveDateTime.utc_now()]
)

  


  
    
      
      Link to this macro
    
    union(query, other_query)

      
       
       View Source
     
      (macro)
  

  
A union query expression.

Combines result sets of multiple queries. The select of each query
must be exactly the same, with the same types in the same order.

Union expression returns only unique rows as if each query returned
distinct results. This may cause performance penalty. If you need
just to combine multiple result sets without removing duplicate rows
consider using union_all/2.

Note that the operations order_by, limit and offset of the
current query apply to the result of the union.


  
  Keywords example



supplier_query = from s in Supplier, select: s.city
from c in Customer, select: c.city, union: ^supplier_query


  
  Expressions example



supplier_query = Supplier |> select([s], s.city)
Customer |> select([c], c.city) |> union(^supplier_query)

  


  
    
      
      Link to this macro
    
    union_all(query, other_query)

      
       
       View Source
     
      (macro)
  

  
A union all query expression.

Combines result sets of multiple queries. The select of each query
must be exactly the same, with the same types in the same order.

Note that the operations order_by, limit and offset of the
current query apply to the result of the union.


  
  Keywords example



supplier_query = from s in Supplier, select: s.city
from c in Customer, select: c.city, union_all: ^supplier_query


  
  Expressions example



supplier_query = Supplier |> select([s], s.city)
Customer |> select([c], c.city) |> union_all(^supplier_query)

  


    
  
    
      
      Link to this macro
    
    update(query, binding \\ [], expr)

      
       
       View Source
     
      (macro)
  

  
An update query expression.

Updates are used to update the filtered entries. In order for
updates to be applied, Ecto.Repo.update_all/3 must be invoked.


  
  Keywords example



from(u in User, update: [set: [name: "new name"]])


  
  Expressions example



User |> update([u], set: [name: "new name"])
User |> update(set: [name: "new name"])


  
  Interpolation



new_name = "new name"
from(u in User, update: [set: [name: ^new_name]])

new_name = "new name"
from(u in User, update: [set: [name: fragment("upper(?)", ^new_name)]])


  
  Operators



The update expression in Ecto supports the following operators:


	set - sets the given field in the table to the given value

from(u in User, update: [set: [name: "new name"]])



	inc - increments (or decrements if the value is negative) the given field in the table by the given value

from(u in User, update: [inc: [accesses: 1]])



	push - pushes (appends) the given value to the end of the array field

from(u in User, update: [push: [tags: "cool"]])



	pull - pulls (removes) the given value from the array field

from(u in User, update: [pull: [tags: "not cool"]])





  


    
  
    
      
      Link to this macro
    
    where(query, binding \\ [], expr)

      
       
       View Source
     
      (macro)
  

  
An AND where query expression.

where expressions are used to filter the result set. If there is more
than one where expression, they are combined with an and operator. All
where expressions have to evaluate to a boolean value.

where also accepts a keyword list where the field given as key is going to
be compared with the given value. The fields will always refer to the source
given in from.


  
  Keywords example



from(c in City, where: c.country == "Sweden")
from(c in City, where: [country: "Sweden"])

It is also possible to interpolate the whole keyword list, allowing you to
dynamically filter the source:

filters = [country: "Sweden"]
from(c in City, where: ^filters)


  
  Expressions example



City |> where([c], c.country == "Sweden")
City |> where(country: "Sweden")

  


    
  
    
      
      Link to this macro
    
    windows(query, binding \\ [], expr)

      
       
       View Source
     
      (macro)
  

  
Defines windows which can be used with Ecto.Query.WindowAPI.

Receives a keyword list where keys are names of the windows
and values are a keyword list with window expressions.


  
  Examples



# Compare each employee's salary with the average salary in his or her department
from e in Employee,
  select: {e.depname, e.empno, e.salary, over(avg(e.salary), :department)},
  windows: [department: [partition_by: e.depname]]

In the example above, we get the average salary per department.
:department is the window name, partitioned by e.depname
and avg/1 is the window function. For more information
on windows functions, see Ecto.Query.WindowAPI.


  
  Window expressions



The following keys are allowed when specifying a window.


  
  :partition_by



A list of fields to partition the window by, for example:

windows: [department: [partition_by: e.depname]]

A list of atoms can also be interpolated for dynamic partitioning:

fields = [:depname, :year]
windows: [dynamic_window: [partition_by: ^fields]]


  
  :order_by



A list of fields to order the window by, for example:

windows: [ordered_names: [order_by: e.name]]

It works exactly as the keyword query version of order_by/3.


  
  :frame



A fragment which defines the frame for window functions.


  
  Examples



# compare each employee's salary for each month with his average salary for previous 3 months
from p in Payroll,
  select: {p.empno, p.date, p.salary, over(avg(p.salary), :prev_months)},
  windows: [prev_months: [partition_by: p.empno, order_by: p.date, frame: fragment("ROWS 3 PRECEDING EXCLUDE CURRENT ROW")]]

  


  
    
      
      Link to this macro
    
    with_cte(query, name, list)

      
       
       View Source
     
      (macro)
  

  
A common table expression (CTE) also known as WITH expression.

name must be a compile-time literal string that is being used
as the table name to join the CTE in the main query or in the
recursive CTE.

IMPORTANT! Beware of using CTEs. In raw SQL, CTEs can be
used as a mechanism to organize queries, but said mechanism
has no purpose in Ecto since Ecto queries are composable by
definition. In other words, if you need to break a large query
into parts, use all of the functionality in Elixir and in this
module to structure your code. Furthermore, breaking a query
into CTEs can negatively impact performance, as the database
may not optimize efficiently across CTEs. The main use case
for CTEs in Ecto is to provide recursive definitions, which
we outline in the following section. Non-recursive CTEs can
often be written as joins or subqueries, which provide better
performance.


  
  Options




	:as – the CTE query itself or a fragment





  
  Recursive CTEs



Use recursive_ctes/2 to enable recursive mode for CTEs.

In the CTE query itself use the same table name to leverage
recursion that has been passed to the name argument. Make sure
to write a stop condition to avoid infinite recursion loop.
Generally speaking, you should only use CTEs in Ecto for
writing recursive queries.


  
  Expression examples



Products and their category names for breadcrumbs:

category_tree_initial_query =
  Category
  |> where([c], is_nil(c.parent_id))

category_tree_recursion_query =
  Category
  |> join(:inner, [c], ct in "category_tree", on: c.parent_id == ct.id)

category_tree_query =
  category_tree_initial_query
  |> union_all(^category_tree_recursion_query)

Product
|> recursive_ctes(true)
|> with_cte("category_tree", as: ^category_tree_query)
|> join(:left, [p], c in "category_tree", on: c.id == p.category_id)
|> group_by([p], p.id)
|> select([p, c], %{p | category_names: fragment("ARRAY_AGG(?)", c.name)})

It's possible to cast CTE result rows as Ecto structs:

{"category_tree", Category}
|> recursive_ctes(true)
|> with_cte("category_tree", as: ^category_tree_query)
|> join(:left, [c], p in assoc(c, :products))
|> group_by([c], c.id)
|> select([c, p], %{c | products_count: count(p.id)})

It's also possible to pass a raw SQL fragment:

@raw_sql_category_tree """
SELECT * FROM categories WHERE c.parent_id IS NULL
UNION ALL
SELECT * FROM categories AS c, category_tree AS ct WHERE ct.id = c.parent_id
"""

Product
|> recursive_ctes(true)
|> with_cte("category_tree", as: fragment(@raw_sql_category_tree))
|> join(:inner, [p], c in "category_tree", on: c.id == p.category_id)

Keyword syntax is not supported for this feature.


  
  Limitation: CTEs on schemas wth source fields



Ecto allows developers to say that a table in their Ecto schema
map to a different column in their database:

field :group_id, :integer, source: :iGroupId

At the moment, using a schema with source fields in CTE may emit
invalid queries. If you are running into such scenarios, your best
option is to use a fragment as your CTE.

  

        

      
  

  
    
    Ecto.Repo - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Repo behaviour    



      
Defines a repository.

A repository maps to an underlying data store, controlled by the
adapter. For example, Ecto ships with a Postgres adapter that
stores data into a PostgreSQL database.

When used, the repository expects the :otp_app and :adapter as
option. The :otp_app should point to an OTP application that has
the repository configuration. For example, the repository:

defmodule Repo do
  use Ecto.Repo,
    otp_app: :my_app,
    adapter: Ecto.Adapters.Postgres
end

Could be configured with:

config :my_app, Repo,
  database: "ecto_simple",
  username: "postgres",
  password: "postgres",
  hostname: "localhost"

Most of the configuration that goes into the config is specific
to the adapter. For this particular example, you can check
Ecto.Adapters.Postgres
for more information. In spite of this, the following configuration values
are shared across all adapters:


	:name- The name of the Repo supervisor process



	:priv - the directory where to keep repository data, like
migrations, schema and more. Defaults to "priv/YOUR_REPO".
It must always point to a subdirectory inside the priv directory



	:url - an URL that specifies storage information. Read below
for more information



	:log - the log level used when logging the query with Elixir's
Logger. If false, disables logging for that repository.
Defaults to :debug



	:pool_size - the size of the pool used by the connection module.
Defaults to 10



	:telemetry_prefix - we recommend adapters to publish events
using the Telemetry library. By default, the telemetry prefix
is based on the module name, so if your module is called
MyApp.Repo, the prefix will be [:my_app, :repo]. See the
"Telemetry Events" section to see which events we recommend
adapters to publish





URLs

Repositories by default support URLs. For example, the configuration
above could be rewritten to:

config :my_app, Repo,
  url: "ecto://postgres:postgres@localhost/ecto_simple"

The schema can be of any value. The path represents the database name
while options are simply merged in.

URL can include query parameters to override shared and adapter-specific
options, like ssl, timeout and pool_size. The following example
shows how to pass these configuration values:

config :my_app, Repo,
  url: "ecto://postgres:postgres@localhost/ecto_simple?ssl=true&pool_size=10"

In case the URL needs to be dynamically configured, for example by
reading a system environment variable, such can be done via the
init/2 repository callback:

def init(_type, config) do
  {:ok, Keyword.put(config, :url, System.get_env("DATABASE_URL"))}
end

Shared options

Almost all of the repository functions outlined in this module accept the following
options:


	:timeout - The time in milliseconds to wait for the query call to
finish. :infinity will wait indefinitely (default: 15000)


	:log - When false, does not log the query


	:telemetry_event - The telemetry event name to dispatch the event under.
See the next section for more information




Telemetry events

We recommend adapters to publish certain Telemetry events listed below.
Those events will use the :telemetry_prefix outlined above which defaults
to [:my_app, :repo].

For instance, to receive all query events published by a repository called
MyApp.Repo, one would define a module:

defmodule MyApp.Telemetry do
  def handle_event([:my_app, :repo, :query], measurements, metadata, config) do
    IO.inspect binding()
  end
end

Then, in the Application.start/2 callback, attach the handler to this event using
a unique handler id:

:ok = :telemetry.attach("my-app-handler-id", [:my_app, :repo, :query], &MyApp.Telemetry.handle_event/4, %{})

For details, see the telemetry documentation.

Below we list all events developers should expect from Ecto. All examples below consider
a repository named MyApp.Repo:

[:my_app, :repo, :query]

This event should be invoked on every query sent to the adapter, including
queries that are related to the transaction management.

The :measurements map will include the following, all given in the
:native time unit:


	:idle_time - the time the connection spent waiting before being checked out for the query


	:queue_time - the time spent waiting to check out a database connection


	:query_time - the time spent executing the query


	:decode_time - the time spent decoding the data received from the database


	:total_time - the sum of the other measurements




All measurements are given in the :native time unit. You can read more
about it in the docs for System.convert_time_unit/3.

A :metadata map is also sent, including parameters, source, the query
string, the repo it was run by, and the query result.

Read-only repositories

You can mark a repository as read-only by passing the :read_only
flag on use:

use Ecto.Repo, otp_app: ..., adapter: ..., read_only: true

By passing the :read_only option, none of the functions that perform
write operations, such as insert/2, insert_all/3, update_all/3,
and friends will be defined.

      Types
    


  
    t()
  



  

  
    
      Callbacks
    


  
    __adapter__()
  

    Returns the adapter tied to the repository.





  
    aggregate(queryable, aggregate, opts)
  

    Calculate the given aggregate.





  
    aggregate(queryable, aggregate, field, opts)
  

    Calculate the given aggregate over the given field.





  
    all(queryable, opts)
  

    Fetches all entries from the data store matching the given query.





  
    checkout(function, opts)
  

    Checks out a connection for the duration of the function.





  
    config()
  

    Returns the adapter configuration stored in the :otp_app environment.





  
    delete(struct_or_changeset, opts)
  

    Deletes a struct using its primary key.





  
    delete!(struct_or_changeset, opts)
  

    Same as delete/2 but returns the struct or raises if the changeset is invalid.





  
    delete_all(queryable, opts)
  

    Deletes all entries matching the given query.





  
    exists?(queryable, opts)
  

    Checks if there exists an entry that matches the given query.





  
    get(queryable, id, opts)
  

    Fetches a single struct from the data store where the primary key matches the
given id.





  
    get!(queryable, id, opts)
  

    Similar to get/3 but raises Ecto.NoResultsError if no record was found.





  
    get_by(queryable, clauses, opts)
  

    Fetches a single result from the query.





  
    get_by!(queryable, clauses, opts)
  

    Similar to get_by/3 but raises Ecto.NoResultsError if no record was found.





  
    get_dynamic_repo()
  

    Returns the atom name or pid of the current repository.





  
    in_transaction?()
  

    Returns true if the current process is inside a transaction.





  
    init(context, config)
  

    A callback executed when the repo starts or when configuration is read.





  
    insert(struct_or_changeset, opts)
  

    Inserts a struct defined via Ecto.Schema or a changeset.





  
    insert!(struct_or_changeset, opts)
  

    Same as insert/2 but returns the struct or raises if the changeset is invalid.





  
    insert_all(schema_or_source, entries, opts)
  

    Inserts all entries into the repository.





  
    insert_or_update(changeset, opts)
  

    Inserts or updates a changeset depending on whether the struct is persisted
or not.





  
    insert_or_update!(changeset, opts)
  

    Same as insert_or_update/2 but returns the struct or raises if the changeset
is invalid.





  
    load(module_or_map, data)
  

    Loads data into a struct or a map.





  
    one(queryable, opts)
  

    Fetches a single result from the query.





  
    one!(queryable, opts)
  

    Similar to one/2 but raises Ecto.NoResultsError if no record was found.





  
    preload(structs_or_struct_or_nil, preloads, opts)
  

    Preloads all associations on the given struct or structs.





  
    prepare_query(operation, query, opts)
  

    A user customizable callback invoked for query-based operations.





  
    put_dynamic_repo(arg1)
  

    Sets the dynamic repository to be used in further interactions.





  
    rollback(value)
  

    Rolls back the current transaction.





  
    start_link(opts)
  

    Starts any connection pooling or supervision and return {:ok, pid}
or just :ok if nothing needs to be done.





  
    stop(timeout)
  

    Shuts down the repository.





  
    stream(queryable, opts)
  

    Returns a lazy enumerable that emits all entries from the data store
matching the given query.





  
    transaction(fun_or_multi, opts)
  

    Runs the given function or Ecto.Multi inside a transaction.





  
    update(changeset, opts)
  

    Updates a changeset using its primary key.





  
    update!(changeset, opts)
  

    Same as update/2 but returns the struct or raises if the changeset is invalid.





  
    update_all(queryable, updates, opts)
  

    Updates all entries matching the given query with the given values.

      Link to this type
    
    t()

      
       
       View Source
     
      
          t() :: module()

      Link to this callback
    
    __adapter__()

      
       
       View Source
     
      
          __adapter__() :: Ecto.Adapter.t()

      

  

  
Returns the adapter tied to the repository.

  


  
    
      
      Link to this callback
    
    aggregate(queryable, aggregate, opts)

      
       
       View Source
     
      (optional)
      
          aggregate(
  queryable :: Ecto.Queryable.t(),
  aggregate :: :count,
  opts :: Keyword.t()
) :: term() | nil

      

  

  
Calculate the given aggregate.

If the query has a limit, offset or distinct set, it will be
automatically wrapped in a subquery in order to return the
proper result.

Any preload or select in the query will be ignored in favor of
the column being aggregated.

The aggregation will fail if any group_by field is set.


  
  Options




	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This will be applied to all from
and joins in the query that did not have a prefix previously given
either via the :prefix option on join/from or via @schema_prefix
in the schema. For more information see the "Query Prefix" section of the
Ecto.Query documentation.




See the "Shared options" section at the module documentation for more options.


  
  Examples



# Returns the number of blog posts
Repo.aggregate(Post, :count)

# Returns the number of blog posts in the "private" schema path
# (in Postgres) or database (in MySQL)
Repo.aggregate(Post, :count, prefix: "private")

  


  
    
      
      Link to this callback
    
    aggregate(queryable, aggregate, field, opts)

      
       
       View Source
     
      (optional)
      
          aggregate(
  queryable :: Ecto.Queryable.t(),
  aggregate :: :avg | :count | :max | :min | :sum,
  field :: atom(),
  opts :: Keyword.t()
) :: term() | nil

      

  

  
Calculate the given aggregate over the given field.

See aggregate/2 for general considerations and options.


  
  Examples



# Returns the number of visits per blog post
Repo.aggregate(Post, :count, :visits)

# Returns the number of visits per blog post in the "private" schema path
# (in Postgres) or database (in MySQL)
Repo.aggregate(Post, :count, :visits, prefix: "private")

# Returns the average number of visits for the top 10
query = from Post, limit: 10
Repo.aggregate(query, :avg, :visits)

  


  
    
      
      Link to this callback
    
    all(queryable, opts)

      
       
       View Source
     
      (optional)
      
          all(queryable :: Ecto.Queryable.t(), opts :: Keyword.t()) :: [Ecto.Schema.t()]

      

  

  
Fetches all entries from the data store matching the given query.

May raise Ecto.QueryError if query validation fails.


  
  Options




	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This will be applied to all from
and joins in the query that did not have a prefix previously given
either via the :prefix option on join/from or via @schema_prefix
in the schema. For more information see the "Query Prefix" section of the
Ecto.Query documentation.




See the "Shared options" section at the module documentation for more options.


  
  Example



# Fetch all post titles
query = from p in Post,
     select: p.title
MyRepo.all(query)

  


  
    
      
      Link to this callback
    
    checkout(function, opts)

      
       
       View Source
     
      
          checkout((() -> result), opts :: Keyword.t()) :: result when result: var

      

  

  
Checks out a connection for the duration of the function.

It returns the result of the function. This is useful when
you need to perform multiple operations against the repository
in a row and you want to avoid checking out the connection
multiple times.

checkout/2 and transaction/2 can be combined and nested
multiple times. If checkout/2 is called inside the function
of another checkout/2 call, the function is simply executed,
without checking out a new connection.


  
  Options



See the "Shared options" section at the module documentation for more options.

  


  
    
      
      Link to this callback
    
    config()

      
       
       View Source
     
      
          config() :: Keyword.t()

      

  

  
Returns the adapter configuration stored in the :otp_app environment.

If the init/2 callback is implemented in the repository,
it will be invoked with the first argument set to :runtime.

  


  
    
      
      Link to this callback
    
    delete(struct_or_changeset, opts)

      
       
       View Source
     
      (optional)
      
          delete(
  struct_or_changeset :: Ecto.Schema.t() | Ecto.Changeset.t(),
  opts :: Keyword.t()
) :: {:ok, Ecto.Schema.t()} | {:error, Ecto.Changeset.t()}

      

  

  
Deletes a struct using its primary key.

If the struct has no primary key, Ecto.NoPrimaryKeyFieldError
will be raised. If the struct has been removed from db prior to
call, Ecto.StaleEntryError will be raised.

It returns {:ok, struct} if the struct has been successfully
deleted or {:error, changeset} if there was a validation
or a known constraint error.


  
  Options




	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This overrides the prefix set
in the query and any @schema_prefix set in the schema.


	:stale_error_field - The field where stale errors will be added in
the returning changeset. This option can be used to avoid raising
Ecto.StaleEntryError.


	:stale_error_message - The message to add to the configured
:stale_error_field when stale errors happen, defaults to "is stale".




See the "Shared options" section at the module documentation for more options.


  
  Example



post = MyRepo.get!(Post, 42)
case MyRepo.delete post do
  {:ok, struct}       -> # Deleted with success
  {:error, changeset} -> # Something went wrong
end

  


  
    
      
      Link to this callback
    
    delete!(struct_or_changeset, opts)

      
       
       View Source
     
      (optional)
      
          delete!(
  struct_or_changeset :: Ecto.Schema.t() | Ecto.Changeset.t(),
  opts :: Keyword.t()
) :: Ecto.Schema.t()

      

  

  
Same as delete/2 but returns the struct or raises if the changeset is invalid.

  


  
    
      
      Link to this callback
    
    delete_all(queryable, opts)

      
       
       View Source
     
      (optional)
      
          delete_all(queryable :: Ecto.Queryable.t(), opts :: Keyword.t()) ::
  {integer(), nil | [term()]}

      

  

  
Deletes all entries matching the given query.

It returns a tuple containing the number of entries and any returned
result as second element. The second element is nil by default
unless a select is supplied in the delete query. Note, however,
not all databases support returning data from DELETEs.


  
  Options




	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This overrides the prefix set
in the query and any @schema_prefix set in the schema.




See the "Shared options" section at the module documentation for
remaining options.


  
  Examples



MyRepo.delete_all(Post)

from(p in Post, where: p.id < 10) |> MyRepo.delete_all

  


  
    
      
      Link to this callback
    
    exists?(queryable, opts)

      
       
       View Source
     
      (optional)
      
          exists?(queryable :: Ecto.Queryable.t(), opts :: Keyword.t()) :: boolean()

      

  

  
Checks if there exists an entry that matches the given query.

Returns a boolean.


  
  Options




	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This will be applied to all from
and joins in the query that did not have a prefix previously given
either via the :prefix option on join/from or via @schema_prefix
in the schema. For more information see the "Query Prefix" section of the
Ecto.Query documentation.




See the "Shared options" section at the module documentation for more options.


  
  Examples



# checks if any posts exist
Repo.exists?(Post)

# checks if any posts exist in the "private" schema path (in Postgres) or
# database (in MySQL)
Repo.exists?(Post, schema: "private")

# checks if any post with a like count greater than 10 exists
query = from p in Post, where: p.like_count > 10
Repo.exists?(query)

  


  
    
      
      Link to this callback
    
    get(queryable, id, opts)

      
       
       View Source
     
      (optional)
      
          get(queryable :: Ecto.Queryable.t(), id :: term(), opts :: Keyword.t()) ::
  Ecto.Schema.t() | nil

      

  

  
Fetches a single struct from the data store where the primary key matches the
given id.

Returns nil if no result was found. If the struct in the queryable
has no or more than one primary key, it will raise an argument error.


  
  Options




	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This will be applied to all from
and joins in the query that did not have a prefix previously given
either via the :prefix option on join/from or via @schema_prefix
in the schema. For more information see the "Query Prefix" section of the
Ecto.Query documentation.




See the "Shared options" section at the module documentation for more options.


  
  Example



MyRepo.get(Post, 42)

MyRepo.get(Post, 42, prefix: "public")

  


  
    
      
      Link to this callback
    
    get!(queryable, id, opts)

      
       
       View Source
     
      (optional)
      
          get!(queryable :: Ecto.Queryable.t(), id :: term(), opts :: Keyword.t()) ::
  Ecto.Schema.t()

      

  

  
Similar to get/3 but raises Ecto.NoResultsError if no record was found.


  
  Options




	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This will be applied to all from
and joins in the query that did not have a prefix previously given
either via the :prefix option on join/from or via @schema_prefix
in the schema. For more information see the "Query Prefix" section of the
Ecto.Query documentation.




See the "Shared options" section at the module documentation for more options.


  
  Example



MyRepo.get!(Post, 42)

MyRepo.get!(Post, 42, prefix: "public")

  


  
    
      
      Link to this callback
    
    get_by(queryable, clauses, opts)

      
       
       View Source
     
      (optional)
      
          get_by(
  queryable :: Ecto.Queryable.t(),
  clauses :: Keyword.t() | map(),
  opts :: Keyword.t()
) :: Ecto.Schema.t() | nil

      

  

  
Fetches a single result from the query.

Returns nil if no result was found. Raises if more than one entry.


  
  Options




	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This will be applied to all from
and joins in the query that did not have a prefix previously given
either via the :prefix option on join/from or via @schema_prefix
in the schema. For more information see the "Query Prefix" section of the
Ecto.Query documentation.




See the "Shared options" section at the module documentation for more options.


  
  Example



MyRepo.get_by(Post, title: "My post")

MyRepo.get_by(Post, [title: "My post"], prefix: "public")

  


  
    
      
      Link to this callback
    
    get_by!(queryable, clauses, opts)

      
       
       View Source
     
      (optional)
      
          get_by!(
  queryable :: Ecto.Queryable.t(),
  clauses :: Keyword.t() | map(),
  opts :: Keyword.t()
) :: Ecto.Schema.t()

      

  

  
Similar to get_by/3 but raises Ecto.NoResultsError if no record was found.

Raises if more than one entry.


  
  Options




	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This will be applied to all from
and joins in the query that did not have a prefix previously given
either via the :prefix option on join/from or via @schema_prefix
in the schema. For more information see the "Query Prefix" section of the
Ecto.Query documentation.




See the "Shared options" section at the module documentation for more options.


  
  Example



MyRepo.get_by!(Post, title: "My post")

MyRepo.get_by!(Post, [title: "My post"], prefix: "public")

  


  
    
      
      Link to this callback
    
    get_dynamic_repo()

      
       
       View Source
     
      
          get_dynamic_repo() :: atom() | pid()

      

  

  
Returns the atom name or pid of the current repository.

See put_dynamic_repo/1 for more information.

  


  
    
      
      Link to this callback
    
    in_transaction?()

      
       
       View Source
     
      (optional)
      
          in_transaction?() :: boolean()

      

  

  
Returns true if the current process is inside a transaction.

If you are using the Ecto.Adapters.SQL.Sandbox in tests, note that even
though each test is inside a transaction, in_transaction?/0 will only
return true inside transactions explicitly created with transaction/2. This
is done so the test environment mimics dev and prod.

If you are trying to debug transaction-related code while using
Ecto.Adapters.SQL.Sandbox, it may be more helpful to configure the database
to log all statements and consult those logs.


  
  Examples



MyRepo.in_transaction?
#=> false

MyRepo.transaction(fn ->
  MyRepo.in_transaction? #=> true
end)

  


  
    
      
      Link to this callback
    
    init(context, config)

      
       
       View Source
     
      (optional)
      
          init(context :: :supervisor | :runtime, config :: Keyword.t()) ::
  {:ok, Keyword.t()} | :ignore

      

  

  
A callback executed when the repo starts or when configuration is read.

The first argument is the context the callback is being invoked. If it
is called because the Repo supervisor is starting, it will be :supervisor.
It will be :runtime if it is called for reading configuration without
actually starting a process.

The second argument is the repository configuration as stored in the
application environment. It must return {:ok, keyword} with the updated
list of configuration or :ignore (only in the :supervisor case).

  


  
    
      
      Link to this callback
    
    insert(struct_or_changeset, opts)

      
       
       View Source
     
      (optional)
      
          insert(
  struct_or_changeset :: Ecto.Schema.t() | Ecto.Changeset.t(),
  opts :: Keyword.t()
) :: {:ok, Ecto.Schema.t()} | {:error, Ecto.Changeset.t()}

      

  

  
Inserts a struct defined via Ecto.Schema or a changeset.

In case a struct is given, the struct is converted into a changeset
with all non-nil fields as part of the changeset.

In case a changeset is given, the changes in the changeset are
merged with the struct fields, and all of them are sent to the
database.

It returns {:ok, struct} if the struct has been successfully
inserted or {:error, changeset} if there was a validation
or a known constraint error.


  
  Options




	:returning - selects which fields to return. It accepts a list
of fields to be returned from the database. When true, returns
all fields. When false, no extra fields are returned. It will
always include all fields in read_after_writes as well as any
autogenerated id. Not all databases support this option and it
may not be available during upserts. See the "Upserts" section
for more information.


	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This overrides the prefix set
in the query and any @schema_prefix set any schemas. Also, the
@schema_prefix for the parent record will override all default
@schema_prefixs set in any child schemas for associations.


	:on_conflict - It may be one of :raise (the default), :nothing,
:replace_all, {:replace_all_except, fields}, {:replace, fields},
a keyword list of update instructions or an Ecto.Query query for updates.
See the "Upserts" section for more information.


	:conflict_target - A list of column names to verify for conflicts.
It is expected those columns to have unique indexes on them that may conflict.
If none is specified, the conflict target is left up to the database.
May also be {:constraint, constraint_name_as_atom} in databases
that support the "ON CONSTRAINT" expression, such as PostgreSQL.


	:stale_error_field - The field where stale errors will be added in
the returning changeset. This option can be used to avoid raising
Ecto.StaleEntryError.


	:stale_error_message - The message to add to the configured
:stale_error_field when stale errors happen, defaults to "is stale".




See the "Shared options" section at the module documentation for more options.


  
  Examples



A typical example is calling MyRepo.insert/1 with a struct
and acting on the return value:

case MyRepo.insert %Post{title: "Ecto is great"} do
  {:ok, struct}       -> # Inserted with success
  {:error, changeset} -> # Something went wrong
end


  
  Upserts



insert/2 provides upserts (update or inserts) via the :on_conflict
option. The :on_conflict option supports the following values:


	:raise - raises if there is a conflicting primary key or unique index


	:nothing - ignores the error in case of conflicts


	:replace_all - replace all values on the existing row with the values
in the schema/changeset, including autogenerated fields such as inserted_at
and updated_at


	{:replace_all_except, fields} - same as above except the given fields are
not replaced


	{:replace, fields} - replace only specific columns. This option requires
conflict_target


	a keyword list of update instructions - such as the one given to
update_all/3, for example: [set: [title: "new title"]]


	an Ecto.Query that will act as an UPDATE statement, such as the
one given to update_all/3. If the struct cannot be found, Ecto.StaleEntryError
will be raised.




Upserts map to "ON CONFLICT" on databases like Postgres and "ON DUPLICATE KEY"
on databases such as MySQL.

As an example, imagine :title is marked as a unique column in
the database:

{:ok, inserted} = MyRepo.insert(%Post{title: "this is unique"})

Now we can insert with the same title but do nothing on conflicts:

{:ok, ignored} = MyRepo.insert(%Post{title: "this is unique"}, on_conflict: :nothing)
assert ignored.id == nil

Because we used on_conflict: :nothing, instead of getting an error,
we got {:ok, struct}. However the returned struct does not reflect
the data in the database. One possible mechanism to detect if an
insert or nothing happened in case of on_conflict: :nothing is by
checking the id field. id will be nil if the field is autogenerated
by the database and no insert happened.

For actual upserts, where an insert or update may happen, the situation
is slightly more complex, as the database does not actually inform us
if an insert or update happened. Let's insert a post with the same title
but use a query to update the body column in case of conflicts:

# In Postgres (it requires the conflict target for updates):
on_conflict = [set: [body: "updated"]]
{:ok, updated} = MyRepo.insert(%Post{title: "this is unique"},
                               on_conflict: on_conflict, conflict_target: :title)

# In MySQL (conflict target is not supported):
on_conflict = [set: [title: "updated"]]
{:ok, updated} = MyRepo.insert(%Post{id: inserted.id, title: "updated"},
                               on_conflict: on_conflict)

In the examples above, even though it returned :ok, we do not know
if we inserted new data or if we updated only the :on_conflict fields.
In case an update happened, the data in the struct most likely does
not match the data in the database. For example, autogenerated fields
such as inserted_at will point to now rather than the time the
struct was actually inserted.

If you need to guarantee the data in the returned struct mirrors the
database, you have three options:


	Use on_conflict: :replace_all, although that will replace all
fields in the database with the ones in the struct/changeset,
including autogenerated fields such as insert_at and updated_at:

MyRepo.insert(%Post{title: "this is unique"},
              on_conflict: :replace_all, conflict_target: :title)



	Specify read_after_writes: true in your schema for choosing
fields that are read from the database after every operation.
Or pass returning: true to insert to read all fields back:

MyRepo.insert(%Post{title: "this is unique"}, returning: true,
              on_conflict: on_conflict, conflict_target: :title)



	Alternatively, read the data again from the database in a separate
query. This option requires the primary key to be generated by the
database:

{:ok, updated} = MyRepo.insert(%Post{title: "this is unique"}, on_conflict: on_conflict)
Repo.get(Post, updated.id)





Because of the inability to know if the struct is up to date or not,
inserting a struct with associations and using the :on_conflict option
at the same time is not recommended, as Ecto will be unable to actually
track the proper status of the association.

  


  
    
      
      Link to this callback
    
    insert!(struct_or_changeset, opts)

      
       
       View Source
     
      (optional)
      
          insert!(
  struct_or_changeset :: Ecto.Schema.t() | Ecto.Changeset.t(),
  opts :: Keyword.t()
) :: Ecto.Schema.t()

      

  

  
Same as insert/2 but returns the struct or raises if the changeset is invalid.

  


  
    
      
      Link to this callback
    
    insert_all(schema_or_source, entries, opts)

      
       
       View Source
     
      (optional)
      
          insert_all(
  schema_or_source :: binary() | {binary(), module()} | module(),
  entries :: [map() | [{atom(), term() | Ecto.Query.t()}]],
  opts :: Keyword.t()
) :: {integer(), nil | [term()]}

      

  

  
Inserts all entries into the repository.

It expects a schema module (MyApp.User) or a source ("users") or
both ({"users", MyApp.User}) as the first argument. The second
argument is a list of entries to be inserted, either as keyword
lists or as maps. The keys of the entries are the field names as
atoms and the value should be the respective value for the field
type or, optionally, an Ecto.Query that returns a single entry
with a single value.

It returns a tuple containing the number of entries
and any returned result as second element. If the database
does not support RETURNING in INSERT statements or no
return result was selected, the second element will be nil.

When a schema module is given, the entries given will be properly dumped
before being sent to the database. If the schema contains an
autogenerated ID field, it will be handled either at the adapter
or the storage layer. However any other autogenerated value, like
timestamps, won't be autogenerated when using insert_all/3.
This is by design as this function aims to be a more direct way
to insert data into the database without the conveniences of
insert/2. This is also consistent with update_all/3 that
does not handle timestamps as well.

It is also not possible to use insert_all to insert across multiple
tables, therefore associations are not supported.

If a source is given, without a schema module, the given fields are passed
as is to the adapter.


  
  Options




	:returning - selects which fields to return. When true,
returns all fields in the given schema. May be a list of
fields, where a struct is still returned but only with the
given fields. Or false, where nothing is returned (the default).
This option is not supported by all databases.


	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This overrides the prefix set
in the query and any @schema_prefix set in the schema.


	:on_conflict - It may be one of :raise (the default), :nothing,
:replace_all, {:replace_all_except, fields}, {:replace, fields},
a keyword list of update instructions or an Ecto.Query
query for updates. See the "Upserts" section for more information.


	:conflict_target - A list of column names to verify for conflicts.
It is expected those columns to have unique indexes on them that may conflict.
If none is specified, the conflict target is left up to the database.
It may also be {:constraint, constraint_name_as_atom} in databases
that support the "ON CONSTRAINT" expression, such as PostgreSQL, or
{:unsafe_fragment, binary_fragment} to pass any expression to the
database without any sanitization, such as
ON CONFLICT (coalesce(firstname, ""), coalesce(lastname, "")).




See the "Shared options" section at the module documentation for
remaining options.


  
  Examples



MyRepo.insert_all(Post, [[title: "My first post"], [title: "My second post"]])

MyRepo.insert_all(Post, [%{title: "My first post"}, %{title: "My second post"}])


  
  Upserts



insert_all/3 provides upserts (update or inserts) via the :on_conflict
option. The :on_conflict option supports the following values:


	:raise - raises if there is a conflicting primary key or unique index


	:nothing - ignores the error in case of conflicts


	:replace_all - replace all values on the existing row by the new entry,
including values not sent explicitly by Ecto, such as database defaults.
This option requires a schema


	{:replace_all_except, fields} - same as above except the given fields
are not replaced. This option requires a schema


	{:replace, fields} - replace only specific columns. This option requires
conflict_target


	a keyword list of update instructions - such as the one given to
update_all/3, for example: [set: [title: "new title"]]


	an Ecto.Query that will act as an UPDATE statement, such as the
one given to update_all/3




Upserts map to "ON CONFLICT" on databases like Postgres and "ON DUPLICATE KEY"
on databases such as MySQL.


  
  Return values



By default, both Postgres and MySQL return the amount of entries
inserted on insert_all/3. However, when the :on_conflict option
is specified, Postgres will only return a row if it was affected
while MySQL returns at least the number of entries attempted.

For example, if :on_conflict is set to :nothing, Postgres will
return 0 if no new entry was added while MySQL will still return
the amount of entries attempted to be inserted, even if no entry
was added. Even worse, if :on_conflict is query, MySQL will return
the number of attempted entries plus the number of entries modified
by the UPDATE query.

  


  
    
      
      Link to this callback
    
    insert_or_update(changeset, opts)

      
       
       View Source
     
      (optional)
      
          insert_or_update(changeset :: Ecto.Changeset.t(), opts :: Keyword.t()) ::
  {:ok, Ecto.Schema.t()} | {:error, Ecto.Changeset.t()}

      

  

  
Inserts or updates a changeset depending on whether the struct is persisted
or not.

The distinction whether to insert or update will be made on the
Ecto.Schema.Metadata field :state. The :state is automatically set by
Ecto when loading or building a schema.

Please note that for this to work, you will have to load existing structs from
the database. So even if the struct exists, this won't work:

struct = %Post{id: "existing_id", ...}
MyRepo.insert_or_update changeset
# => {:error, changeset} # id already exists


  
  Options




	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This overrides the prefix set
in the query and any @schema_prefix set any schemas. Also, the
@schema_prefix for the parent record will override all default
@schema_prefixs set in any child schemas for associations.


	:stale_error_field - The field where stale errors will be added in
the returning changeset. This option can be used to avoid raising
Ecto.StaleEntryError. Only applies to updates.


	:stale_error_message - The message to add to the configured
:stale_error_field when stale errors happen, defaults to "is stale".
Only applies to updates.




See the "Shared options" section at the module documentation for more options.


  
  Example



result =
  case MyRepo.get(Post, id) do
    nil  -> %Post{id: id} # Post not found, we build one
    post -> post          # Post exists, let's use it
  end
  |> Post.changeset(changes)
  |> MyRepo.insert_or_update

case result do
  {:ok, struct}       -> # Inserted or updated with success
  {:error, changeset} -> # Something went wrong
end

  


  
    
      
      Link to this callback
    
    insert_or_update!(changeset, opts)

      
       
       View Source
     
      (optional)
      
          insert_or_update!(changeset :: Ecto.Changeset.t(), opts :: Keyword.t()) ::
  Ecto.Schema.t()

      

  

  
Same as insert_or_update/2 but returns the struct or raises if the changeset
is invalid.

  


  
    
      
      Link to this callback
    
    load(module_or_map, data)

      
       
       View Source
     
      
          load(
  module_or_map :: module() | map(),
  data :: map() | Keyword.t() | {list(), list()}
) :: Ecto.Schema.t() | map()

      

  

  
Loads data into a struct or a map.

The first argument can be a a schema module, or a
map (of types) and determines the return value:
a struct or a map, respectively.

The second argument data specifies fields and values that are to be loaded.
It can be a map, a keyword list, or a {fields, values} tuple.
Fields can be atoms or strings.

Fields that are not present in the schema (or types map) are ignored.
If any of the values has invalid type, an error is raised.


  
  Examples



iex> MyRepo.load(User, %{name: "Alice", age: 25})
%User{name: "Alice", age: 25}

iex> MyRepo.load(User, [name: "Alice", age: 25])
%User{name: "Alice", age: 25}

data can also take form of {fields, values}:

iex> MyRepo.load(User, {[:name, :age], ["Alice", 25]})
%User{name: "Alice", age: 25, ...}

The first argument can also be a types map:

iex> types = %{name: :string, age: :integer}
iex> MyRepo.load(types, %{name: "Alice", age: 25})
%{name: "Alice", age: 25}

This function is especially useful when parsing raw query results:

iex> result = Ecto.Adapters.SQL.query!(MyRepo, "SELECT * FROM users", [])
iex> Enum.map(result.rows, &MyRepo.load(User, {result.columns, &1}))
[%User{...}, ...]

  


  
    
      
      Link to this callback
    
    one(queryable, opts)

      
       
       View Source
     
      (optional)
      
          one(queryable :: Ecto.Queryable.t(), opts :: Keyword.t()) ::
  Ecto.Schema.t() | nil

      

  

  
Fetches a single result from the query.

Returns nil if no result was found. Raises if more than one entry.


  
  Options




	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This will be applied to all from
and joins in the query that did not have a prefix previously given
either via the :prefix option on join/from or via @schema_prefix
in the schema. For more information see the "Query Prefix" section of the
Ecto.Query documentation.




See the "Shared options" section at the module documentation for more options.


  
  Examples



Repo.one(Post)

Repo.one(from p in Post, where: p.like_count > 10)

query = from p in Post, where: p.like_count > 10
Repo.one(query, prefix: "private")

  


  
    
      
      Link to this callback
    
    one!(queryable, opts)

      
       
       View Source
     
      (optional)
      
          one!(queryable :: Ecto.Queryable.t(), opts :: Keyword.t()) :: Ecto.Schema.t()

      

  

  
Similar to one/2 but raises Ecto.NoResultsError if no record was found.

Raises if more than one entry.


  
  Options




	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This will be applied to all from
and joins in the query that did not have a prefix previously given
either via the :prefix option on join/from or via @schema_prefix
in the schema. For more information see the "Query Prefix" section of the
Ecto.Query documentation.




See the "Shared options" section at the module documentation for more options.

  


  
    
      
      Link to this callback
    
    preload(structs_or_struct_or_nil, preloads, opts)

      
       
       View Source
     
      (optional)
      
          preload(structs_or_struct_or_nil, preloads :: term(), opts :: Keyword.t()) ::
  structs_or_struct_or_nil
when structs_or_struct_or_nil: [Ecto.Schema.t()] | Ecto.Schema.t() | nil

      

  

  
Preloads all associations on the given struct or structs.

This is similar to Ecto.Query.preload/3 except it allows
you to preload structs after they have been fetched from the
database.

In case the association was already loaded, preload won't attempt
to reload it.


  
  Options




	:force - By default, Ecto won't preload associations that
are already loaded. By setting this option to true, any existing
association will be discarded and reloaded.


	:in_parallel - If the preloads must be done in parallel. It can
only be performed when we have more than one preload and the
repository is not in a transaction. Defaults to true.


	:prefix - the prefix to fetch preloads from. By default, queries
will use the same prefix as the one in the given collection. This
option allows the prefix to be changed.




See the "Shared options" section at the module documentation for more options.


  
  Examples



# Use a single atom to preload an association
posts = Repo.preload posts, :comments

# Use a list of atoms to preload multiple associations
posts = Repo.preload posts, [:comments, :authors]

# Use a keyword list to preload nested associations as well
posts = Repo.preload posts, [comments: [:replies, :likes], authors: []]

# Use a keyword list to customize how associations are queried
posts = Repo.preload posts, [comments: from(c in Comment, order_by: c.published_at)]

# Use a two-element tuple for a custom query and nested association definition
query = from c in Comment, order_by: c.published_at
posts = Repo.preload posts, [comments: {query, [:replies, :likes]}]

The query given to preload may also preload its own associations.

  


  
    
      
      Link to this callback
    
    prepare_query(operation, query, opts)

      
       
       View Source
     
      (optional)
      
          prepare_query(operation, query :: Ecto.Query.t(), opts :: Keyword.t()) ::
  {Ecto.Query.t(), Keyword.t()}
when operation: :all | :update_all | :delete_all | :stream

      

  

  
A user customizable callback invoked for query-based operations.

This callback can be used to further modify the query and options
before it is transformed and sent to the database.

This callback is invoked for all query APIs, including the stream
function, but it is not invoked for insert_all nor any of the
schema functions.


  
  Examples



Let's say you want to filter out records that were "soft-deleted" (have deleted_at
column set) from all operations unless an admin is running the query; you can define
the callback like this:

@impl true
def prepare_query(_operation, query, opts) do
  if opts[:admin] do
    {query, opts}
  else
    query = from(x in query, where: is_nil(x.deleted_at))
    {query, opts}
  end
end

And then execute the query:

Repo.all(query)              # only non-deleted records are returned
Repo.all(query, admin: true) # all records are returned

  


  
    
      
      Link to this callback
    
    put_dynamic_repo(arg1)

      
       
       View Source
     
      
          put_dynamic_repo(atom() | pid()) :: atom() | pid()

      

  

  
Sets the dynamic repository to be used in further interactions.

Sometimes you may want a single Ecto repository to talk to
many different database instances. By default, when you call
MyApp.Repo.start_link/1, it will start a repository with
name MyApp.Repo. But if you want to start multiple repositories,
you can give each of them a different name:

MyApp.Repo.start_link(name: :tenant_foo, hostname: "foo.example.com")
MyApp.Repo.start_link(name: :tenant_bar, hostname: "bar.example.com")

You can also start repositories without names by explicitly
setting the name to nil:

MyApp.Repo.start_link(name: nil, hostname: "temp.example.com")

However, once the repository is started, you can't directly interact with
it, since all operations in MyApp.Repo are sent by default to the repository
named MyApp.Repo. You can change the default repo at compile time with:

use Ecto.Repo, default_dynamic_repo: :name_of_repo

Or you can change it anytime at runtime by calling put_dynamic_repo/1:

MyApp.Repo.put_dynamic_repo(:tenant_foo)

From this moment on, all future queries done by the current process will
run on :tenant_foo.

Note this feature is experimental and may be changed or removed in future
releases.

  


  
    
      
      Link to this callback
    
    rollback(value)

      
       
       View Source
     
      (optional)
      
          rollback(value :: any()) :: no_return()

      

  

  
Rolls back the current transaction.

The transaction will return the value given as {:error, value}.

  


  
    
      
      Link to this callback
    
    start_link(opts)

      
       
       View Source
     
      
          start_link(opts :: Keyword.t()) ::
  {:ok, pid()} | {:error, {:already_started, pid()}} | {:error, term()}

      

  

  
Starts any connection pooling or supervision and return {:ok, pid}
or just :ok if nothing needs to be done.

Returns {:error, {:already_started, pid}} if the repo is already
started or {:error, term} in case anything else goes wrong.


  
  Options



See the configuration in the moduledoc for options shared between adapters,
for adapter-specific configuration see the adapter's documentation.

  


  
    
      
      Link to this callback
    
    stop(timeout)

      
       
       View Source
     
      
          stop(timeout()) :: :ok

      

  

  
Shuts down the repository.

  


  
    
      
      Link to this callback
    
    stream(queryable, opts)

      
       
       View Source
     
      (optional)
      
          stream(queryable :: Ecto.Queryable.t(), opts :: Keyword.t()) :: Enum.t()

      

  

  
Returns a lazy enumerable that emits all entries from the data store
matching the given query.

SQL adapters, such as Postgres and MySQL, can only enumerate a stream
inside a transaction.

May raise Ecto.QueryError if query validation fails.


  
  Options




	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This will be applied to all from
and joins in the query that did not have a prefix previously given
either via the :prefix option on join/from or via @schema_prefix
in the schema. For more information see the "Query Prefix" section of the
Ecto.Query documentation.



	:max_rows - The number of rows to load from the database as we stream.
It is supported at least by Postgres and MySQL and defaults to 500.





See the "Shared options" section at the module documentation for more options.


  
  Example



# Fetch all post titles
query = from p in Post,
     select: p.title
stream = MyRepo.stream(query)
MyRepo.transaction(fn() ->
  Enum.to_list(stream)
end)

  


  
    
      
      Link to this callback
    
    transaction(fun_or_multi, opts)

      
       
       View Source
     
      (optional)
      
          transaction(
  fun_or_multi :: (... -> any()) | Ecto.Multi.t(),
  opts :: Keyword.t()
) ::
  {:ok, any()}
  | {:error, any()}
  | {:error, Ecto.Multi.name(), any(),
     %{required(Ecto.Multi.name()) => any()}}

      

  

  
Runs the given function or Ecto.Multi inside a transaction.


  
  Use with function



transaction/2 can be called with both a function of arity
zero or one. The arity zero function will just be executed as is,
while the arity one function will receive the repo of the transaction
as its first argument, similar to Ecto.Multi.run.

If an unhandled error occurs the transaction will be rolled back
and the error will bubble up from the transaction function.
If no error occurred the transaction will be committed when the
function returns. A transaction can be explicitly rolled back
by calling rollback/1, this will immediately leave the function
and return the value given to rollback as {:error, value}.

A successful transaction returns the value returned by the function
wrapped in a tuple as {:ok, value}.

If transaction/2 is called inside another transaction, the function
is simply executed, without wrapping the new transaction call in any
way. If there is an error in the inner transaction and the error is
rescued, or the inner transaction is rolled back, the whole outer
transaction is marked as tainted, guaranteeing nothing will be committed.


  
  Use with Ecto.Multi



Besides functions transaction can be used with an Ecto.Multi struct.
Transaction will be started, all operations applied and in case of
success committed returning {:ok, changes}. In case of any errors
the transaction will be rolled back and
{:error, failed_operation, failed_value, changes_so_far} will be
returned.

You can read more about using transactions with Ecto.Multi as well as
see some examples in the Ecto.Multi documentation.


  
  Options



See the "Shared options" section at the module documentation for more options.


  
  Examples



import Ecto.Changeset, only: [change: 2]

MyRepo.transaction(fn ->
  MyRepo.update!(change(alice, balance: alice.balance - 10))
  MyRepo.update!(change(bob, balance: bob.balance + 10))
end)

# When passing a function of arity 1, it receives the repository itself
MyRepo.transaction(fn repo -> 
  repo.insert!(%Post{})
end)

# Roll back a transaction explicitly
MyRepo.transaction(fn ->
  p = MyRepo.insert!(%Post{})
  if not Editor.post_allowed?(p) do
    MyRepo.rollback(:posting_not_allowed)
  end
end)

# With Ecto.Multi
Ecto.Multi.new()
|> Ecto.Multi.insert(:post, %Post{})
|> MyRepo.transaction

  


  
    
      
      Link to this callback
    
    update(changeset, opts)

      
       
       View Source
     
      (optional)
      
          update(changeset :: Ecto.Changeset.t(), opts :: Keyword.t()) ::
  {:ok, Ecto.Schema.t()} | {:error, Ecto.Changeset.t()}

      

  

  
Updates a changeset using its primary key.

A changeset is required as it is the only mechanism for
tracking dirty changes. Only the fields present in the changes part
of the changeset are sent to the database. Any other, in-memory
changes done to the schema are ignored.

If the struct has no primary key, Ecto.NoPrimaryKeyFieldError
will be raised.

If the struct cannot be found, Ecto.StaleEntryError will be raised.

It returns {:ok, struct} if the struct has been successfully
updated or {:error, changeset} if there was a validation
or a known constraint error.


  
  Options




	:returning - selects which fields to return. It accepts a list
of fields to be returned from the database. When true, returns
all fields. When false, no extra fields are returned. It will
always include all fields in read_after_writes. Not all
databases support this option.


	:force - By default, if there are no changes in the changeset,
update/2 is a no-op. By setting this option to true, update
callbacks will always be executed, even if there are no changes
(including timestamps).


	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This overrides the prefix set
in the query and any @schema_prefix set any schemas. Also, the
@schema_prefix for the parent record will override all default
@schema_prefixs set in any child schemas for associations.


	:stale_error_field - The field where stale errors will be added in
the returning changeset. This option can be used to avoid raising
Ecto.StaleEntryError.


	:stale_error_message - The message to add to the configured
:stale_error_field when stale errors happen, defaults to "is stale".




See the "Shared options" section at the module documentation for more options.


  
  Example



post = MyRepo.get!(Post, 42)
post = Ecto.Changeset.change post, title: "New title"
case MyRepo.update post do
  {:ok, struct}       -> # Updated with success
  {:error, changeset} -> # Something went wrong
end

  


  
    
      
      Link to this callback
    
    update!(changeset, opts)

      
       
       View Source
     
      (optional)
      
          update!(changeset :: Ecto.Changeset.t(), opts :: Keyword.t()) :: Ecto.Schema.t()

      

  

  
Same as update/2 but returns the struct or raises if the changeset is invalid.

  


  
    
      
      Link to this callback
    
    update_all(queryable, updates, opts)

      
       
       View Source
     
      (optional)
      
          update_all(
  queryable :: Ecto.Queryable.t(),
  updates :: Keyword.t(),
  opts :: Keyword.t()
) :: {integer(), nil | [term()]}

      

  

  
Updates all entries matching the given query with the given values.

It returns a tuple containing the number of entries and any returned
result as second element. The second element is nil by default
unless a select is supplied in the update query. Note, however,
not all databases support returning data from UPDATEs.

Keep in mind this update_all will not update autogenerated
fields like the updated_at columns.

See Ecto.Query.update/3 for update operations that can be
performed on fields.


  
  Options




	:prefix - The prefix to run the query on (such as the schema path
in Postgres or the database in MySQL). This overrides the prefix set
in the query and any @schema_prefix set in the schema.




See the "Shared options" section at the module documentation for
remaining options.


  
  Examples



MyRepo.update_all(Post, set: [title: "New title"])

MyRepo.update_all(Post, inc: [visits: 1])

from(p in Post, where: p.id < 10, select: p.visits)
|> MyRepo.update_all(set: [title: "New title"])

from(p in Post, where: p.id < 10, update: [set: [title: "New title"]])
|> MyRepo.update_all([])

from(p in Post, where: p.id < 10, update: [set: [title: ^new_title]])
|> MyRepo.update_all([])

from(p in Post, where: p.id < 10, update: [set: [title: fragment("upper(?)", ^new_title)]])
|> MyRepo.update_all([])

Defines a schema.

An Ecto schema is used to map any data source into an Elixir struct.
The definition of the schema is possible through two main APIs:
schema/2 and embedded_schema/1.

schema/2 is typically used to map data from a persisted source,
usually a database table, into Elixir structs and vice-versa. For
this reason, the first argument of schema/2 is the source (table)
name. Structs defined with schema/2 also contain a __meta__ field
with metadata holding the status of the struct, for example, if it
has been built, loaded or deleted.

On the other hand, embedded_schema/1 is used for defining schemas
that are embedded in other schemas or only exist in-memory. For example,
you can use such schemas to receive data from a command line interface
and validate it, without ever persisting it elsewhere. Such structs
do not contain a __meta__ field, as they are never persisted.

Besides working as data mappers, embedded_schema/1 and schema/2 can
also be used together to decouple how the data is represented in your
applications from the database. Let's see some examples.

Example

defmodule User do
  use Ecto.Schema

  schema "users" do
    field :name, :string
    field :age, :integer, default: 0
    has_many :posts, Post
  end
end

By default, a schema will automatically generate a primary key which is named
id and of type :integer. The field macro defines a field in the schema
with given name and type. has_many associates many posts with the user
schema. Schemas are regular structs and can be created and manipulated directly
using Elixir's struct API:

iex> user = %User{name: "jane"}
iex> %{user | age: 30}

However, most commonly, structs are cast, validated and manipulated with the
Ecto.Changeset module.

Note that the name of the database table does not need to correlate to your
module name.  For example, if you are working with a legacy database, you can
reference the table name when you define your schema:

defmodule User do
  use Ecto.Schema

  schema "legacy_users" do
    # ... fields ...
  end
end

Embedded schemas are defined similarly to source-based schemas. For example,
you can use an embedded schema to represent your UI, mapping and validating
its inputs, and then you convert such embedded schema to other schemas that
are persisted to the database:

defmodule SignUp do
  use Ecto.Schema

  embedded_schema do
    field :name, :string
    field :age, :integer
    field :email, :string
    field :accepts_conditions, :boolean
  end
end

defmodule Profile do
  use Ecto.Schema

  schema "profiles" do
    field :name
    field :age
    belongs_to :account, Account
  end
end

defmodule Account do
  use Ecto.Schema

  schema "accounts" do
    field :email
  end
end

The SignUp schema can be cast and validated with the help of the
Ecto.Changeset module, and afterwards, you can copy its data to
the Profile and Account structs that will be persisted to the
database with the help of Ecto.Repo.

Schema attributes

Supported attributes for configuring the defined schema. They must
be set after the use Ecto.Schema call and before the schema/2
definition.

These attributes are:


	@primary_key - configures the schema primary key. It expects
a tuple {field_name, type, options} with the primary key field
name, type (typically :id or :binary_id, but can be any type) and
options. It also accepts false to disable the generation of a primary
key field. Defaults to {:id, :id, autogenerate: true}. 



	@schema_prefix - configures the schema prefix. Defaults to nil,
which generates structs and queries without prefix. When set, the
prefix will be used by every built struct and on queries whenever
the schema is used in a from or a join. In PostgreSQL, the prefix
is called "SCHEMA" (typically set via Postgres' search_path).
In MySQL the prefix points to databases.



	@foreign_key_type - configures the default foreign key type
used by belongs_to associations. It must be set in the same
module that defines the belongs_to. Defaults to :id;



	@timestamps_opts - configures the default timestamps type
used by timestamps. Defaults to [type: :naive_datetime];



	@derive - the same as @derive available in Kernel.defstruct/1
as the schema defines a struct behind the scenes;



	@field_source_mapper - a function that receives the current field name
and returns the mapping of this field name in the underlying source.
In other words, it is a mechanism to automatically generate the :source
option for the field macro. It defaults to fn x -> x end, where no
field transformation is done;





The advantage of configuring the schema via those attributes is
that they can be set with a macro to configure application wide
defaults.

For example, if your database does not support autoincrementing
primary keys and requires something like UUID or a RecordID, you
can configure and use :binary_id as your primary key type as follows:

# Define a module to be used as base
defmodule MyApp.Schema do
  defmacro __using__(_) do
    quote do
      use Ecto.Schema
      @primary_key {:id, :binary_id, autogenerate: true}
      @foreign_key_type :binary_id
    end
  end
end

# Now use MyApp.Schema to define new schemas
defmodule MyApp.Comment do
  use MyApp.Schema

  schema "comments" do
    belongs_to :post, MyApp.Post
  end
end

Any schemas using MyApp.Schema will get the :id field with type
:binary_id as the primary key. We explain what the :binary_id type
entails in the next section.

The belongs_to association on MyApp.Comment will also define
a :post_id field with :binary_id type that references the :id
field of the MyApp.Post schema.

Primary keys

Ecto supports two ID types, called :id and :binary_id, which are
often used as the type for primary keys and associations.

The :id type is used when the primary key is an integer while the
:binary_id is used for primary keys in particular binary formats,
which may be Ecto.UUID for databases like PostgreSQL and MySQL,
or some specific ObjectID or RecordID often imposed by NoSQL databases.

In both cases, both types have their semantics specified by the
underlying adapter/database. If you use the :id type with
:autogenerate, it means the database will be responsible for
auto-generation of the id. This is often the case for primary keys
in relational databases which are auto-incremented.

There are two ways to define primary keys in Ecto: using the @primary_key
module attribute and using primary_key: true as option for field/3 in
your schema definition. They are not mutually exclusive and can be used
together.

Using @primary_key should be prefered for single field primary keys and
sharing primary key definitions between multiple schemas using macros.

Ecto also supports composite primary keys, which is where you need to use
primary_key: true for the fields in your schema. This usually goes along 
with setting @primary_key false to disable generation of additional 
primary key fields.

Besides :id and :binary_id, which are often used by primary
and foreign keys, Ecto provides a huge variety of types to be used
by any field.

Types and casting

When defining the schema, types need to be given. Types are split
into two categories, primitive types and custom types.

Primitive types

The primitive types are:




	Ecto type	Elixir type	Literal syntax in query





	:id	integer	1, 2, 3



	:binary_id	binary	<<int, int, int, ...>>



	:integer	integer	1, 2, 3



	:float	float	1.0, 2.0, 3.0



	:boolean	boolean	true, false



	:string	UTF-8 encoded string	"hello"



	:binary	binary	<<int, int, int, ...>>



	{:array, inner_type}	list	[value, value, value, ...]



	:map	map	



	{:map, inner_type}	map	



	:decimal	Decimal	



	:date	Date	



	:time	Time	



	:time_usec	Time	



	:naive_datetime	NaiveDateTime	



	:naive_datetime_usec	NaiveDateTime	



	:utc_datetime	DateTime	



	:utc_datetime_usec	DateTime	





Notes:


	For the {:array, inner_type} and {:map, inner_type} type,
replace inner_type with one of the valid types, such as :string.



	For the :decimal type, +Infinity, -Infinity, and NaN values
are not supported, even though the Decimal library handles them.
To support them, you can create a custom type.



	For calendar types with and without microseconds, the precision is
enforced when persisting to the DB. For example, casting ~T[09:00:00]
as :time_usec will succeed and result in ~T[09:00:00.000000], but
persisting a type without microseconds as :time_usec will fail.
Similarly, casting ~T[09:00:00.000000] as :time will succeed, but
persisting will not. This is the same behaviour as seen in other types,
where casting has to be done explicitly and is never performed
implicitly when loading from or dumping to the database.





Custom types

Besides providing primitive types, Ecto allows custom types to be
implemented by developers, allowing Ecto behaviour to be extended.

A custom type is a module that implements the Ecto.Type behaviour.
By default, Ecto provides the following custom types:




	Custom type	Database type	Elixir type





	Ecto.UUID	:uuid	uuid-string





Read the Ecto.Type documentation for more information on implementing
your own types.

Finally, schemas can also have virtual fields by passing the
virtual: true option. These fields are not persisted to the database
and can optionally not be type checked by declaring type :any.

The datetime types

Four different datetime primitive types are available:


	naive_datetime - has a precision of seconds and casts values
to Elixir's NaiveDateTime struct which has no timezone information.



	naive_datetime_usec - has a default precision of microseconds and
also casts values to NaiveDateTime with no timezone information.



	utc_datetime - has a precision of seconds and casts values to
Elixir's DateTime struct and expects the time zone to be set to UTC.



	utc_datetime_usec has a default precision of microseconds and also
casts values to DateTime expecting the time zone be set to UTC.





All of those types are represented by the same timestamp/datetime in the
underlying data storage, the difference are in their precision and how the
data is loaded into Elixir.

Having different precisions allows developers to choose a type that will
be compatible with the database and your project's precision requirements.
For example, some older versions of MySQL do not support microseconds in
datetime fields.

When choosing what datetime type to work with, keep in mind that Elixir
functions like NaiveDateTime.utc_now/0 have a default precision of 6.
Casting a value with a precision greater than 0 to a non-usec type will
truncate all microseconds and set the precision to 0.

The map type

The map type allows developers to store an Elixir map directly
in the database:

# In your migration
create table(:users) do
  add :data, :map
end

# In your schema
field :data, :map

# Now in your code
user = Repo.insert! %User{data: %{"foo" => "bar"}}

Keep in mind that we advise the map keys to be strings or integers
instead of atoms. Atoms may be accepted depending on how maps are
serialized but the database will always convert atom keys to strings
due to security reasons.

In order to support maps, different databases may employ different
techniques. For example, PostgreSQL will store those values in jsonb
fields, allowing you to just query parts of it. MSSQL, on
the other hand, does not yet provide a JSON type, so the value will be
stored in a text field.

For maps to work in such databases, Ecto will need a JSON library.
By default Ecto will use Jason
which needs to be added to your deps in mix.exs:

{:jason, "~> 1.0"}

You can however configure the adapter to use another library. For example,
if using Postgres:

config :postgrex, :json_library, YourLibraryOfChoice

Or if using MySQL:

config :mariaex, :json_library, YourLibraryOfChoice

If changing the JSON library, remember to recompile the adapter afterwards
by cleaning the current build:

mix deps.clean --build postgrex

Casting

When directly manipulating the struct, it is the responsibility of
the developer to ensure the field values have the proper type. For
example, you can create a user struct with an invalid value
for age:

iex> user = %User{age: "0"}
iex> user.age
"0"

However, if you attempt to persist the struct above, an error will
be raised since Ecto validates the types when sending them to the
adapter/database.

Therefore, when working with and manipulating external data, it is
recommended to use Ecto.Changeset's that are able to filter
and properly cast external data:

changeset = Ecto.Changeset.cast(%User{}, %{"age" => "0"}, [:age])
user = Repo.insert!(changeset)

You can use Ecto schemas and changesets to cast and validate any kind
of data, regardless if the data will be persisted to an Ecto repository
or not.

Reflection

Any schema module will generate the __schema__ function that can be
used for runtime introspection of the schema:


	__schema__(:source) - Returns the source as given to schema/2;



	__schema__(:prefix) - Returns optional prefix for source provided by
@schema_prefix schema attribute;



	__schema__(:primary_key) - Returns a list of primary key fields (empty if there is none);



	__schema__(:fields) - Returns a list of all non-virtual field names;



	__schema__(:field_source, field) - Returns the alias of the given field;



	__schema__(:type, field) - Returns the type of the given non-virtual field;



	__schema__(:associations) - Returns a list of all association field names;



	__schema__(:association, assoc) - Returns the association reflection of the given assoc;



	__schema__(:embeds) - Returns a list of all embedded field names;



	__schema__(:embed, embed) - Returns the embedding reflection of the given embed;



	__schema__(:read_after_writes) - Non-virtual fields that must be read back
from the database after every write (insert or update);



	__schema__(:autogenerate_id) - Primary key that is auto generated on insert;





Furthermore, both __struct__ and __changeset__ functions are
defined so structs and changeset functionalities are available.

       View Source
     
      (macro)
  

  
Indicates a one-to-one or many-to-one association with another schema.

The current schema belongs to zero or one records of the other schema. The other
schema often has a has_one or a has_many field with the reverse association.

You should use belongs_to in the table that contains the foreign key. Imagine
a company <-> employee relationship. If the employee contains the company_id in
the underlying database table, we say the employee belongs to company.

In fact, when you invoke this macro, a field with the name of foreign key is
automatically defined in the schema for you.


  
  Options




	:foreign_key - Sets the foreign key field name, defaults to the name
of the association suffixed by _id. For example, belongs_to :company
will define foreign key of :company_id. The associated has_one or has_many
field in the other schema should also have its :foreign_key option set
with the same value.



	:references - Sets the key on the other schema to be used for the
association, defaults to: :id



	:define_field - When false, does not automatically define a :foreign_key
field, implying the user is defining the field manually elsewhere



	:type - Sets the type of automatically defined :foreign_key.
Defaults to: :integer and can be set per schema via @foreign_key_type



	:on_replace - The action taken on associations when the record is
replaced when casting or manipulating parent changeset. May be
:raise (default), :mark_as_invalid, :nilify, :update, or :delete.
See Ecto.Changeset's section on related data for more info.



	:defaults - Default values to use when building the association.
It may be a keyword list of options that override the association schema
or a {module, function, args} that receive the struct and the owner as
arguments. For example, if you set Post.has_many :comments, defaults: [public: true],
then when using Ecto.build_assoc(post, :comments) that comment will have
comment.public == true. Alternatively, you can set it to
Post.has_many :comments, defaults: {__MODULE__, :update_comment, []}
and Post.update_comment(comment, post) will be invoked.



	:primary_key - If the underlying belongs_to field is a primary key



	:source - Defines the name that is to be used in database for this field



	:where - A filter for the association. See "Filtering associations"
in has_many/3.






  
  Examples



defmodule Comment do
  use Ecto.Schema

  schema "comments" do
    belongs_to :post, Post
  end
end

# The post can come preloaded on the comment record
[comment] = Repo.all(from(c in Comment, where: c.id == 42, preload: :post))
comment.post #=> %Post{...}

If you need custom options on the underlying field, you can define the
field explicitly and then pass define_field: false to belongs_to:

defmodule Comment do
  use Ecto.Schema

  schema "comments" do
    field :post_id, :integer, ... # custom options
    belongs_to :post, Post, define_field: false
  end
end


  
  Polymorphic associations



One common use case for belongs to associations is to handle
polymorphism. For example, imagine you have defined a Comment
schema and you wish to use it for commenting on both tasks and
posts.

Some abstractions would force you to define some sort of
polymorphic association with two fields in your database:

* commentable_type
* commentable_id

The problem with this approach is that it breaks references in
the database. You can't use foreign keys and it is very inefficient,
both in terms of query time and storage.

In Ecto, we have three ways to solve this issue. The simplest
is to define multiple fields in the Comment schema, one for each
association:

* task_id
* post_id

Unless you have dozens of columns, this is simpler for the developer,
more DB friendly and more efficient in all aspects.

Alternatively, because Ecto does not tie a schema to a given table,
we can use separate tables for each association. Let's start over
and define a new Comment schema:

defmodule Comment do
  use Ecto.Schema

  schema "abstract table: comments" do
    # This will be used by associations on each "concrete" table
    field :assoc_id, :integer
  end
end

Notice we have changed the table name to "abstract table: comments".
You can choose whatever name you want, the point here is that this
particular table will never exist.

Now in your Post and Task schemas:

defmodule Post do
  use Ecto.Schema

  schema "posts" do
    has_many :comments, {"posts_comments", Comment}, foreign_key: :assoc_id
  end
end

defmodule Task do
  use Ecto.Schema

  schema "tasks" do
    has_many :comments, {"tasks_comments", Comment}, foreign_key: :assoc_id
  end
end

Now each association uses its own specific table, "posts_comments"
and "tasks_comments", which must be created on migrations. The
advantage of this approach is that we never store unrelated data
together, also ensuring we keep database references fast and correct.

When using this technique, the only limitation is that you cannot
build comments directly. For example, the command below

Repo.insert!(%Comment{})

will attempt to use the abstract table. Instead, one should use

Repo.insert!(build_assoc(post, :comments))

leveraging the Ecto.build_assoc/3 function. You can also
use Ecto.assoc/2 or pass a tuple in the query syntax
to easily retrieve associated comments to a given post or
task:

# Fetch all comments associated with the given task
Repo.all(Ecto.assoc(task, :comments))

Or all comments in a given table:

Repo.all from(c in {"posts_comments", Comment}), ...)

The third and final option is to use many_to_many/3 to
define the relationships between the resources. In this case,
the comments table won't have the foreign key, instead there
is a intermediary table responsible for associating the entries:

defmodule Comment do
  use Ecto.Schema
  schema "comments" do
    # ...
  end
end

In your posts and tasks:

defmodule Post do
  use Ecto.Schema

  schema "posts" do
    many_to_many :comments, Comment, join_through: "posts_comments"
  end
end

defmodule Task do
  use Ecto.Schema

  schema "tasks" do
    many_to_many :comments, Comment, join_through: "tasks_comments"
  end
end

See many_to_many/3 for more information on this particular approach.

Defines an embedded schema with the given field definitions.

An embedded schema is either embedded into another
schema or kept exclusively in memory. For this reason,
an embedded schema does not require a source name and
it does not include a metadata field.

Embedded schemas by default set the primary key type
to :binary_id but such can be configured with the
@primary_key attribute.

  


    
  
    
      
      Link to this macro
    
    embeds_many(name, schema, opts \\ [])

      
       
       View Source
     
      (macro)
  

  
Indicates an embedding of many schemas.

The current schema has zero or more records of the other schema embedded
inside of it. Embeds have all the things regular schemas have.

It is recommended to declare your embeds_many/3 field with type :map
in your migrations, instead of using {:array, :map}. Ecto can work with
both maps and arrays as the container for embeds (and in most databases
map are represented as JSON which allows Ecto to choose what works best).

The embedded may or may not have a primary key. Ecto uses the primary keys
to detect if an embed is being updated or not. If a primary is not present
and you still want the list of embeds to be updated, :on_replace must be
set to :delete, forcing all current embeds to be deleted and replaced by
new ones whenever a new list of embeds is set.

For encoding and decoding of embeds, please read the docs for
embeds_one/3.


  
  Options




	:on_replace - The action taken on associations when the embed is
replaced when casting or manipulating parent changeset. May be
:raise (default), :mark_as_invalid, or :delete.
See Ecto.Changeset's section on related data for more info.





  
  Examples



defmodule Order do
  use Ecto.Schema

  schema "orders" do
    embeds_many :items, Item
  end
end

defmodule Item do
  use Ecto.Schema

  embedded_schema do
    field :title
  end
end

# The items are loaded with the order
order = Repo.get!(Order, 42)
order.items #=> [%Item{...}, ...]

Adding and removal of embeds can only be done via the Ecto.Changeset
API so Ecto can properly track the embed life-cycle:

# Order has no items
order = Repo.get!(Order, 42)
order.items
# => []

items  = [%Item{title: "Soap"}]

# Generate a changeset
changeset = Ecto.Changeset.change(order)

# Put a one or more new items
changeset = Ecto.Changeset.put_embed(changeset, :items, items)

# Update the order and fetch items
items = Repo.update!(changeset).items

# Items are generated with a unique identification
items
# => [%Item{id: "20a97d94-f79b-4e63-a875-85deed7719b7", title: "Soap"}]

Updating of embeds must be done using a changeset for each changed embed.

# Order has an existing items
order = Repo.get!(Order, 42)
order.items
# => [%Item{id: "20a97d94-f79b-4e63-a875-85deed7719b7", title: "Soap"}]

# Generate a changeset
changeset = Ecto.Changeset.change(order)

# Put the updated item as a changeset
current_item = List.first(order.items)
item_changeset = Ecto.Changeset.change(current_item, title: "Mujju's Soap")
order_changeset = Ecto.Changeset.put_embed(changeset, :items, [item_changeset])

# Update the order and fetch items
items = Repo.update!(order_changeset).items

# Item has the updated title
items
# => [%Item{id: "20a97d94-f79b-4e63-a875-85deed7719b7", title: "Mujju's Soap"}]


  
  Inline embedded schema



The schema module can be defined inline in the parent schema in simple
cases:

defmodule Parent do
  use Ecto.Schema

  schema "parents" do
    field :name, :string

    embeds_many :children, Child do
      field :name, :string
      field :age,  :integer
    end
  end
end

Primary keys are automatically set up for  embedded  schemas as well,
defaulting  to  {:id, :binary_id, autogenerate: true}. You can
customize it by passing a :primary_key option with the same arguments
as @primary_key (see the Schema attributes
section for more info).

Defining embedded schema in such a way will define a Parent.Child module
with the appropriate struct. In order to properly cast the embedded schema.
When casting the inline-defined embedded schemas you need to use the :with
option of cast_embed/3 to provide the proper function to do the casting.
For example:

def changeset(schema, params) do
  schema
  |> cast(params, [:name])
  |> cast_embed(:children, with: &child_changeset/2)
end

defp child_changeset(schema, params) do
  schema
  |> cast(params, [:name, :age])
end

  


  
    
      
      Link to this macro
    
    embeds_many(name, schema, opts, list)

      
       
       View Source
     
      (macro)
  

  
Indicates an embedding of many schemas.

For options and examples see documentation of embeds_many/3.

  


    
  
    
      
      Link to this macro
    
    embeds_one(name, schema, opts \\ [])

      
       
       View Source
     
      (macro)
  

  
Indicates an embedding of a schema.

The current schema has zero or one records of the other schema embedded
inside of it. It uses a field similar to the :map type for storage,
but allows embeds to have all the things regular schema can.

You must declare your embeds_one/3 field with type :map at the
database level.

The embedded may or may not have a primary key. Ecto uses the primary keys
to detect if an embed is being updated or not. If a primary is not present,
:on_replace should be set to either :update or :delete if there is a
desire to either update or delete the current embed when a new one is set.


  
  Options




	:on_replace - The action taken on associations when the embed is
replaced when casting or manipulating parent changeset. May be
:raise (default), :mark_as_invalid, :update, or :delete.
See Ecto.Changeset's section on related data for more info.





  
  Examples



defmodule Order do
  use Ecto.Schema

  schema "orders" do
    embeds_one :item, Item
  end
end

defmodule Item do
  use Ecto.Schema

  embedded_schema do
    field :title
  end
end

# The item is loaded with the order
order = Repo.get!(Order, 42)
order.item #=> %Item{...}

Adding and removal of embeds can only be done via the Ecto.Changeset
API so Ecto can properly track the embed life-cycle:

order = Repo.get!(Order, 42)
item  = %Item{title: "Soap"}

# Generate a changeset
changeset = Ecto.Changeset.change(order)

# Put a new embed to the changeset
changeset = Ecto.Changeset.put_embed(changeset, :item, item)

# Update the order, and fetch the item
item = Repo.update!(changeset).item

# Item is generated with a unique identification
item
# => %Item{id: "20a97d94-f79b-4e63-a875-85deed7719b7", title: "Soap"}


  
  Inline embedded schema



The schema module can be defined inline in the parent schema in simple
cases:

defmodule Parent do
  use Ecto.Schema

  schema "parents" do
    field :name, :string

    embeds_one :child, Child do
      field :name, :string
      field :age,  :integer
    end
  end
end

Options should be passed before the do block like this:

embeds_one :child, Child, on_replace: :delete do
  field :name, :string
  field :age,  :integer
end

Primary keys are automatically set up for  embedded  schemas as well,
defaulting  to  {:id, :binary_id, autogenerate: true}. You can
customize it by passing a :primary_key option with the same arguments
as @primary_key (see the Schema attributes
section for more info).

Defining embedded schema in such a way will define a Parent.Child module
with the appropriate struct. In order to properly cast the embedded schema.
When casting the inline-defined embedded schemas you need to use the :with
option of Ecto.Changeset.cast_embed/3 to provide the proper function to do the casting.
For example:

def changeset(schema, params) do
  schema
  |> cast(params, [:name])
  |> cast_embed(:child, with: &child_changeset/2)
end

defp child_changeset(schema, params) do
  schema
  |> cast(params, [:name, :age])
end


  
  Encoding and decoding



Because many databases do not support direct encoding and decoding
of embeds, it is often emulated by Ecto by using specific encoding
and decoding rules.

For example, PostgreSQL will store embeds on top of JSONB columns,
which means types in embedded schemas won't go through the usual
dump->DB->load cycle but rather encode->DB->decode->cast. This means
that, when using embedded schemas with databases like PG or MySQL,
make sure all of your types can be JSON encoded/decoded correctly.
Ecto provides this guarantee for all built-in types.

  


  
    
      
      Link to this macro
    
    embeds_one(name, schema, opts, list)

      
       
       View Source
     
      (macro)
  

  
Indicates an embedding of a schema.

For options and examples see documentation of embeds_one/3.

  


    
    
  
    
      
      Link to this macro
    
    field(name, type \\ :string, opts \\ [])

      
       
       View Source
     
      (macro)
  

  
Defines a field on the schema with given name and type.

The field name will be used as is to read and write to the database
by all of the built-in adapters unless overridden with the :source
option.


  
  Options




	:default - Sets the default value on the schema and the struct.
The default value is calculated at compilation time, so don't use
expressions like DateTime.utc_now or Ecto.UUID.generate as
they would then be the same for all records.



	:source - Defines the name that is to be used in database for this field.
This is useful when attaching to an existing database. The value should be
an atom.



	:autogenerate - a {module, function, args} tuple for a function
to call to generate the field value before insertion if value is not set.
A shorthand value of true is equivalent to {type, :autogenerate, []}.



	:read_after_writes - When true, the field is always read back
from the database after insert and updates.

For relational databases, this means the RETURNING option of those
statements is used. For this reason, MySQL does not support this
option and will raise an error if a schema is inserted/updated with
read after writes fields.



	:virtual - When true, the field is not persisted to the database.
Notice virtual fields do not support :autogenerate nor
:read_after_writes.



	:primary_key - When true, the field is used as part of the
composite primary key.



	:load_in_query - When false, the field will not be loaded when
selecting the whole struct in a query, such as from p in Post, select: p.
Defaults to true.





  


    
  
    
      
      Link to this macro
    
    has_many(name, queryable, opts \\ [])

      
       
       View Source
     
      (macro)
  

  
Indicates a one-to-many association with another schema.

The current schema has zero or more records of the other schema. The other
schema often has a belongs_to field with the reverse association.


  
  Options




	:foreign_key - Sets the foreign key, this should map to a field on the
other schema, defaults to the underscored name of the current schema
suffixed by _id



	:references - Sets the key on the current schema to be used for the
association, defaults to the primary key on the schema



	:through - Allow this association to be defined in terms of existing
associations. Read the section on :through associations for more info



	:on_delete - The action taken on associations when parent record
is deleted. May be :nothing (default), :nilify_all and :delete_all.
Using this option is DISCOURAGED for most relational databases. Instead,
in your migration, set references(:parent_id, on_delete: :delete_all).
Opposite to the migration option, this option cannot guarantee integrity
and it is only triggered for Ecto.Repo.delete/2 (and not on
Ecto.Repo.delete_all/2) and it never cascades. If posts has many comments,
which has many tags, and you delete a post, only comments will be deleted.
If your database does not support references, cascading can be manually
implemented by using Ecto.Multi or Ecto.Changeset.prepare_changes/2.



	:on_replace - The action taken on associations when the record is
replaced when casting or manipulating parent changeset. May be
:raise (default), :mark_as_invalid, :nilify, or :delete.
See Ecto.Changeset's section about ":on_replace" for more info.



	:defaults - Default values to use when building the association.
It may be a keyword list of options that override the association schema
or a {module, function, args} that receive the struct and the owner as
arguments. For example, if you set Post.has_many :comments, defaults: [public: true],
then when using Ecto.build_assoc(post, :comments) that comment will have
comment.public == true. Alternatively, you can set it to
Post.has_many :comments, defaults: {__MODULE__, :update_comment, []}
and Post.update_comment(comment, post) will be invoked.



	:where - A filter for the association. See "Filtering associations" below.
It does not apply to :through associations.






  
  Examples



defmodule Post do
  use Ecto.Schema
  schema "posts" do
    has_many :comments, Comment
  end
end

# Get all comments for a given post
post = Repo.get(Post, 42)
comments = Repo.all assoc(post, :comments)

# The comments can come preloaded on the post struct
[post] = Repo.all(from(p in Post, where: p.id == 42, preload: :comments))
post.comments #=> [%Comment{...}, ...]

has_many can be used to define hierarchical relationships within a single
schema, for example threaded comments.

defmodule Comment do
  use Ecto.Schema
  schema "comments" do
    field :content, :string
    field :parent_id, :integer
    belongs_to :parent, Comment, foreign_key: :id, references: :parent_id, define_field: false
    has_many :children, Comment, foreign_key: :parent_id, references: :id
  end
end


  
  Filtering associations



It is possible to specify a :where option that will filter the records
returned by the association. Querying, joining or preloading the association
will use the given conditions as shown next:

defmodule Post do
  use Ecto.Schema

  schema "posts" do
    has_many :public_comments, Comment,
      where: [public: true]
  end
end

The :where option expects a keyword list where the key is an atom
representing the field and the value is either:


	nil - which specifies the field must be nil


	{:not, nil} - which specifies the field must not be nil


	{:in, list} - which specifies the field must be one of the values in a list


	{:fragment, expr} - which specifies a fragment string as the filter
(see Ecto.Query.API.fragment/1) with the field's value given to it
as the only argument


	or any other value which the field is compared directly against




Note the values above are distinctly different from the values you
would pass to where when building a query. For example, if you
attempt to build a query such as

from Post, where: [id: nil]

it will emit an error. This is because queries can be built dynamically,
and therefore passing nil can lead to security errors. However, the
:where values for an association are given at compile-time, which is
less dynamic and cannot leverage the full power of Ecto queries, which
explains why they have different APIs.

Important! Please use this feature only when strictly necessary,
otherwise it is very easy to end-up with large schemas with dozens of
different associations polluting your schema and affecting your
application performance. For instance, if you are using associations
only for different querying purposes, then it is preferable to build
and compose queries, rather than defining multiple associations:

posts
|> Ecto.assoc(:comments)
|> Comment.deleted()

Or when preloading:

from posts, preload: [comments: ^Comment.deleted()]


  
  has_many/has_one :through



Ecto also supports defining associations in terms of other associations
via the :through option. Let's see an example:

defmodule Post do
  use Ecto.Schema

  schema "posts" do
    has_many :comments, Comment
    has_one :permalink, Permalink

    # In the has_many :through example below, the `:comments`
    # in the list [:comments, :author] refers to the
    # `has_many :comments` in the Post own schema and the
    # `:author` refers to the `belongs_to :author` of the
    # Comment's schema (the module below).
    # (see the description below for more details)
    has_many :comments_authors, through: [:comments, :author]

    # Specify the association with custom source
    has_many :tags, {"posts_tags", Tag}
  end
end

defmodule Comment do
  use Ecto.Schema

  schema "comments" do
    belongs_to :author, Author
    belongs_to :post, Post
    has_one :post_permalink, through: [:post, :permalink]
  end
end

In the example above, we have defined a has_many :through association
named :comments_authors. A :through association always expects a list
and the first element of the list must be a previously defined association
in the current module. For example, :comments_authors first points to
:comments in the same module (Post), which then points to :author in
the next schema, Comment.

This :through association will return all authors for all comments
that belongs to that post:

# Get all comments authors for a given post
post = Repo.get(Post, 42)
authors = Repo.all assoc(post, :comments_authors)

Although we used the :through association in the example above, Ecto
also allows developers to dynamically build the through associations using
the Ecto.assoc/2 function:

assoc(post, [:comments, :author])

In fact, given :through associations are read-only, using the Ecto.assoc/2
format is the preferred mechanism for working with through associations. Use
the schema-based one only if you need to store the through data alongside of
the parent struct, in specific cases such as preloading.

:through associations can also be preloaded. In such cases, not only
the :through association is preloaded but all intermediate steps are
preloaded too:

[post] = Repo.all(from(p in Post, where: p.id == 42, preload: :comments_authors))
post.comments_authors #=> [%Author{...}, ...]

# The comments for each post will be preloaded too
post.comments #=> [%Comment{...}, ...]

# And the author for each comment too
hd(post.comments).author #=> %Author{...}

When the :through association is expected to return one or zero items,
has_one :through should be used instead, as in the example at the beginning
of this section:

# How we defined the association above
has_one :post_permalink, through: [:post, :permalink]

# Get a preloaded comment
[comment] = Repo.all(Comment) |> Repo.preload(:post_permalink)
comment.post_permalink #=> %Permalink{...}

  


    
  
    
      
      Link to this macro
    
    has_one(name, queryable, opts \\ [])

      
       
       View Source
     
      (macro)
  

  
Indicates a one-to-one association with another schema.

The current schema has zero or one records of the other schema. The other
schema often has a belongs_to field with the reverse association.


  
  Options




	:foreign_key - Sets the foreign key, this should map to a field on the
other schema, defaults to the underscored name of the current module
suffixed by _id



	:references  - Sets the key on the current schema to be used for the
association, defaults to the primary key on the schema



	:through - If this association must be defined in terms of existing
associations. Read the section in has_many/3 for more information



	:on_delete - The action taken on associations when parent record
is deleted. May be :nothing (default), :nilify_all and :delete_all.
Using this option is DISCOURAGED for most relational databases. Instead,
in your migration, set references(:parent_id, on_delete: :delete_all).
Opposite to the migration option, this option cannot guarantee integrity
and it is only triggered for Ecto.Repo.delete/2 (and not on
Ecto.Repo.delete_all/2) and it never cascades. If posts has many comments,
which has many tags, and you delete a post, only comments will be deleted.
If your database does not support references, cascading can be manually
implemented by using Ecto.Multi or Ecto.Changeset.prepare_changes/2



	:on_replace - The action taken on associations when the record is
replaced when casting or manipulating parent changeset. May be
:raise (default), :mark_as_invalid, :nilify, :update, or
:delete. See Ecto.Changeset's section on related data for more info.



	:defaults - Default values to use when building the association.
It may be a keyword list of options that override the association schema
or a {module, function, args} that receive the struct and the owner as
arguments. For example, if you set Post.has_many :comments, defaults: [public: true],
then when using Ecto.build_assoc(post, :comments) that comment will have
comment.public == true. Alternatively, you can set it to
Post.has_many :comments, defaults: {__MODULE__, :update_comment, []}
and Post.update_comment(comment, post) will be invoked.



	:where - A filter for the association. See "Filtering associations"
in has_many/3. It does not apply to :through associations.






  
  Examples



defmodule Post do
  use Ecto.Schema

  schema "posts" do
    has_one :permalink, Permalink

    # Specify the association with custom source
    has_one :category, {"posts_categories", Category}
  end
end

# The permalink can come preloaded on the post struct
[post] = Repo.all(from(p in Post, where: p.id == 42, preload: :permalink))
post.permalink #=> %Permalink{...}

  


    
  
    
      
      Link to this macro
    
    many_to_many(name, queryable, opts \\ [])

      
       
       View Source
     
      (macro)
  

  
Indicates a many-to-many association with another schema.

The association happens through a join schema or source, containing
foreign keys to the associated schemas. For example, the association
below:

# from MyApp.Post
many_to_many :tags, MyApp.Tag, join_through: "posts_tags"

is backed by relational databases through a join table as follows:

[Post] <-> [posts_tags] <-> [Tag]
  id   <--   post_id
              tag_id    -->  id

More information on the migration for creating such a schema is shown
below.


  
  Options




	:join_through - specifies the source of the associated data.
It may be a string, like "posts_tags", representing the
underlying storage table or an atom, like MyApp.PostTag,
representing a schema. This option is required.



	:join_keys - specifies how the schemas are associated. It
expects a keyword list with two entries, the first being how
the join table should reach the current schema and the second
how the join table should reach the associated schema. In the
example above, it defaults to: [post_id: :id, tag_id: :id].
The keys are inflected from the schema names.



	:on_delete - The action taken on associations when the parent record
is deleted. May be :nothing (default) or :delete_all.
Using this option is DISCOURAGED for most relational databases. Instead,
in your migration, set references(:parent_id, on_delete: :delete_all).
Opposite to the migration option, this option cannot guarantee integrity
and it is only triggered for Ecto.Repo.delete/2 (and not on
Ecto.Repo.delete_all/2). This option can only remove data from the
join source, never the associated records, and it never cascades.



	:on_replace - The action taken on associations when the record is
replaced when casting or manipulating parent changeset. May be
:raise (default), :mark_as_invalid, or :delete.
:delete will only remove data from the join source, never the
associated records. See Ecto.Changeset's section on related data
for more info.



	:defaults - Default values to use when building the association.
It may be a keyword list of options that override the association schema
or a {module, function, args} that receive the struct and the owner as
arguments. For example, if you set Post.has_many :comments, defaults: [public: true],
then when using Ecto.build_assoc(post, :comments) that comment will have
comment.public == true. Alternatively, you can set it to
Post.has_many :comments, defaults: {__MODULE__, :update_comment, []}
and Post.update_comment(comment, post) will be invoked.



	:join_defaults - The same as :defaults but it applies to the join schema
instead. This option will raise if it is given and the :join_through value
is not a schema.



	:unique - When true, checks if the associated entries are unique
whenever the association is cast or changed via the parent record.
For instance, it would verify that a given tag cannot be attached to
the same post more than once. This exists mostly as a quick check
for user feedback, as it does not guarantee uniqueness at the database
level. Therefore, you should also set a unique index in the database
join table, such as: create unique_index(:posts_tags, [:post_id, :tag_id])



	:where - A filter for the association. See "Filtering associations"
in has_many/3






  
  Removing data



If you attempt to remove associated many_to_many data, Ecto will
always remove data from the join schema and never from the target
associations be it by setting :on_replace to :delete, :on_delete
to :delete_all or by using changeset functions such as
Ecto.Changeset.put_assoc/3. For example, if a Post has a many to many
relationship with Tag, setting :on_delete to :delete_all will
only delete entries from the "posts_tags" table in case Post is
deleted.


  
  Migration



How your migration should be structured depends on the value you pass
in :join_through. If :join_through is simply a string, representing
a table, you may define a table without primary keys and you must not
include any further columns, as those values won't be set by Ecto:

create table(:posts_tags, primary_key: false) do
  add :post_id, references(:posts)
  add :tag_id, references(:tags)
end

However, if your :join_through is a schema, like MyApp.PostTag, your
join table may be structured as any other table in your codebase,
including timestamps:

create table(:posts_tags) do
  add :post_id, references(:posts)
  add :tag_id, references(:tags)
  timestamps()
end

Because :join_through contains a schema, in such cases, autogenerated
values and primary keys will be automatically handled by Ecto.


  
  Examples



defmodule Post do
  use Ecto.Schema
  schema "posts" do
    many_to_many :tags, Tag, join_through: "posts_tags"
  end
end

# Let's create a post and a tag
post = Repo.insert!(%Post{})
tag = Repo.insert!(%Tag{name: "introduction"})

# We can associate at any time post and tags together using changesets
post
|> Repo.preload(:tags) # Load existing data
|> Ecto.Changeset.change() # Build the changeset
|> Ecto.Changeset.put_assoc(:tags, [tag]) # Set the association
|> Repo.update!

# In a later moment, we may get all tags for a given post
post = Repo.get(Post, 42)
tags = Repo.all(assoc(post, :tags))

# The tags may also be preloaded on the post struct for reading
[post] = Repo.all(from(p in Post, where: p.id == 42, preload: :tags))
post.tags #=> [%Tag{...}, ...]


  
  Join Schema Example



You may prefer to use a join schema to handle many_to_many associations. The
decoupled nature of Ecto allows us to create a "join" struct which
belongs_to both sides of the many to many association.

In our example, a User has and belongs to many Organizations:

defmodule MyApp.Repo.Migrations.CreateUserOrganization do
  use Ecto.Migration

  def change do
    create table(:users_organizations) do
      add :user_id, references(:users)
      add :organization_id, references(:organizations)

      timestamps()
    end
  end
end

defmodule UserOrganization do
  use Ecto.Schema

  @primary_key false
  schema "users_organizations" do
    belongs_to :user, User
    belongs_to :organization, Organization
    timestamps() # Added bonus, a join schema will also allow you to set timestamps
  end

  def changeset(struct, params \\ %{}) do
    struct
    |> Ecto.Changeset.cast(params, [:user_id, :organization_id])
    |> Ecto.Changeset.validate_required([:user_id, :organization_id])
    # Maybe do some counter caching here!
  end
end

defmodule User do
  use Ecto.Schema

  schema "users" do
    many_to_many :organizations, Organization, join_through: UserOrganization
  end
end

defmodule Organization do
  use Ecto.Schema

  schema "organizations" do
    many_to_many :users, User, join_through: UserOrganization
  end
end

# Then to create the association, pass in the ID's of an existing
# User and Organization to UserOrganization.changeset
changeset = UserOrganization.changeset(%UserOrganization{}, %{user_id: id, organization_id: id})

case Repo.insert(changeset) do
  {:ok, assoc} -> # Assoc was created!
  {:error, changeset} -> # Handle the error
end

  


  
    
      
      Link to this macro
    
    schema(source, list)

      
       
       View Source
     
      (macro)
  

  
Defines a schema struct with a source name and field definitions.

An additional field called __meta__ is added to the struct for storing
internal Ecto state. This field always has a Ecto.Schema.Metadata struct
as value and can be manipulated with the Ecto.put_meta/2 function.

  


    
  
    
      
      Link to this macro
    
    timestamps(opts \\ [])

      
       
       View Source
     
      (macro)
  

  
Generates :inserted_at and :updated_at timestamp fields.

The fields generated by this macro will automatically be set to
the current time when inserting and updating values in a repository.


  
  Options




	:inserted_at - the ecto schema name of the field for insertion times or false


	:updated_at - the ecto schema name of the field for update times or false


	:inserted_at_source - the name of the database column for insertion times or false


	:updated_at_source - the name of the database column for update times or false


	:type - the timestamps type, defaults to :naive_datetime.


	:autogenerate - a module-function-args tuple used for generating
both inserted_at and updated_at timestamps




All options can be pre-configured by setting @timestamps_opts.

  

        

      
  

  
    
    Ecto.Schema.Metadata - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Schema.Metadata    



      
Stores metadata of a struct.

State

The state of the schema is stored in the :state field and allows
following values:


	:built - the struct was constructed in memory and is not persisted
to database yet;


	:loaded - the struct was loaded from database and represents
persisted data;


	:deleted - the struct was deleted and no longer represents persisted
data.




Source

The :source tracks the (table or collection) where the struct is or should
be persisted to.

Prefix

Tracks the source prefix in the data storage.

Context

The :context field represents additional state some databases require
for proper updates of data. It is not used by the built-in adapters of
Ecto.Adapters.Postgres and Ecto.Adapters.MySQL.

Schema

The :schema field refers the module name for the schema this metadata belongs to.

      Link to this type
    
    state()

      
       
       View Source
     
      
          state() :: :built | :loaded | :deleted

      

  

  
  


  
    
      
      Link to this type
    
    t()

      
       
       View Source
     
      
          t() :: %Ecto.Schema.Metadata{
  context: any(),
  prefix: Ecto.Schema.prefix(),
  schema: module(),
  source: Ecto.Schema.source(),
  state: state()
}

      

  

  
  

        

      
  

  
    
    Ecto.Type - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Type behaviour    



      
Defines functions and the Ecto.Type behaviour for implementing
custom types.

A custom type expects 4 functions to be implemented, all documented
and described below. We also provide two examples of how custom
types can be used in Ecto to augment existing types or providing
your own types.

Note: nil values are always bypassed and cannot be handled by
custom types.

Example

Imagine you want to store an URI struct as part of a schema in an
url-shortening service. There isn't an Ecto field type to support
that value at runtime, therefore a custom one is needed.

You also want to query not only by the full url, but for example
by specific ports used. This is possible by putting the URI data
into a map field instead of just storing the plain
string representation.

from s in ShortUrl,
  where: fragment("?->>? ILIKE ?", s.original_url, "port", "443")

So the custom type does need to handle the conversion from
external data to runtime data (cast/1) as well as
transforming that runtime data into the :map Ecto native type and
back (dump/1 and load/1).

defmodule EctoURI do
  use Ecto.Type
  def type, do: :map

  # Provide custom casting rules.
  # Cast strings into the URI struct to be used at runtime
  def cast(uri) when is_binary(uri) do
    {:ok, URI.parse(uri)}
  end

  # Accept casting of URI structs as well
  def cast(%URI{} = uri), do: {:ok, uri}

  # Everything else is a failure though
  def cast(_), do: :error

  # When loading data from the database, as long as it's a map,
  # we just put the data back into an URI struct to be stored in
  # the loaded schema struct.
  def load(data) when is_map(data) do
    data =
      for {key, val} <- data do
        {String.to_existing_atom(key), val}
      end
    {:ok, struct!(URI, data)}
  end

  # When dumping data to the database, we *expect* an URI struct
  # but any value could be inserted into the schema struct at runtime,
  # so we need to guard against them.
  def dump(%URI{} = uri), do: {:ok, Map.from_struct(uri)}
  def dump(_), do: :error
end

Now we can use our new field type above in our schemas:

defmodule ShortUrl do
  use Ecto.Schema

  schema "posts" do
    field :original_url, EctoURI
  end
end

    composite()
  




  
    custom()
  

    Custom types are represented by user-defined modules.





  
    primitive()
  

    Primitive Ecto types (handled by Ecto).





  
    t()
  

    An Ecto type, primitive or custom.




  

  
    
      Functions
    


  
    base?(atom)
  

    Checks if the given atom can be used as base type.





  
    cast(type, value)
  

    Casts a value to the given type.





  
    composite?(atom)
  

    Checks if the given atom can be used as composite type.





  
    dump(type, value)
  

    Dumps a value to the given type.





  
    dump(type, value, dumper)
  

    Dumps a value to the given type.





  
    embed_as(base, format)
  

    Gets how the type is treated inside embeds for the given format.





  
    embedded_dump(type, value, format)
  

    Dumps the value for type considering it will be embedded in format.





  
    embedded_load(type, value, format)
  

    Loads the value for type considering it was embedded in format.





  
    equal?(type, term1, term2)
  

    Checks if two terms are equal.





  
    load(type, value)
  

    Loads a value with the given type.





  
    load(type, value, loader)
  

    Loads a value with the given type.





  
    match?(schema_type, query_type)
  

    Checks if a given type matches with a primitive type
that can be found in queries.





  
    primitive?(base)
  

    Checks if we have a primitive type.





  
    type(type)
  

    Retrieves the underlying schema type for the given, possibly custom, type.




  

  
    
      Callbacks
    


  
    cast(term)
  

    Casts the given input to the custom type.





  
    dump(term)
  

    Dumps the given term into an Ecto native type.





  
    embed_as(format)
  

    Dictates how the type should be treated inside embeds.





  
    equal?(term, term)
  

    Checks if two terms are semantically equal.





  
    load(term)
  

    Loads the given term into a custom type.





  
    type()
  

    Returns the underlying schema type for the custom type.

      Link to this type
    
    base()

      
       
       View Source
     
      
          base() ::
  :integer
  | :float
  | :boolean
  | :string
  | :map
  | :binary
  | :decimal
  | :id
  | :binary_id
  | :utc_datetime
  | :naive_datetime
  | :date
  | :time
  | :any
  | :utc_datetime_usec
  | :naive_datetime_usec
  | :time_usec

      

  

  
  


  
    
      
      Link to this type
    
    composite()

      
       
       View Source
     
      
          composite() :: {:array, t()} | {:map, t()} | private_composite()

      

  

  
  


  
    
      
      Link to this type
    
    custom()

      
       
       View Source
     
      
          custom() :: module()

      

  

  
Custom types are represented by user-defined modules.

  


  
    
      
      Link to this type
    
    primitive()

      
       
       View Source
     
      
          primitive() :: base() | composite()

      

  

  
Primitive Ecto types (handled by Ecto).

  


  
    
      
      Link to this type
    
    t()

      
       
       View Source
     
      
          t() :: primitive() | custom()

      

  

  
An Ecto type, primitive or custom.

      Link to this function
    
    base?(atom)

      
       
       View Source
     
      
          base?(atom()) :: boolean()

      

  

  
Checks if the given atom can be used as base type.

iex> base?(:string)
true
iex> base?(:array)
false
iex> base?(Custom)
false

  


  
    
      
      Link to this function
    
    cast(type, value)

      
       
       View Source
     
      
          cast(t(), term()) :: {:ok, term()} | {:error, keyword()} | :error

      

  

  
Casts a value to the given type.

cast/2 is used by the finder queries and changesets to cast outside values to
specific types.

Note that nil can be cast to all primitive types as data stores allow nil to be
set on any column.

NaN and infinite decimals are not supported, use custom types instead.

iex> cast(:any, "whatever")
{:ok, "whatever"}

iex> cast(:any, nil)
{:ok, nil}
iex> cast(:string, nil)
{:ok, nil}

iex> cast(:integer, 1)
{:ok, 1}
iex> cast(:integer, "1")
{:ok, 1}
iex> cast(:integer, "1.0")
:error

iex> cast(:id, 1)
{:ok, 1}
iex> cast(:id, "1")
{:ok, 1}
iex> cast(:id, "1.0")
:error

iex> cast(:float, 1.0)
{:ok, 1.0}
iex> cast(:float, 1)
{:ok, 1.0}
iex> cast(:float, "1")
{:ok, 1.0}
iex> cast(:float, "1.0")
{:ok, 1.0}
iex> cast(:float, "1-foo")
:error

iex> cast(:boolean, true)
{:ok, true}
iex> cast(:boolean, false)
{:ok, false}
iex> cast(:boolean, "1")
{:ok, true}
iex> cast(:boolean, "0")
{:ok, false}
iex> cast(:boolean, "whatever")
:error

iex> cast(:string, "beef")
{:ok, "beef"}
iex> cast(:binary, "beef")
{:ok, "beef"}

iex> cast(:decimal, Decimal.new("1.0"))
{:ok, Decimal.new("1.0")}

iex> cast({:array, :integer}, [1, 2, 3])
{:ok, [1, 2, 3]}
iex> cast({:array, :integer}, ["1", "2", "3"])
{:ok, [1, 2, 3]}
iex> cast({:array, :string}, [1, 2, 3])
:error
iex> cast(:string, [1, 2, 3])
:error

  


  
    
      
      Link to this function
    
    composite?(atom)

      
       
       View Source
     
      
          composite?(atom()) :: boolean()

      

  

  
Checks if the given atom can be used as composite type.

iex> composite?(:array)
true
iex> composite?(:string)
false

  


  
    
      
      Link to this function
    
    dump(type, value)

      
       
       View Source
     
      
          dump(t(), term()) :: {:ok, term()} | :error

      

  

  
Dumps a value to the given type.

Opposite to casting, dumping requires the returned value
to be a valid Ecto type, as it will be sent to the
underlying data store.

iex> dump(:string, nil)
{:ok, nil}
iex> dump(:string, "foo")
{:ok, "foo"}

iex> dump(:integer, 1)
{:ok, 1}
iex> dump(:integer, "10")
:error

iex> dump(:binary, "foo")
{:ok, "foo"}
iex> dump(:binary, 1)
:error

iex> dump({:array, :integer}, [1, 2, 3])
{:ok, [1, 2, 3]}
iex> dump({:array, :integer}, [1, "2", 3])
:error
iex> dump({:array, :binary}, ["1", "2", "3"])
{:ok, ["1", "2", "3"]}

  


  
    
      
      Link to this function
    
    dump(type, value, dumper)

      
       
       View Source
     
      
          dump(t(), term(), (t(), term() -> {:ok, term()} | :error)) ::
  {:ok, term()} | :error

      

  

  
Dumps a value to the given type.

This function behaves the same as dump/2, except for composite types
the given dumper function is used.

  


  
    
      
      Link to this function
    
    embed_as(base, format)

      
       
       View Source
     
  

  
Gets how the type is treated inside embeds for the given format.

See embed_as/1.

  


  
    
      
      Link to this function
    
    embedded_dump(type, value, format)

      
       
       View Source
     
  

  
Dumps the value for type considering it will be embedded in format.


  
  Examples



iex> Ecto.Type.embedded_dump(:decimal, Decimal.new("1"), :json)
{:ok, Decimal.new("1")}

  


  
    
      
      Link to this function
    
    embedded_load(type, value, format)

      
       
       View Source
     
  

  
Loads the value for type considering it was embedded in format.


  
  Examples



iex> Ecto.Type.embedded_load(:decimal, "1", :json)
{:ok, Decimal.new("1")}

  


  
    
      
      Link to this function
    
    equal?(type, term1, term2)

      
       
       View Source
     
      
          equal?(t(), term(), term()) :: boolean()

      

  

  
Checks if two terms are equal.

Depending on the given type performs a structural or semantical comparison.


  
  Examples



iex> equal?(:integer, 1, 1)
true
iex> equal?(:decimal, Decimal.new("1"), Decimal.new("1.00"))
true

  


  
    
      
      Link to this function
    
    load(type, value)

      
       
       View Source
     
      
          load(t(), term()) :: {:ok, term()} | :error

      

  

  
Loads a value with the given type.

iex> load(:string, nil)
{:ok, nil}
iex> load(:string, "foo")
{:ok, "foo"}

iex> load(:integer, 1)
{:ok, 1}
iex> load(:integer, "10")
:error

  


  
    
      
      Link to this function
    
    load(type, value, loader)

      
       
       View Source
     
      
          load(t(), term(), (t(), term() -> {:ok, term()} | :error)) ::
  {:ok, term()} | :error

      

  

  
Loads a value with the given type.

This function behaves the same as load/2, except for composite types
the given loader function is used.

  


  
    
      
      Link to this function
    
    match?(schema_type, query_type)

      
       
       View Source
     
      
          match?(t(), primitive()) :: boolean()

      

  

  
Checks if a given type matches with a primitive type
that can be found in queries.

iex> match?(:string, :any)
true
iex> match?(:any, :string)
true
iex> match?(:string, :string)
true

iex> match?({:array, :string}, {:array, :any})
true

iex> match?(Ecto.UUID, :uuid)
true
iex> match?(Ecto.UUID, :string)
false

  


  
    
      
      Link to this function
    
    primitive?(base)

      
       
       View Source
     
      
          primitive?(t()) :: boolean()

      

  

  
Checks if we have a primitive type.

iex> primitive?(:string)
true
iex> primitive?(Another)
false

iex> primitive?({:array, :string})
true
iex> primitive?({:array, Another})
true

  


  
    
      
      Link to this function
    
    type(type)

      
       
       View Source
     
      
          type(t()) :: t()

      

  

  
Retrieves the underlying schema type for the given, possibly custom, type.

iex> type(:string)
:string
iex> type(Ecto.UUID)
:uuid

iex> type({:array, :string})
{:array, :string}
iex> type({:array, Ecto.UUID})
{:array, :uuid}

iex> type({:map, Ecto.UUID})
{:map, :uuid}

  

        

      
      
        
          
            
            Anchor for this section
          
Callbacks        

        

  
    
      
      Link to this callback
    
    cast(term)

      
       
       View Source
     
      
          cast(term()) :: {:ok, term()} | {:error, keyword()} | :error

      

  

  
Casts the given input to the custom type.

This callback is called on external input and can return any type,
as long as the dump/1 function is able to convert the returned
value into an Ecto native type. There are two situations where
this callback is called:


	When casting values by Ecto.Changeset


	When passing arguments to Ecto.Query




When returning {:error, keyword()}, the returned keyword list
will be preserved in the changeset errors, similar to
Changeset.add_error/4. Passing a :message key, will override
the default message. It is not possible to override the :type key.

For {:array, CustomType} or {:map, CustomType} the returned
keyword list will be erased and the default error will be shown.

  


  
    
      
      Link to this callback
    
    dump(term)

      
       
       View Source
     
      
          dump(term()) :: {:ok, term()} | :error

      

  

  
Dumps the given term into an Ecto native type.

This callback is called with any term that was stored in the struct
and it needs to validate them and convert it to an Ecto native type.

  


  
    
      
      Link to this callback
    
    embed_as(format)

      
       
       View Source
     
      
          embed_as(format :: atom()) :: :self | :dump

      

  

  
Dictates how the type should be treated inside embeds.

By default, the type is sent as itself, without calling
dumping to keep the higher level representation. But
it can be set to :dump to it is dumped before encoded.

  


  
    
      
      Link to this callback
    
    equal?(term, term)

      
       
       View Source
     
      
          equal?(term(), term()) :: boolean()

      

  

  
Checks if two terms are semantically equal.

  


  
    
      
      Link to this callback
    
    load(term)

      
       
       View Source
     
      
          load(term()) :: {:ok, term()} | :error

      

  

  
Loads the given term into a custom type.

This callback is called when loading data from the database and
receives an Ecto native type. It can return any type, as long as
the dump/1 function is able to convert the returned value back
into an Ecto native type.

  


  
    
      
      Link to this callback
    
    type()

      
       
       View Source
     
      
          type() :: t()

      

  

  
Returns the underlying schema type for the custom type.

For example, if you want to provide your own date
structures, the type function should return :date.

Note this function is not required to return Ecto primitive
types, the type is only required to be known by the adapter.

  

        

      
  

  
    
    Ecto.UUID - Ecto v3.3.2
    
    

    

  
  
    
Ecto.UUID    



      
An Ecto type for UUIDs strings.

      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Types
    


  
    raw()
  

    A raw binary representation of a UUID.





  
    t()
  

    A hex-encoded UUID string.




  

  
    
      Functions
    


  
    bingenerate()
  

    Generates a version 4 (random) UUID in the binary format.





  
    cast(binary)
  

    Casts to UUID.





  
    cast!(value)
  

    Same as cast/1 but raises Ecto.CastError on invalid arguments.





  
    dump(arg1)
  

    Converts a string representing a UUID into a binary.





  
    embed_as(_)
  

    Dictates how the type should be treated inside embeds.





  
    equal?(term1, term2)
  

    Checks if two terms are semantically equal.





  
    generate()
  

    Generates a version 4 (random) UUID.





  
    load(uuid)
  

    Converts a binary UUID into a string.




  

      

      
        
          
            
            Anchor for this section
          
Types        

        

  
    
      
      Link to this type
    
    raw()

      
       
       View Source
     
      
          raw() :: 
  
    
    Mix.Ecto - Ecto v3.3.2
    
    

    

  
  
    
Mix.Ecto    



      
Conveniences for writing Ecto related Mix tasks.

      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Functions
    


  
    ensure_implements(module, behaviour, message)
  

    Returns true if module implements behaviour.





  
    ensure_repo(repo, args)
  

    Ensures the given module is an Ecto.Repo.





  
    no_umbrella!(task)
  

    Gets a path relative to the application path.





  
    open?(file)
  

    Asks if the user wants to open a file based on ECTO_EDITOR.





  
    parse_repo(args)
  

    Parses the repository option from the given command line args list.




  

      

      
        
          
            
            Anchor for this section
          
Functions        

        

  
    
      
      Link to this function
    
    ensure_implements(module, behaviour, message)

      
       
       View Source
     
  

  
Returns true if module implements behaviour.

  


  
    
      
      Link to this function
    
    ensure_repo(repo, args)

      
       
       View Source
     
      
          ensure_repo(module(), list()) :: Ecto.Repo.t()

      

  

  
Ensures the given module is an Ecto.Repo.

  


  
    
      
      Link to this function
    
    no_umbrella!(task)

      
       
       View Source
     
  

  
Gets a path relative to the application path.

Raises on umbrella application.

  


  
    
      
      Link to this function
    
    open?(file)

      
       
       View Source
     
      
          open?(binary()) :: boolean()

      

  

  
Asks if the user wants to open a file based on ECTO_EDITOR.

  


  
    
      
      Link to this function
    
    parse_repo(args)

      
       
       View Source
     
      
          parse_repo([term()]) :: [Ecto.Repo.t()]

      

  

  
Parses the repository option from the given command line args list.

If no repo option is given, it is retrieved from the application environment.

  

        

      
  

  
    
    Ecto.Query.API - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Query.API    



      
Lists all functions allowed in the query API.


	Comparison operators: ==, !=, <=, >=, <, >


	Arithmetic operators: +, -, *, /


	Boolean operators: and, or, not


	Inclusion operator: in/2


	Search functions: like/2 and ilike/2


	Null check functions: is_nil/1


	Aggregates: count/0, count/1, avg/1, sum/1, min/1, max/1


	Date/time intervals: datetime_add/3, date_add/3, from_now/2, ago/2


	Inside select: struct/2, map/2, merge/2 and literals (map, tuples, lists, etc)


	General: fragment/1, field/2 and type/2




Note the functions in this module exist for documentation
purposes and one should never need to invoke them directly.
Furthermore, it is possible to define your own macros and
use them in Ecto queries (see docs for fragment/1).

Window API

Ecto also supports many of the windows functions found
in SQL databases. See Ecto.Query.WindowAPI for more
information.

About the arithmetic operators

The Ecto implementation of these operators provide only
a thin layer above the adapters. So if your adapter allows you
to use them in a certain way (like adding a date and an
interval in PostgreSQL), it should work just fine in Ecto
queries.

      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Functions
    


  
    left != right
  

    Binary != operation.





  
    left * right
  

    Binary * operation.





  
    left + right
  

    Binary + operation.





  
    left - right
  

    Binary - operation.





  
    left / right
  

    Binary / operation.





  
    left < right
  

    Binary < operation.





  
    left <= right
  

    Binary <= operation.





  
    left == right
  

    Binary == operation.





  
    left > right
  

    Binary > operation.





  
    left >= right
  

    Binary >= operation.





  
    ago(count, interval)
  

    Subtracts the given interval from the current time in UTC.





  
    left and right
  

    Binary and operation.





  
    avg(value)
  

    Calculates the average for the given entry.





  
    coalesce(value, expr)
  

    Takes whichever value is not null, or null if they both are.





  
    count()
  

    Counts the entries in the table.





  
    count(value)
  

    Counts the given entry.





  
    count(value, atom)
  

    Counts the distinct values in given entry.





  
    date_add(date, count, interval)
  

    Adds a given interval to a date.





  
    datetime_add(datetime, count, interval)
  

    Adds a given interval to a datetime.





  
    field(source, field)
  

    Allows a field to be dynamically accessed.





  
    filter(value, filter)
  

    Applies the given expression as a FILTER clause against an
aggregate. This is currently only supported by Postgres.





  
    fragment(fragments)
  

    Send fragments directly to the database.





  
    from_now(count, interval)
  

    Adds the given interval to the current time in UTC.





  
    ilike(string, search)
  

    Searches for search in string in a case insensitive fashion.





  
    left in right
  

    Checks if the left-value is included in the right one.





  
    is_nil(value)
  

    Checks if the given value is nil.





  
    like(string, search)
  

    Searches for search in string.





  
    map(source, fields)
  

    Used in select to specify which fields should be returned as a map.





  
    max(value)
  

    Calculates the maximum for the given entry.





  
    merge(left_map, right_map)
  

    Merges the map on the right over the map on the left.





  
    min(value)
  

    Calculates the minimum for the given entry.





  
    not(value)
  

    Unary not operation.





  
    left or right
  

    Binary or operation.





  
    struct(source, fields)
  

    Used in select to specify which struct fields should be returned.





  
    sum(value)
  

    Calculates the sum for the given entry.





  
    type(interpolated_value, type)
  

    Casts the given value to the given type at the database level.




  

      

      
        
          
            
            Anchor for this section
          
Functions        

        

  
    
      
      Link to this function
    
    left != right

      
       
       View Source
     
  

  
Binary != operation.

  


  
    
      
      Link to this function
    
    left * right

      
       
       View Source
     
  

  
Binary * operation.

  


  
    
      
      Link to this function
    
    left + right

      
       
       View Source
     
  

  
Binary + operation.

  


  
    
      
      Link to this function
    
    left - right

      
       
       View Source
     
  

  
Binary - operation.

  


  
    
      
      Link to this function
    
    left / right

      
       
       View Source
     
  

  
Binary / operation.

  


  
    
      
      Link to this function
    
    left < right

      
       
       View Source
     
  

  
Binary < operation.

  


  
    
      
      Link to this function
    
    left <= right

      
       
       View Source
     
  

  
Binary <= operation.

  


  
    
      
      Link to this function
    
    left == right

      
       
       View Source
     
  

  
Binary == operation.

  


  
    
      
      Link to this function
    
    left > right

      
       
       View Source
     
  

  
Binary > operation.

  


  
    
      
      Link to this function
    
    left >= right

      
       
       View Source
     
  

  
Binary >= operation.

  


  
    
      
      Link to this function
    
    ago(count, interval)

      
       
       View Source
     
  

  
Subtracts the given interval from the current time in UTC.

The current time in UTC is retrieved from Elixir and
not from the database.


  
  Examples



from p in Post, where: p.published_at > ago(3, "month")

  


  
    
      
      Link to this function
    
    left and right

      
       
       View Source
     
  

  
Binary and operation.

  


  
    
      
      Link to this function
    
    avg(value)

      
       
       View Source
     
  

  
Calculates the average for the given entry.

from p in Payment, select: avg(p.value)

  


  
    
      
      Link to this function
    
    coalesce(value, expr)

      
       
       View Source
     
  

  
Takes whichever value is not null, or null if they both are.

In SQL, COALESCE takes any number of arguments, but in ecto
it only takes two, so it must be chained to achieve the same
effect.

from p in Payment, select: p.value |> coalesce(p.backup_value) |> coalesce(0)

  


  
    
      
      Link to this function
    
    count()

      
       
       View Source
     
  

  
Counts the entries in the table.

from p in Post, select: count()

  


  
    
      
      Link to this function
    
    count(value)

      
       
       View Source
     
  

  
Counts the given entry.

from p in Post, select: count(p.id)

  


  
    
      
      Link to this function
    
    count(value, atom)

      
       
       View Source
     
  

  
Counts the distinct values in given entry.

from p in Post, select: count(p.id, :distinct)

  


  
    
      
      Link to this function
    
    date_add(date, count, interval)

      
       
       View Source
     
  

  
Adds a given interval to a date.

See datetime_add/3 for more information.

  


  
    
      
      Link to this function
    
    datetime_add(datetime, count, interval)

      
       
       View Source
     
  

  
Adds a given interval to a datetime.

The first argument is a datetime, the second one is the count
for the interval, which may be either positive or negative and
the interval value:

# Get all items published since the last month
from p in Post, where: p.published_at >
                       datetime_add(^NaiveDateTime.utc_now(), -1, "month")

In the example above, we used datetime_add/3 to subtract one month
from the current datetime and compared it with the p.published_at.
If you want to perform operations on date, date_add/3 could be used.

The following intervals are supported: year, month, week, day, hour,
minute, second, millisecond and microsecond.

  


  
    
      
      Link to this function
    
    field(source, field)

      
       
       View Source
     
  

  
Allows a field to be dynamically accessed.

def at_least_four(doors_or_tires) do
  from c in Car,
    where: field(c, ^doors_or_tires) >= 4
end

In the example above, both at_least_four(:doors) and at_least_four(:tires)
would be valid calls as the field is dynamically generated.

  


  
    
      
      Link to this function
    
    filter(value, filter)

      
       
       View Source
     
  

  
Applies the given expression as a FILTER clause against an
aggregate. This is currently only supported by Postgres.

from p in Payment, select: filter(avg(p.value), p.value > 0 and p.value < 100)

from p in Payment, select: avg(p.value) |> filter(p.value < 0)

  


  
    
      
      Link to this function
    
    fragment(fragments)

      
       
       View Source
     
  

  
Send fragments directly to the database.

It is not possible to represent all possible database queries using
Ecto's query syntax. When such is required, it is possible to use
fragments to send any expression to the database:

def unpublished_by_title(title) do
  from p in Post,
    where: is_nil(p.published_at) and
           fragment("lower(?)", p.title) == ^title
end

Every occurence of the ? character will be interpreted as a place
for additional argument. If the literal character ? is required,
it can be escaped with \\? (one escape for strings, another for
fragment).

In the example above, we are using the lower procedure in the
database to downcase the title column.

It is very important to keep in mind that Ecto is unable to do any
type casting described above when fragments are used. You can
however use the type/2 function to give Ecto some hints:

fragment("lower(?)", p.title) == type(^title, :string)

Or even say the right side is of the same type as p.title:

fragment("lower(?)", p.title) == type(^title, p.title)

It is possible to make use of PostgreSQL's JSON/JSONB data type
with fragments, as well:

fragment("?->>? ILIKE ?", p.map, "key_name", ^some_value)


  
  Keyword fragments



In order to support databases that do not have string-based
queries, like MongoDB, fragments also allow keywords to be given:

from p in Post,
    where: fragment(title: ["$eq": ^some_value])


  
  Defining custom functions using macros and fragment



You can add a custom Ecto query function using macros.  For example
to expose SQL's coalesce function you can define this macro:

defmodule CustomFunctions do
  defmacro coalesce(left, right) do
    quote do
      fragment("coalesce(?, ?)", unquote(left), unquote(right))
    end
  end
end

To have coalesce/2 available, just import the module that defines it.

import CustomFunctions

The only downside is that it will show up as a fragment when
inspecting the Elixir query.  Other than that, it should be
equivalent to a built-in Ecto query function.

  


  
    
      
      Link to this function
    
    from_now(count, interval)

      
       
       View Source
     
  

  
Adds the given interval to the current time in UTC.

The current time in UTC is retrieved from Elixir and
not from the database.


  
  Examples



from a in Account, where: a.expires_at < from_now(3, "month")

  


  
    
      
      Link to this function
    
    ilike(string, search)

      
       
       View Source
     
  

  
Searches for search in string in a case insensitive fashion.

from p in Post, where: ilike(p.body, "Chapter%")

Translates to the underlying SQL ILIKE query. This operation is
only available on PostgreSQL.

  


  
    
      
      Link to this function
    
    left in right

      
       
       View Source
     
  

  
Checks if the left-value is included in the right one.

from p in Post, where: p.id in [1, 2, 3]

The right side may either be a list, a literal list
or even a column in the database with array type:

from p in Post, where: "elixir" in p.tags

  


  
    
      
      Link to this function
    
    is_nil(value)

      
       
       View Source
     
  

  
Checks if the given value is nil.

from p in Post, where: is_nil(p.published_at)

To check if a given value is not nil use:

from p in Post, where: not is_nil(p.published_at)

  


  
    
      
      Link to this function
    
    like(string, search)

      
       
       View Source
     
  

  
Searches for search in string.

from p in Post, where: like(p.body, "Chapter%")

Translates to the underlying SQL LIKE query, therefore
its behaviour is dependent on the database. In particular,
PostgreSQL will do a case-sensitive operation, while the
majority of other databases will be case-insensitive. For
performing a case-insensitive like in PostgreSQL, see ilike/2.

You should be very careful when allowing user sent data to be used
as part of LIKE query, since they allow to perform
LIKE-injections.

  


  
    
      
      Link to this function
    
    map(source, fields)

      
       
       View Source
     
  

  
Used in select to specify which fields should be returned as a map.

For example, if you don't need all fields to be returned or
neither need a struct, you can use map/2 to achieve both:

from p in Post,
  select: map(p, [:title, :body])

map/2 can also be used to dynamically select fields:

fields = [:title, :body]
from p in Post, select: map(p, ^fields)

If the same source is selected multiple times with a map,
the fields are merged in order to avoid fetching multiple copies
from the database. In other words, the expression below:

from(city in City, preload: :country,
     select: {map(city, [:country_id]), map(city, [:name])}

is expanded to:

from(city in City, preload: :country,
     select: {map(city, [:country_id, :name]), map(city, [:country_id, :name])}

For preloads, the selected fields may be specified from the parent:

from(city in City, preload: :country,
     select: map(city, [:country_id, :name, country: [:id, :population]]))

IMPORTANT: When filtering fields for associations, you
MUST include the foreign keys used in the relationship,
otherwise Ecto will be unable to find associated records.

  


  
    
      
      Link to this function
    
    max(value)

      
       
       View Source
     
  

  
Calculates the maximum for the given entry.

from p in Payment, select: max(p.value)

  


  
    
      
      Link to this function
    
    merge(left_map, right_map)

      
       
       View Source
     
  

  
Merges the map on the right over the map on the left.

If the map on the left side is a struct, Ecto will check
all of the field on the right previously exist on the left
before merging.

from(city in City, select: merge(city, %{virtual_field: "some_value"}))

This function is primarily used by Ecto.Query.select_merge/3
to merge different select clauses.

  


  
    
      
      Link to this function
    
    min(value)

      
       
       View Source
     
  

  
Calculates the minimum for the given entry.

from p in Payment, select: min(p.value)

  


  
    
      
      Link to this function
    
    not(value)

      
       
       View Source
     
  

  
Unary not operation.

  


  
    
      
      Link to this function
    
    left or right

      
       
       View Source
     
  

  
Binary or operation.

  


  
    
      
      Link to this function
    
    struct(source, fields)

      
       
       View Source
     
  

  
Used in select to specify which struct fields should be returned.

For example, if you don't need all fields to be returned
as part of a struct, you can filter it to include only certain
fields by using struct/2:

from p in Post,
  select: struct(p, [:title, :body])

struct/2 can also be used to dynamically select fields:

fields = [:title, :body]
from p in Post, select: struct(p, ^fields)

As a convenience, select allows developers to take fields
without an explicit call to struct/2:

from p in Post, select: [:title, :body]

Or even dynamically:

fields = [:title, :body]
from p in Post, select: ^fields

For preloads, the selected fields may be specified from the parent:

from(city in City, preload: :country,
     select: struct(city, [:country_id, :name, country: [:id, :population]]))

If the same source is selected multiple times with a struct,
the fields are merged in order to avoid fetching multiple copies
from the database. In other words, the expression below:

from(city in City, preload: :country,
     select: {struct(city, [:country_id]), struct(city, [:name])}

is expanded to:

from(city in City, preload: :country,
     select: {struct(city, [:country_id, :name]), struct(city, [:country_id, :name])}

IMPORTANT: When filtering fields for associations, you
MUST include the foreign keys used in the relationship,
otherwise Ecto will be unable to find associated records.

  


  
    
      
      Link to this function
    
    sum(value)

      
       
       View Source
     
  

  
Calculates the sum for the given entry.

from p in Payment, select: sum(p.value)

  


  
    
      
      Link to this function
    
    type(interpolated_value, type)

      
       
       View Source
     
  

  
Casts the given value to the given type at the database level.

Most of the times, Ecto is able to proper cast interpolated
values due to its type checking mechanism. In some situations
though, you may want to tell Ecto that a parameter has some
particular type:

type(^title, :string)

It is also possible to say the type must match the same of a column:

type(^title, p.title)

Ecto will ensure ^title is cast to the given type and enforce such
type at the database level. If the value is returned in a select,
Ecto will also enforce the proper type throughout.

When performing arithmetic operations, type/2 can be used to cast
all the parameters in the operation to the same type:

from p in Post,
  select: type(p.visits + ^a_float + ^a_integer, :decimal)

Inside select, type/2 can also be used to cast fragments:

type(fragment("NOW"), :naive_datetime)

Or to type fields from schemaless queries:

from p in "posts", select: type(p.cost, :decimal)

Or to type aggregation results:

from p in Post, select: type(avg(p.cost), :integer)
from p in Post, select: type(filter(avg(p.cost), p.cost > 0), :integer)

  

        

      
  

  
    
    Ecto.Query.WindowAPI - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Query.WindowAPI    



      
Lists all windows functions.

Windows functions must always be used as the first argument
of over/2 where the second argument is the name of a window:

from e in Employee,
  select: {e.depname, e.empno, e.salary, over(avg(e.salary), :department)},
  windows: [department: [partition_by: e.depname]]

In the example above, we get the average salary per department.
:department is the window name, partitioned by e.depname
and avg/1 is the window function.

However, note that defining a window is not necessary, as the
window definition can be given as the second argument to over:

from e in Employee,
  select: {e.depname, e.empno, e.salary, over(avg(e.salary), partition_by: e.depname)}

Both queries are equivalent. However, if you are using the same
partitioning over and over again, defining a window will reduce
the query size. See Ecto.Query.windows/3 for all possible window
expressions, such as :partition_by and :order_by.

      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Functions
    


  
    avg(value)
  

    Calculates the average for the given entry.





  
    count()
  

    Counts the entries in the table.





  
    count(value)
  

    Counts the given entry.





  
    cume_dist()
  

    Returns relative rank of the current row:
(number of rows preceding or peer with current row) / (total rows).





  
    dense_rank()
  

    Returns rank of the current row without gaps; this function counts peer groups.





  
    filter(value, filter)
  

    Applies the given expression as a FILTER clause against an
aggregate. This is currently only supported by Postgres.





  
    first_value(value)
  

    Returns value evaluated at the row that is the first row of the window frame.





  
    lag(value, offset \\ 1, default \\ nil)
  

    Returns value evaluated at the row that is offset rows before
the current row within the partition.





  
    last_value(value)
  

    Returns value evaluated at the row that is the last row of the window frame.





  
    lead(value, offset \\ 1, default \\ nil)
  

    Returns value evaluated at the row that is offset rows after
the current row within the partition.





  
    max(value)
  

    Calculates the maximum for the given entry.





  
    min(value)
  

    Calculates the minimum for the given entry.





  
    nth_value(value, nth)
  

    Returns value evaluated at the row that is the nth row of the window
frame (counting from 1); nil if no such row.





  
    ntile(num_buckets)
  

    Returns integer ranging from 1 to the argument value, dividing the partition as equally as possible.





  
    percent_rank()
  

    Returns relative rank of the current row: (rank - 1) / (total rows - 1).





  
    rank()
  

    Returns rank of the current row with gaps; same as row_number/0 of its first peer.





  
    row_number()
  

    Returns number of the current row within its partition, counting from 1.





  
    sum(value)
  

    Calculates the sum for the given entry.




  

      

      
        
          
            
            Anchor for this section
          
Functions        

        

  
    
      
      Link to this function
    
    avg(value)

      
       
       View Source
     
  

  
Calculates the average for the given entry.

from p in Payment, select: avg(p.value)

  


  
    
      
      Link to this function
    
    count()

      
       
       View Source
     
  

  
Counts the entries in the table.

from p in Post, select: count()

  


  
    
      
      Link to this function
    
    count(value)

      
       
       View Source
     
  

  
Counts the given entry.

from p in Post, select: count(p.id)

  


  
    
      
      Link to this function
    
    cume_dist()

      
       
       View Source
     
  

  
Returns relative rank of the current row:
(number of rows preceding or peer with current row) / (total rows).

from p in Post,
     select: cume_dist() |> over(partition_by: p.category_id, order_by: p.date)

Note that this function must be invoked using window function syntax.

  


  
    
      
      Link to this function
    
    dense_rank()

      
       
       View Source
     
  

  
Returns rank of the current row without gaps; this function counts peer groups.

from p in Post,
     select: dense_rank() |> over(partition_by: p.category_id, order_by: p.date)

Note that this function must be invoked using window function syntax.

  


  
    
      
      Link to this function
    
    filter(value, filter)

      
       
       View Source
     
  

  
Applies the given expression as a FILTER clause against an
aggregate. This is currently only supported by Postgres.

from p in Post,
     select: avg(p.value)
             |> filter(p.value > 0 and p.value < 100)
             |> over(partition_by: p.category_id, order_by: p.date)

  


  
    
      
      Link to this function
    
    first_value(value)

      
       
       View Source
     
  

  
Returns value evaluated at the row that is the first row of the window frame.

from p in Post,
     select: first_value(p.id) |> over(partition_by: p.category_id, order_by: p.date)

Note that this function must be invoked using window function syntax.

  


    
    
  
    
      
      Link to this function
    
    lag(value, offset \\ 1, default \\ nil)

      
       
       View Source
     
  

  
Returns value evaluated at the row that is offset rows before
the current row within the partition.

If there is no such row, instead return default (which must be of the
same type as value). Both offset and default are evaluated with respect
to the current row. If omitted, offset defaults to 1 and default to nil.

from e in Events,
     windows: [w: [partition_by: e.name, order_by: e.tick]],
     select: {
       e.tick,
       e.action,
       e.name,
       lag(e.action) |> over(:w), # previous_action
       lead(e.action) |> over(:w) # next_action
     }

Note that this function must be invoked using window function syntax.

  


  
    
      
      Link to this function
    
    last_value(value)

      
       
       View Source
     
  

  
Returns value evaluated at the row that is the last row of the window frame.

from p in Post,
     select: last_value(p.id) |> over(partition_by: p.category_id, order_by: p.date)

Note that this function must be invoked using window function syntax.

  


    
    
  
    
      
      Link to this function
    
    lead(value, offset \\ 1, default \\ nil)

      
       
       View Source
     
  

  
Returns value evaluated at the row that is offset rows after
the current row within the partition.

If there is no such row, instead return default (which must be of the
same type as value). Both offset and default are evaluated with respect
to the current row. If omitted, offset defaults to 1 and default to nil.

from e in Events,
     windows: [w: [partition_by: e.name, order_by: e.tick],
     select: {
       e.tick,
       e.action,
       e.name,
       lag(e.action) |> over(:w), # previous_action
       lead(e.action) |> over(:w) # next_action
     }

Note that this function must be invoked using window function syntax.

  


  
    
      
      Link to this function
    
    max(value)

      
       
       View Source
     
  

  
Calculates the maximum for the given entry.

from p in Payment, select: max(p.value)

  


  
    
      
      Link to this function
    
    min(value)

      
       
       View Source
     
  

  
Calculates the minimum for the given entry.

from p in Payment, select: min(p.value)

  


  
    
      
      Link to this function
    
    nth_value(value, nth)

      
       
       View Source
     
  

  
Returns value evaluated at the row that is the nth row of the window
frame (counting from 1); nil if no such row.

from p in Post,
     select: nth_value(p.id, 4) |> over(partition_by: p.category_id, order_by: p.date)

Note that this function must be invoked using window function syntax.

  


  
    
      
      Link to this function
    
    ntile(num_buckets)

      
       
       View Source
     
  

  
Returns integer ranging from 1 to the argument value, dividing the partition as equally as possible.

from p in Post,
     select: ntile(10) |> over(partition_by: p.category_id, order_by: p.date)

Note that this function must be invoked using window function syntax.

  


  
    
      
      Link to this function
    
    percent_rank()

      
       
       View Source
     
  

  
Returns relative rank of the current row: (rank - 1) / (total rows - 1).

from p in Post,
     select: percent_rank() |> over(partition_by: p.category_id, order_by: p.date)

Note that this function must be invoked using window function syntax.

  


  
    
      
      Link to this function
    
    rank()

      
       
       View Source
     
  

  
Returns rank of the current row with gaps; same as row_number/0 of its first peer.

from p in Post,
     select: rank() |> over(partition_by: p.category_id, order_by: p.date)

Note that this function must be invoked using window function syntax.

  


  
    
      
      Link to this function
    
    row_number()

      
       
       View Source
     
  

  
Returns number of the current row within its partition, counting from 1.

from p in Post,
     select: row_number() |> over(partition_by: p.category_id, order_by: p.date)

Note that this function must be invoked using window function syntax.

  


  
    
      
      Link to this function
    
    sum(value)

      
       
       View Source
     
  

  
Calculates the sum for the given entry.

from p in Payment, select: sum(p.value)

  

        

      
  

  
    
    Ecto.Queryable - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Queryable protocol    



      
Converts a data structure into an Ecto.Query.

      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Types
    


  
    t()
  



  

  
    
      Functions
    


  
    to_query(data)
  

    Converts the given data into an Ecto.Query.




  

      

      
        
          
            
            Anchor for this section
          
Types        

        

  
    
      
      Link to this type
    
    t()

      
       
       View Source
     
      
          t() :: term()

      

  

  
  

        

      
      
        
          
            
            Anchor for this section
          
Functions        

        

  
    
      
      Link to this function
    
    to_query(data)

      
       
       View Source
     
  

  
Converts the given data into an Ecto.Query.

  

        

      
  

  
    
    Ecto.SubQuery - Ecto v3.3.2
    
    

    

  
  
    
Ecto.SubQuery    



      
A struct representing subqueries.

See Ecto.Query.subquery/2 for more information.

      


  

  
    
    Ecto.Adapter - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Adapter behaviour    



      
Specifies the minimal API required from adapters.

      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Types
    


  
    adapter_meta()
  

    The metadata returned by the adapter init/1.





  
    t()
  



  

  
    
      Functions
    


  
    lookup_meta(repo_name_or_pid)
  

    Returns the adapter metadata from the init/1 callback.




  

  
    
      Callbacks
    


  
    __before_compile__(env)
  

    The callback invoked in case the adapter needs to inject code.





  
    checkout(adapter_meta, config, function)
  

    Checks out a connection for the duration of the given function.





  
    dumpers(primitive_type, ecto_type)
  

    Returns the dumpers for a given type.





  
    ensure_all_started(config, type)
  

    Ensure all applications necessary to run the adapter are started.





  
    init(config)
  

    Initializes the adapter supervision tree by returning the children and adapter metadata.





  
    loaders(primitive_type, ecto_type)
  

    Returns the loaders for a given type.




  

      

      
        
          
            
            Anchor for this section
          
Types        

        

  
    
      
      Link to this type
    
    adapter_meta()

      
       
       View Source
     
      
          adapter_meta() :: map()

      

  

  
The metadata returned by the adapter init/1.

It must be a map and Ecto itself will always inject
two keys into the meta:


	the :cache key, which as ETS table that can be used as a cache (if available)


	the :pid key, which is the PID returned by the child spec returned in init/1




  


  
    
      
      Link to this type
    
    t()

      
       
       View Source
     
      
          t() :: module()

      

  

  
  

        

      
      
        
          
            
            Anchor for this section
          
Functions        

        

  
    
      
      Link to this function
    
    lookup_meta(repo_name_or_pid)

      
       
       View Source
     
  

  
Returns the adapter metadata from the init/1 callback.

It expects a name or a PID representing a repo.

  

        

      
      
        
          
            
            Anchor for this section
          
Callbacks        

        

  
    
      
      Link to this macrocallback
    
    __before_compile__(env)

      
       
       View Source
     
      
          __before_compile__(term(), env :: Macro.Env.t()) :: Macro.t()

      

  

  
The callback invoked in case the adapter needs to inject code.

  


  
    
      
      Link to this callback
    
    checkout(adapter_meta, config, function)

      
       
       View Source
     
      
          checkout(adapter_meta(), config :: Keyword.t(), (() -> result)) :: result
when result: var

      

  

  
Checks out a connection for the duration of the given function.

In case the adapter provides a pool, this guarantees all of the code
inside the given fun runs against the same connection, which
might improve performance by for instance allowing multiple related
calls to the datastore to share cache information:

Repo.checkout(fn ->
  for _ <- 100 do
    Repo.insert!(%Post{})
  end
end)

If the adapter does not provide a pool, just calling the passed function
and returning its result are enough.

If the adapter provides a pool, it is supposed to "check out" one of the
pool connections for the duration of the function call. Which connection
is checked out is not passed to the calling function, so it should be done
using a stateful method like using the current process' dictionary, process
tracking, or some kind of other lookup method. Make sure that this stored
connection is then used in the other callbacks implementations, such as
Ecto.Adapter.Queryable and Ecto.Adapter.Schema.

  


  
    
      
      Link to this callback
    
    dumpers(primitive_type, ecto_type)

      
       
       View Source
     
      
          dumpers(primitive_type :: Ecto.Type.primitive(), ecto_type :: Ecto.Type.t()) ::
  [(term() -> {:ok, term()} | :error) | Ecto.Type.t()]

      

  

  
Returns the dumpers for a given type.

It receives the primitive type and the Ecto type (which may be
primitive as well). It returns a list of dumpers with the given
type usually at the beginning.

This allows developers to properly translate values coming from
the Ecto into adapter ones. For example, if the database does not
support booleans but instead returns 0 and 1 for them, you could
add:

def dumpers(:boolean, type), do: [type, &bool_encode/1]
def dumpers(_primitive, type), do: [type]

defp bool_encode(false), do: {:ok, 0}
defp bool_encode(true), do: {:ok, 1}

All adapters are required to implement a clause for :binary_id types,
since they are adapter specific. If your adapter does not provide
binary ids, you may simply use Ecto.UUID:

def dumpers(:binary_id, type), do: [type, Ecto.UUID]
def dumpers(_primitive, type), do: [type]

  


  
    
      
      Link to this callback
    
    ensure_all_started(config, type)

      
       
       View Source
     
      
          ensure_all_started(config :: Keyword.t(), type :: :application.restart_type()) ::
  {:ok, [atom()]} | {:error, atom()}

      

  

  
Ensure all applications necessary to run the adapter are started.

  


  
    
      
      Link to this callback
    
    init(config)

      
       
       View Source
     
      
          init(config :: Keyword.t()) :: {:ok, :supervisor.child_spec(), adapter_meta()}

      

  

  
Initializes the adapter supervision tree by returning the children and adapter metadata.

  


  
    
      
      Link to this callback
    
    loaders(primitive_type, ecto_type)

      
       
       View Source
     
      
          loaders(primitive_type :: Ecto.Type.primitive(), ecto_type :: Ecto.Type.t()) ::
  [(term() -> {:ok, term()} | :error) | Ecto.Type.t()]

      

  

  
Returns the loaders for a given type.

It receives the primitive type and the Ecto type (which may be
primitive as well). It returns a list of loaders with the given
type usually at the end.

This allows developers to properly translate values coming from
the adapters into Ecto ones. For example, if the database does not
support booleans but instead returns 0 and 1 for them, you could
add:

def loaders(:boolean, type), do: [&bool_decode/1, type]
def loaders(_primitive, type), do: [type]

defp bool_decode(0), do: {:ok, false}
defp bool_decode(1), do: {:ok, true}

All adapters are required to implement a clause for :binary_id types,
since they are adapter specific. If your adapter does not provide binary
ids, you may simply use Ecto.UUID:

def loaders(:binary_id, type), do: [Ecto.UUID, type]
def loaders(_primitive, type), do: [type]

  

        

      
  

  
    
    Ecto.Adapter.Queryable - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Adapter.Queryable behaviour    



      
Specifies the query API required from adapters.

If your adapter is only able to respond to one or a couple of the query functions,
add custom implementations of those functions directly to the Repo
by using Ecto.Adapter.__before_compile__/1 instead.

      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Types
    


  
    adapter_meta()
  

    Proxy type to the adapter meta





  
    cached()
  




  
    options()
  




  
    prepared()
  




  
    query_cache()
  

    Cache query metadata





  
    query_meta()
  

    Ecto.Query metadata fields (stored in cache)




  

  
    
      Functions
    


  
    plan_query(operation, adapter, queryable)
  

    Plans a query using the given adapter.





  
    prepare_query(operation, repo_name_or_pid, queryable)
  

    Plans and prepares a query for the given repo, leveraging its query cache.




  

  
    
      Callbacks
    


  
    execute(adapter_meta, query_meta, query_cache, params, options)
  

    Executes a previously prepared query.





  
    prepare(atom, query)
  

    Commands invoked to prepare a query for Ecto.Repo.all/2, Ecto.Repo.update_all/3,
and Ecto.Repo.delete_all/2.





  
    stream(adapter_meta, query_meta, query_cache, params, options)
  

    Streams a previously prepared query.




  

      

      
        
          
            
            Anchor for this section
          
Types        

        

  
    
      
      Link to this type
    
    adapter_meta()

      
       
       View Source
     
      
          adapter_meta() :: Ecto.Adapter.adapter_meta()

      

  

  
Proxy type to the adapter meta

  


  
    
      
      Link to this type
    
    cached()

      
       
       View Source
     
      
          cached() :: term()

      

  

  
  


  
    
      
      Link to this type
    
    options()

      
       
       View Source
     
      
          options() :: Keyword.t()

      

  

  
  


  
    
      
      Link to this type
    
    prepared()

      
       
       View Source
     
      
          prepared() :: term()

      

  

  
  


  
    
      
      Link to this type
    
    query_cache()

      
       
       View Source
     
      
          query_cache() ::
  {:nocache, prepared()}
  | {:cache, (cached() -> :ok), prepared()}
  | {:cached, (cached() -> :ok), (prepared() -> :ok), cached()}

      

  

  
Cache query metadata

  


  
    
      
      Link to this type
    
    query_meta()

      
       
       View Source
     
      
          query_meta() :: %{sources: tuple(), preloads: term(), select: map()}

      

  

  
Ecto.Query metadata fields (stored in cache)

Functions        

        

  
    
      
      Link to this function
    
    plan_query(operation, adapter, queryable)

      
       
       View Source
     
  

  
Plans a query using the given adapter.

This does not expect the repository and therefore does not leverage the cache.

  


  
    
      
      Link to this function
    
    prepare_query(operation, repo_name_or_pid, queryable)

      
       
       View Source
     
  

  
Plans and prepares a query for the given repo, leveraging its query cache.

This operation uses the query cache if one is available.

Callbacks        

        

  
    
      
      Link to this callback
    
    execute(adapter_meta, query_meta, query_cache, params, options)

      
       
       View Source
     
      
          execute(
  adapter_meta(),
  query_meta(),
  query_cache(),
  params :: list(),
  options()
) :: {integer(), [[term()]] | nil}

      

  

  
Executes a previously prepared query.

It must return a tuple containing the number of entries and
the result set as a list of lists. The result set may also be
nil if a particular operation does not support them.

The adapter_meta field is a map containing some of the fields found
in the Ecto.Query struct.

  


  
    
      
      Link to this callback
    
    prepare(atom, query)

      
       
       View Source
     
      
          prepare(atom :: :all | :update_all | :delete_all, query :: Ecto.Query.t()) ::
  {:cache, prepared()} | {:nocache, prepared()}

      

  

  
Commands invoked to prepare a query for Ecto.Repo.all/2, Ecto.Repo.update_all/3,
and Ecto.Repo.delete_all/2.

The returned result is given to execute/5.

  


  
    
      
      Link to this callback
    
    stream(adapter_meta, query_meta, query_cache, params, options)

      
       
       View Source
     
      
          stream(adapter_meta(), query_meta(), query_cache(), params :: list(), options()) ::
  Enumerable.t()

      

  

  
Streams a previously prepared query.

It returns a stream of values.

The adapter_meta field is a map containing some of the fields found
in the Ecto.Query struct.

  

        

      
  

  
    
    Ecto.Adapter.Schema - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Adapter.Schema behaviour    



      
Specifies the schema API required from adapters.

      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Types
    


  
    adapter_meta()
  

    Proxy type to the adapter meta





  
    constraints()
  




  
    fields()
  




  
    filters()
  




  
    on_conflict()
  




  
    options()
  




  
    returning()
  




  
    schema_meta()
  

    Ecto.Schema metadata fields




  

  
    
      Callbacks
    


  
    autogenerate(field_type)
  

    Called to autogenerate a value for id/embed_id/binary_id.





  
    delete(adapter_meta, schema_meta, filters, options)
  

    Deletes a single struct with the given filters.





  
    insert(adapter_meta, schema_meta, fields, on_conflict, returning, options)
  

    Inserts a single new struct in the data store.





  
    insert_all(adapter_meta, schema_meta, header, list, on_conflict, returning, options)
  

    Inserts multiple entries into the data store.





  
    update(adapter_meta, schema_meta, fields, filters, returning, options)
  

    Updates a single struct with the given filters.




  

      

      
        
          
            
            Anchor for this section
          
Types        

        

  
    
      
      Link to this type
    
    adapter_meta()

      
       
       View Source
     
      
          adapter_meta() :: Ecto.Adapter.adapter_meta()

      

  

  
Proxy type to the adapter meta

  


  
    
      
      Link to this type
    
    constraints()

      
       
       View Source
     
      
          constraints() :: Keyword.t()

      

  

  
  


  
    
      
      Link to this type
    
    fields()

      
       
       View Source
     
      
          fields() :: Keyword.t()

      

  

  
  


  
    
      
      Link to this type
    
    filters()

      
       
       View Source
     
      
          filters() :: Keyword.t()

      

  

  
  


  
    
      
      Link to this type
    
    on_conflict()

      
       
       View Source
     
      
          on_conflict() ::
  {:raise, list(), []}
  | {:nothing, list(), [atom()]}
  | {[atom()], list(), [atom()]}
  | {Ecto.Query.t(), list(), [atom()]}

      

  

  
  


  
    
      
      Link to this type
    
    options()

      
       
       View Source
     
      
          options() :: Keyword.t()

      

  

  
  


  
    
      
      Link to this type
    
    returning()

      
       
       View Source
     
      
          returning() :: [atom()]

      

  

  
  


  
    
      
      Link to this type
    
    schema_meta()

      
       
       View Source
     
      
          schema_meta() :: %{
  autogenerate_id:
    {schema_field :: atom(), source_field :: atom(), Ecto.Type.t()},
  context: term(),
  prefix: binary() | nil,
  schema: atom(),
  source: binary()
}

      

  

  
Ecto.Schema metadata fields

      Link to this callback
    
    autogenerate(field_type)

      
       
       View Source
     
      
          autogenerate(field_type :: :id | :binary_id | :embed_id) :: term() | nil

      

  

  
Called to autogenerate a value for id/embed_id/binary_id.

Returns the autogenerated value, or nil if it must be
autogenerated inside the storage or raise if not supported.

  


  
    
      
      Link to this callback
    
    delete(adapter_meta, schema_meta, filters, options)

      
       
       View Source
     
      
          delete(adapter_meta(), schema_meta(), filters(), options()) ::
  {:ok, fields()} | {:invalid, constraints()} | {:error, :stale}

      

  

  
Deletes a single struct with the given filters.

While filters can be any record column, it is expected that
at least the primary key (or any other key that uniquely
identifies an existing record) be given as a filter. Therefore,
in case there is no record matching the given filters,
{:error, :stale} is returned.

  


  
    
      
      Link to this callback
    
    insert(adapter_meta, schema_meta, fields, on_conflict, returning, options)

      
       
       View Source
     
      
          insert(
  adapter_meta(),
  schema_meta(),
  fields(),
  on_conflict(),
  returning(),
  options()
) :: {:ok, fields()} | {:invalid, constraints()}

      

  

  
Inserts a single new struct in the data store.


  
  Autogenerate



The primary key will be automatically included in returning if the
field has type :id or :binary_id and no value was set by the
developer or none was autogenerated by the adapter.

  


  
    
      
      Link to this callback
    
    insert_all(adapter_meta, schema_meta, header, list, on_conflict, returning, options)

      
       
       View Source
     
      
          insert_all(
  adapter_meta(),
  schema_meta(),
  header :: [atom()],
  [[{atom(), term() | {Ecto.Query.t(), list()}}]],
  on_conflict(),
  returning(),
  options()
) :: {integer(), [[term()]] | nil}

      

  

  
Inserts multiple entries into the data store.

In case an Ecto.Query given as any of the field values by the user,
it will be sent to the adapter as a tuple with in the shape of
{query, params}.

  


  
    
      
      Link to this callback
    
    update(adapter_meta, schema_meta, fields, filters, returning, options)

      
       
       View Source
     
      
          update(
  adapter_meta(),
  schema_meta(),
  fields(),
  filters(),
  returning(),
  options()
) :: {:ok, fields()} | {:invalid, constraints()} | {:error, :stale}

      

  

  
Updates a single struct with the given filters.

While filters can be any record column, it is expected that
at least the primary key (or any other key that uniquely
identifies an existing record) be given as a filter. Therefore,
in case there is no record matching the given filters,
{:error, :stale} is returned.

  

        

      
  

  
    
    Ecto.Adapter.Storage - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Adapter.Storage behaviour    



      
Specifies the adapter storage API.

      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Callbacks
    


  
    storage_down(options)
  

    Drops the storage given by options.





  
    storage_status(options)
  

    Returns the status of a storage given by options.





  
    storage_up(options)
  

    Creates the storage given by options.




  

      

      
        
          
            
            Anchor for this section
          
Callbacks        

        

  
    
      
      Link to this callback
    
    storage_down(options)

      
       
       View Source
     
      
          storage_down(options :: Keyword.t()) ::
  :ok | {:error, :already_down} | {:error, term()}

      

  

  
Drops the storage given by options.

Returns :ok if it was dropped successfully.

Returns {:error, :already_down} if the storage has already been dropped or
{:error, term} in case anything else goes wrong.


  
  Examples



storage_down(username: "postgres",
             database: "ecto_test",
             hostname: "localhost")

  


  
    
      
      Link to this callback
    
    storage_status(options)

      
       
       View Source
     
      
          storage_status(options :: Keyword.t()) :: :up | :down | {:error, term()}

      

  

  
Returns the status of a storage given by options.

Can return :up, :down or {:error, term} in case anything goes wrong.


  
  Examples



storage_status(username: "postgres",
               database: "ecto_test",
               hostname: "localhost")

  


  
    
      
      Link to this callback
    
    storage_up(options)

      
       
       View Source
     
      
          storage_up(options :: Keyword.t()) ::
  :ok | {:error, :already_up} | {:error, term()}

      

  

  
Creates the storage given by options.

Returns :ok if it was created successfully.

Returns {:error, :already_up} if the storage has already been created or
{:error, term} in case anything else goes wrong.


  
  Examples



storage_up(username: "postgres",
           database: "ecto_test",
           hostname: "localhost")

  

        

      
  

  
    
    Ecto.Adapter.Transaction - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Adapter.Transaction behaviour    



      
Specifies the adapter transactions API.

      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Types
    


  
    adapter_meta()
  



  

  
    
      Callbacks
    


  
    in_transaction?(adapter_meta)
  

    Returns true if the given process is inside a transaction.





  
    rollback(adapter_meta, value)
  

    Rolls back the current transaction.





  
    transaction(adapter_meta, options, function)
  

    Runs the given function inside a transaction.




  

      

      
        
          
            
            Anchor for this section
          
Types        

        

  
    
      
      Link to this type
    
    adapter_meta()

      
       
       View Source
     
      
          adapter_meta() :: Ecto.Adapter.adapter_meta()

      

  

  
  

        

      
      
        
          
            
            Anchor for this section
          
Callbacks        

        

  
    
      
      Link to this callback
    
    in_transaction?(adapter_meta)

      
       
       View Source
     
      
          in_transaction?(adapter_meta()) :: boolean()

      

  

  
Returns true if the given process is inside a transaction.

  


  
    
      
      Link to this callback
    
    rollback(adapter_meta, value)

      
       
       View Source
     
      
          rollback(adapter_meta(), value :: any()) :: no_return()

      

  

  
Rolls back the current transaction.

The transaction will return the value given as {:error, value}.

See Ecto.Repo.rollback/1.

  


  
    
      
      Link to this callback
    
    transaction(adapter_meta, options, function)

      
       
       View Source
     
      
          transaction(adapter_meta(), options :: Keyword.t(), function :: (... -> any())) ::
  {:ok, any()} | {:error, any()}

      

  

  
Runs the given function inside a transaction.

Returns {:ok, value} if the transaction was successful where value
is the value return by the function or {:error, value} if the transaction
was rolled back where value is the value given to rollback/1.

  

        

      
  

  
    
    Ecto.Association.BelongsTo - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Association.BelongsTo    



      
The association struct for a belongs_to association.

Its fields are:


	cardinality - The association cardinality


	field - The name of the association field on the schema


	owner - The schema where the association was defined


	owner_key - The key on the owner schema used for the association


	related - The schema that is associated


	related_key - The key on the related schema used for the association


	queryable - The real query to use for querying association


	defaults - Default fields used when building the association


	relationship - The relationship to the specified schema, default :parent


	on_replace - The action taken on associations when schema is replaced




      


  

  
    
    Ecto.Association.Has - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Association.Has    



      
The association struct for has_one and has_many associations.

Its fields are:


	cardinality - The association cardinality


	field - The name of the association field on the schema


	owner - The schema where the association was defined


	related - The schema that is associated


	owner_key - The key on the owner schema used for the association


	related_key - The key on the related schema used for the association


	queryable - The real query to use for querying association


	on_delete - The action taken on associations when schema is deleted


	on_replace - The action taken on associations when schema is replaced


	defaults - Default fields used when building the association


	relationship - The relationship to the specified schema, default is :child




      


  

  
    
    Ecto.Association.HasThrough - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Association.HasThrough    



      
The association struct for has_one and has_many through associations.

Its fields are:


	cardinality - The association cardinality


	field - The name of the association field on the schema


	owner - The schema where the association was defined


	owner_key - The key on the owner schema used for the association


	through - The through associations


	relationship - The relationship to the specified schema, default :child




      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Functions
    


  
    on_repo_change(map, _, _, _, _)
  

    Performs the repository change on the association.




  

      

      
        
          
            
            Anchor for this section
          
Functions        

        

  
    
      
      Link to this function
    
    on_repo_change(map, _, _, _, _)

      
       
       View Source
     
  

  
Performs the repository change on the association.

Receives the parent changeset, the current changesets
and the repository action options. Must return the
persisted struct (or nil) or the changeset error.

Callback implementation for c:Ecto.Association.on_repo_change/5.

  

        

      
  

  
    
    Ecto.Association.ManyToMany - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Association.ManyToMany    



      
The association struct for many_to_many associations.

Its fields are:


	cardinality - The association cardinality


	field - The name of the association field on the schema


	owner - The schema where the association was defined


	related - The schema that is associated


	owner_key - The key on the owner schema used for the association


	queryable - The real query to use for querying association


	on_delete - The action taken on associations when schema is deleted


	on_replace - The action taken on associations when schema is replaced


	defaults - Default fields used when building the association


	relationship - The relationship to the specified schema, default :child


	join_keys - The keyword list with many to many join keys


	join_through - Atom (representing a schema) or a string (representing a table)
for many to many associations


	join_defaults - A list of defaults for join associations




      


  

  
    
    Ecto.Association.NotLoaded - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Association.NotLoaded    



      
Struct returned by associations when they are not loaded.

The fields are:


	__field__ - the association field in owner


	__owner__ - the schema that owns the association


	__cardinality__ - the cardinality of the association




      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Types
    


  
    t()
  



  

      

      
        
          
            
            Anchor for this section
          
Types        

        

  
    
      
      Link to this type
    
    t()

      
       
       View Source
     
      
          t() :: %Ecto.Association.NotLoaded{
  __cardinality__: atom(),
  __field__: atom(),
  __owner__: any()
}

      

  

  
  

        

      
  

  
    
    Ecto.CastError - Ecto v3.3.2
    
    

    

  
  
    
Ecto.CastError exception    



      
Raised when a changeset can't cast a value.

      


  

  
    
    Ecto.ChangeError - Ecto v3.3.2
    
    

    

  
  
    
Ecto.ChangeError exception    





  

  
    
    Ecto.ConstraintError - Ecto v3.3.2
    
    

    

  
  
    
Ecto.ConstraintError exception    





  

  
    
    Ecto.InvalidChangesetError - Ecto v3.3.2
    
    

    

  
  
    
Ecto.InvalidChangesetError exception    



      
Raised when we cannot perform an action because the
changeset is invalid.

      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Functions
    


  
    message(map)
  

    Callback implementation for Exception.message/1.




  

      

      
        
          
            
            Anchor for this section
          
Functions        

        

  
    
      
      Link to this function
    
    message(map)

      
       
       View Source
     
  

  
Callback implementation for Exception.message/1.

  

        

      
  

  
    
    Ecto.InvalidURLError - Ecto v3.3.2
    
    

    

  
  
    
Ecto.InvalidURLError exception    





  

  
    
    Ecto.MigrationError - Ecto v3.3.2
    
    

    

  
  
    
Ecto.MigrationError exception    





  

  
    
    Ecto.MultiplePrimaryKeyError - Ecto v3.3.2
    
    

    

  
  
    
Ecto.MultiplePrimaryKeyError exception    





  

  
    
    Ecto.MultipleResultsError - Ecto v3.3.2
    
    

    

  
  
    
Ecto.MultipleResultsError exception    





  

  
    
    Ecto.NoPrimaryKeyFieldError - Ecto v3.3.2
    
    

    

  
  
    
Ecto.NoPrimaryKeyFieldError exception    



      
Raised at runtime when an operation that requires a primary key is invoked
with a schema that does not define a primary key by using @primary_key false

      


  

  
    
    Ecto.NoPrimaryKeyValueError - Ecto v3.3.2
    
    

    

  
  
    
Ecto.NoPrimaryKeyValueError exception    



      
Raised at runtime when an operation that requires a primary key is invoked
with a schema missing value for its primary key

      


  

  
    
    Ecto.NoResultsError - Ecto v3.3.2
    
    

    

  
  
    
Ecto.NoResultsError exception    





  

  
    
    Ecto.Query.CastError - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Query.CastError exception    



      
Raised at runtime when a value cannot be cast.

      


  

  
    
    Ecto.Query.CompileError - Ecto v3.3.2
    
    

    

  
  
    
Ecto.Query.CompileError exception    



      
Raised at compilation time when the query cannot be compiled.

      


  

  
    
    Ecto.QueryError - Ecto v3.3.2
    
    

    

  
  
    
Ecto.QueryError exception    



      
Raised at runtime when the query is invalid.

      


  

  
    
    Ecto.StaleEntryError - Ecto v3.3.2
    
    

    

  
  
    
Ecto.StaleEntryError exception    





  

  
    
    Ecto.SubQueryError - Ecto v3.3.2
    
    

    

  
  
    
Ecto.SubQueryError exception    



      
Raised at runtime when a subquery is invalid.

      


  

  
    
    mix ecto - Ecto v3.3.2
    
    

    

  
  
    
mix ecto    



      
Prints Ecto tasks and their information.

mix ecto

      


  

  
    
    mix ecto.create - Ecto v3.3.2
    
    

    

  
  
    
mix ecto.create    



      
Create the storage for the given repository.

The repositories to create are the ones specified under the
:ecto_repos option in the current app configuration. However,
if the -r option is given, it replaces the :ecto_repos config.

Since Ecto tasks can only be executed once, if you need to create
multiple repositories, set :ecto_repos accordingly or pass the -r
flag multiple times.

Examples

mix ecto.create
mix ecto.create -r Custom.Repo

Command line options


	-r, --repo - the repo to create


	--quiet - do not log output


	--no-compile - do not compile before dropping


	--no-deps-check - do not compile before dropping




      


  

  
    
    mix ecto.drop - Ecto v3.3.2
    
    

    

  
  
    
mix ecto.drop    



      
Drop the storage for the given repository.

The repositories to drop are the ones specified under the
:ecto_repos option in the current app configuration. However,
if the -r option is given, it replaces the :ecto_repos config.

Since Ecto tasks can only be executed once, if you need to drop
multiple repositories, set :ecto_repos accordingly or pass the -r
flag multiple times.

Examples

mix ecto.drop
mix ecto.drop -r Custom.Repo

Command line options


	-r, --repo - the repo to drop


	-q, --quiet - run the command quietly


	-f, --force - do not ask for confirmation when dropping the database.
Configuration is asked only when :start_permanent is set to true
(typically in production)


	--no-compile - do not compile before dropping


	--no-deps-check - do not compile before dropping




      


  

  
    
    mix ecto.gen.repo - Ecto v3.3.2
    
    

    

  
  
    
mix ecto.gen.repo    



      
Generates a new repository.

The repository will be placed in the lib directory.

Examples

mix ecto.gen.repo -r Custom.Repo

This generator will automatically open the config/config.exs
after generation if you have ECTO_EDITOR set in your environment
variable.

Command line options


	-r, --repo - the repo to generate

